

40 programmes en Basic pour la maison

AMSTRAD EN FAMILLE



Jean-François Séhan

**AMSTRAD
EN FAMILLE**

CONNAISSEZ-VOUS TOUTE LA COLLECTION AMSTRAD CHEZ P.S.I. ?

Pour les Amstrad CPC 464, 664 et 6128

Initiation :

- La découverte de l'Amstrad - Daniel-Jean David
- Exercices en BASIC pour Amstrad - Maurice Charbit

Programmation BASIC :

- Super jeux pour Amstrad - Jean-François Sehan
- 102 programmes pour Amstrad - Jacques Deconchat
- Super générateur de caractères sur Amstrad - Jean-François Sehan
- Photographie sur Amstrad et Apple II - Pierrick Moigneau et Xavier de la Tullaye
- Amstrad en musique - Daniel Lemahieu

Maîtrise du BASIC :

- BASIC Amstrad - 1. Méthodes pratiques - Jacques Boisgontier et Bruno César
- BASIC Amstrad - 2. Programmes et fichiers - Jacques Boisgontier
- BASIC Plus, 80 routines sur Amstrad - Michel Martin
- Périphériques et fichiers sur Amstrad - Daniel-Jean David

Assembleur et Pascal :

- Assembleur de l'Amstrad - Marcel Henrot
- Turbo Pascal sur Amstrad - Pierre Brandeis et Frédéric Blanc

Système :

- Clefs pour Amstrad - 1. Système de base - Daniel Martin
- CP/M Plus sur Amstrad 6128 et 8256 - Yvon Dargery
- Clefs pour Amstrad - 2. Système disque - Daniel Martin et Philippe Jadoul

A paraître :

- Intelligence artificielle : langage et formes sur Amstrad - Thierry Lévy-Abégnoilli et Olivier Magnan
- Animation graphique sur Amstrad - Gilles Fouchard et Jean-Yves Corre
- Simulation et intelligence artificielle sur Amstrad - René Descamps
- Le livre de l'Amstrad 8256 - Eric Baumart

Pour tout problème rencontré dans les ouvrages P.S.I.
vous pouvez nous contacter au numéro ci-dessous :

Numéro Vert/Appel Gratuit en France

05 21 22 01

(Composer tous les chiffres, même en région parisienne)

La loi du 11 mars 1957 n'autorisant, aux termes des alinéas 2 et 3 de l'article 41, d'une part, que les « copies ou reproductions strictement réservées à l'usage privé du copiste et non destinées à une utilisation collective », et, d'autre part, que les analyses et les courtes citations dans un but d'exemple et d'illustration, « toute représentation ou reproduction intégrale, ou partielle, faite sans le consentement de l'auteur ou de ses ayants droit ou ayants cause, est illicite » (alinéa 1^{er} de l'article 40).

Cette représentation ou reproduction, par quelque procédé que ce soit, constituerait donc une contrefaçon sanctionnée par les articles 425 et suivants du Code Pénal.

© Éditions du P.S.I. - B.P. 86 - 77402 Lagny/Marne cedex
1986

ISBN 2-86595-240-1

**40 programmes en Basic
pour la maison**

AMSTRAD EN FAMILLE

Jean-François SEHAN



**ÉDITIONS DU P.S.I.
1986**

Du même auteur aux éditions du P.S.I.

- Etudes pour ZX-81 tome I
 - Etudes pour ZX-81 tome II
 - Suites pour PC-1500
 - Le TI-99/4A à l'affiche
 - Clefs pour ZX-81
 - Le VIC à l'affiche
 - Clefs pour ZX SPECTRUM
 - L'ORIC à l'affiche
 - Le SPECTRUM à l'affiche
 - Le COMMODORE 64 à l'affiche
 - Le MO5 et TO7/70 à l'affiche
 - Le MO5 et TO7/70 en famille
 - Le COMMODORE 64 en famille
 - Le MSX en famille
 - Super-jeux pour MSX
 - Super-jeux Amstrad
 - Super-jeux YENO
 - Apple en famille
 - Super-jeux Apple
-

AVERTISSEMENT

Si vous voulez utiliser correctement les programmes de ce livre, lisez attentivement les conseils suivants :

- Avant de taper un programme, vous devez initialiser Amstrad en appuyant simultanément sur les touches [SHIFT], [CTRL] et [ESC].
Effectuez la même opération avant de charger un programme à partir du datacorder.
- Pour gagner en rapidité d'exécution certaines instructions NEXT ne sont pas suivies d'une variable (gain de 10 à 20 %).
- Si vous n'obtenez pas les mêmes résultats que sur les exemples, vérifiez bien toutes les lignes et surtout celles contenant des DATAs.
- Dès que les programmes sont entrés dans votre AMSTRAD, sauvez-les sur le datacorder. Pour les programmes très longs, il est vivement conseillé d'effectuer une première sauvegarde au milieu de leur frappe. N'attendez pas qu'ils soient entièrement entrés et testés... une coupure de courant est si vite arrivée !
- Quand vous entrez des chiffres décimaux, assurez-vous que vous tapez un point et non une virgule entre deux chiffres.
- Assurez-vous que vous êtes en majuscules dès la mise en route, en appuyant sur CAPS LOCK.

L'auteur remercie la société VISMO pour son aide technique.

SOMMAIRE

Thème 1 - LES FINANCES

Compte bancaire ou postal	10
Placement chez l'écureuil	17
Budget familial	23
Coût annuel de votre voiture	28
Calcul d'impôts	33

Thème 2 - PÉDAGOGIE

Musique	42
Géographie	49
Apprendre à taper au clavier	56
Alphabet	60
Apprendre à compter	64
Tables de multiplication	69
Les quatre opérations	73
Questions-réponses	77
Conjugaisons	84
Suites logiques	90

Thème 3 - CUISINE ET TRICOT

Fichier de recettes	96
Recettes de cuisine	102
Menus aléatoires	108
Gestion d'un congélateur	113
Tricot	119

Thème 4 - LES JEUX NATIONAUX

Le Tiercé : pronostics	124
Le Loto : grilles aléatoires	130
Le Loto : statistiques	134

Thème 5 - LE TEMPS

Arbre généalogique	142
Astrologie	150
Astrologie chinoise	155
Espérance de vie	160

Biorythme	166
Jour de la semaine	170
Calendrier	175
Horloge	181
Agenda électronique	185

Thème 6 - VOTRE FORME


Poids idéal	192
Courbes poids/taille	196

Thème 7 - BRICOLAGE

Papier peint	202
Disposition des meubles	206

Thème 8 - UTILITAIRES

Cryptogrammes	214
Gestion des QSO	218
Gestion d'une cave	224
Répertoire téléphonique	230



THEME 1 LES FINANCES


COMPTE BANCAIRE

Plus qu'une simple gestion de compte bancaire, ce programme vous donne l'occasion de connaître vos dépenses en nourriture, loyer, vêtements, essence, etc., mais aussi le total des dépenses et des rentrées de l'année.

Si vous l'utilisez pour la première fois, AMSTRAD vous demande le montant du compte actuellement ou au 1^{er} janvier si vous créez une nouvelle année. Puis un menu de six options s'affiche.

- 1 – **SAISIE** : vous devez entrer tous les mouvements en donnant des valeurs négatives aux sommes retirées (chèques, virements, carte de crédit). La date, le numéro et la description sont à titre indicatif, mais il est très intéressant de donner des noms par groupe pour faciliter les recherches. Par exemple si vous donnez les descriptions "ESSENCE VOITURE", "VIDANGE VOITURE", "CREDIT VOITURE", vous pourrez obtenir le total des dépenses pour votre voiture mais aussi les dépenses uniquement d'essence ou de vidanges ainsi que la somme de tous les crédits (pas uniquement celui de la voiture).
- 2 – **MODIFICATION** : si vous avez fait une erreur dans la saisie, il est toujours possible de la modifier en utilisant cette option. Vous devez donner le numéro de l'opération (si vous l'avez oublié utilisez les options 4 et 5 puis le numéro de l'élément à modifier). Quand vous ne désirez plus effectuer de modification, tapez "0" à la question "Numéro de l'opération".
- 3 – **SUPPRESSION** : supprime une des opérations déjà entrées. Le programme demande confirmation avant la suppression (répondre par "O" ou "N"). Comme pour la modification il faut répondre "0" pour revenir au menu.
- 4 – **RECHERCHE** : le programme recherche dans tout le fichier le nom, le numéro ou la date que vous lui donnez. Par exemple le nom "CREDIT" affiche et calcule le total des sommes concernant des crédits (crédits logement, voiture, etc.).
- 5 – **LISTING** : cette opération permet de lister sur l'écran la totalité des mouvements effectués sur le compte. Le listing s'arrête automatiquement en bas de page. Vous devez appuyer sur n'importe quelle touche pour continuer.
- 6 – **SAUVEGARDE** : dès que vous n'avez plus besoin du programme, cette option sauve sur la cassette la totalité du fichier.

COMPTE BANCAIRE



```

10 REM COMPTE BANCAIRE OU POSTAL
20 REM -----
30 DIM D$(4,500)
40 L$(1)="MONTANT":L$(2)="DATE":L$(3)="NUMERO OU R
EPERE":L$(4)="DESCRIPTION"
50 L(2)=8:L(3)=7:L(4)=7
60 MODE 1:INK 0,0:INK 1,15:BORDER 0
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$<>"0" THEN 110
90 CLS:INPUT"MONTANT INITIAL DU COMPTE ";M
100 GOTO 300
110 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
120 OPENIN "!COMPTE"
130 INPUT#9,N,C,D,M
140 FOR I=1 TO N
150 FOR J=1 TO 4
160 INPUT#9,D$(J,I)
170 NEXT J,I
180 CLOSEIN
280 REM -----
290 REM MENU
300 CLS
310 PRINT"SOLDE : ";M+C-D:PRINT"CREDIT: ";C:PRINT"DE
BIT : ";D
320 PRINT"NB D'OPERATIONS: ";N:PRINT:PRINT
330 PRINT"1 SAISIE"
340 PRINT"2 MODIFICATION"
350 PRINT"3 SUPPRESSION"
360 PRINT"4 RECHERCHE"
370 PRINT"5 LISTING"
380 PRINT"6 SAUVEGARDE"
390 PRINT:PRINT"VOTRE CHOIX ?";
400 R$=INKEY$:IF R$="" THEN 400
410 R=VAL(R$):IF R<1 OR R>6 THEN 400
420 ON R GOTO 500,1000,1200,700,1500,1700
480 REM -----
490 REM SAISIE
500 CLS:PRINT"SAISIE D'UNE OPERATION":PRINT
510 IF N=500 THEN PRINT"FICHER PLEIN";:LINE INPUT
R$:GOTO 300
520 N=N+1
530 FOR I=1 TO 4
540 PRINT L$(I);": ";:INPUT" ",D$(I,N)
550 IF I=1 AND VAL(D$(1,N))=0 THEN N=N-1:GOTO 300
560 IF I=2 AND LEN(D$(2,N))<>8 THEN 540
570 NEXT I
580 IF VAL(D$(1,N))<0 THEN D=D+ABS(VAL(D$(1,N))) E
LSE C=C+VAL(D$(1,N))
590 GOTO 500
680 REM -----
690 REM RECHERCHE
700 CLS:F=0:T=0
710 PRINT"RECHERCHE PAR : "
720 FOR I=1 TO 4:PRINT I;L$(I):NEXT
730 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
740 R$=INKEY$:IF R$="" THEN 740
750 R=VAL(R$):IF R<1 OR R>4 THEN 300
760 PRINT L$(R);": ";:INPUT" ",D$:PRINT
770 FOR I=1 TO N

```

```

780 IF INSTR(D$(R,I),D$)=0 THEN 840
790 F=1:PRINT USING"### ";I;
800 FOR J=1 TO 4
810 IF J=1 THEN PRINT USING"#####.## ";VAL(D$(J,I
)); ELSE PRINT LEFT$(D$(J,I)+"
",L(J));" ";
820 NEXT J:PRINT
830 T=T+VAL(D$(1,I))
840 NEXT I:PRINT"-----"
850 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE" ELSE PR
INT USING"#####.##";T
860 IF INKEY$="" THEN 860 ELSE 700
980 REM -----
990 REM MODIFICATION
1000 CLS:PRINT"MODIFICATION":PRINT
1010 INPUT"NUMERO DE L'OPERATION";NU
1020 IF NU<1 OR NU>N THEN 300
1030 FOR I=1 TO 4
1040 PRINT I;L$(I);": ";D$(I,NU)
1050 NEXT I
1060 PRINT:INPUT"MODIFICATION DU";R
1070 IF R<1 OR R>4 THEN 300
1080 IF R=1 THEN A=VAL(D$(1,NU))
1090 PRINT L$(R);:INPUT D$(R,NU)
1100 IF R=1 AND VAL(D$(1,NU))=0 THEN 1090
1110 IF R<>1 THEN 1000
1120 IF A<0 THEN D=D+A ELSE C=C-A
1130 IF VAL(D$(1,NU))<0 THEN D=D-VAL(D$(1,NU)) ELS
E C=C+VAL(D$(1,NU))
1140 GOTO 1000
1180 REM -----
1190 REM SUPPRESSION
1200 CLS:PRINT"SUPPRESSION":PRINT
1210 INPUT"NUMERO DE L'OPERATION";NU
1220 IF NU<1 OR NU>N THEN 300
1230 PRINT
1240 FOR I=1 TO 4
1250 PRINT I;L$(I);": ";D$(I,NU)
1260 NEXT I:PRINT
1270 INPUT"VOUS ETES SUR DE SUPPRIMER
CETTE OPERATION";R$
1280 IF R$<>"0" THEN 1200
1290 A=VAL(D$(1,NU))
1300 IF A<0 THEN D=D+A ELSE C=C-A
1310 N=N-1
1320 FOR I=NU TO N
1330 FOR J=1 TO 4
1340 D$(J,I)=D$(J,I+1)
1350 NEXT J,I
1360 GOTO 1200
1480 REM -----
1490 REM LISTING
1500 MODE 2:L(4)=40
1510 FOR I=1 TO N
1520 PRINT USING"### ";I;
1530 FOR J=1 TO 4
1540 IF J=1 THEN PRINT USING"#####.## ";VAL(D$(J,
I)); ELSE PRINT LEFT$(D$(J,I)+"
",L(J));" ";
1550 NEXT J:PRINT

```

```

1560 IF I/20=INT(I/20) THEN PRINT:PRINT"APPUYEZ SU
R UNE TOUCHE POUR LA SUITE";:WHILE INKEY$="":WEND:
CLS
1570 NEXT I
1580 PRINT:PRINT"    APPUYEZ SUR UNE TOUCHE POUR L
E MENU";
1590 IF INKEY$="" THEN 1590 ELSE MODE 1:L(4)=7:GOT
O 300
1680 REM -----
1690 REM SAUVEGARDE
1700 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORD
ER EST PRET POUR L'ENREGISTREMENT"
1710 LINE INPUT R$
1720 OPENOUT "!COMPTE"
1730 PRINT#9,N,C,D,M
1740 FOR I=1 TO N
1750 FOR J=1 TO 4
1760 PRINT#9,DS(J,I)
1770 NEXT J,I
1780 CLOSEOUT
1790 CLS:PRINT"A LA PROCHAINE OPERATION !"
1800 END

```

```

RECHERCHE PAR :
1 MONTANT
2 DATE
3 NUMERO OU REPERE
4 DESCRIPTION

QUEL TYPE DE DONNEE ?
DESCRIPTION:VOITURE

1   -218.00  08/01/85  123100  ESSENCE
2   -225.00  15/01/85  123101  VIDANGE
3   -435.67  05/02/85  123115  REVISIO
4   -187.00  10/02/85  123124  ESSENCE
-----
=  -1065.67

```

LE PROGRAMME

Lignes 30 à 180 : initialisation

On place dans le tableau L\$() les libellés des quatre données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 300 à 420 : menu

Ces lignes affichent le solde du compte, le total des crédits et des débits, puis ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 500 à 590 : saisie d'une opération

On saisit par une boucle FOR/NEXT les quatre données du fichier et l'on met à jour le total crédit ou débit selon le cas (ligne 580).

Si aucun montant n'a été entré, on revient au menu (ligne 550).

Lignes 700 à 860 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$).

Les lignes 770 à 840 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$. La variable T totalise les montants.

Lignes 1000 à 1140 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU). Si c'est le montant qui doit être changé, les lignes 1120 et 1130 mettent à jour le total des crédits et des débits.

Lignes 1200 à 1360 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1320 à 1350 décalent le tableau vers le bas.

Lignes 1500 à 1590 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1560 dès que l'écran est plein.

Lignes 1700 à 1800 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A	ancienne valeur du montant pour la modification
C	total des crédits
D	total des débits
D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. si F=0, aucun enregistrement n'a été trouvé
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	longueur maximum d'affichage de chaque donnée
L\$()	libellés des données
M	montant initial du compte
N	nombre d'opérations dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$
T	total des montants dans la recherche

PLACEMENT CHEZ L'ECUREUIL

Que rapportent les économies que vous avez placées à la Caisse d'Épargne ? Laissez le soin à votre AMSTRAD d'en effectuer les calculs.

Le programme demande chaque quinzaine le total des dépôts et des retraits que vous avez effectués sur votre livret. En effet, les intérêts de la Caisse d'Épargne sont calculés tous les quinze jours (mais déposés sur le compte seulement le 31 décembre). Tous les retraits comptent à partir du premier jour de la quinzaine, les dépôts le premier jour de la quinzaine suivante. Il est préférable de se méfier pour les chèques "Reçu sous réserve d'encaissement" car ils prennent plusieurs jours avant d'être réellement déposés sur le livret. Un chèque déposé le 14 est encaissé après le 15, il comptera donc seulement dans les intérêts du mois suivant.


En plus des dépôts et des retraits, le programme demande aussi le taux d'intérêt, celui-ci pouvant être modifié d'une quinzaine à l'autre (aux bons soins de monsieur le ministre des Finances).

Ces formalités étant remplies, AMSTRAD affiche par quinzaine le capital, les dépôts, les retraits, le capital sur lequel sont calculés les intérêts et les intérêts eux-mêmes.

DATE	CAPIT	DEPOT	RETR.	I SUR	INTERETS
15/ 1	9057	0	1000	9057	24.53
31/ 1	9057	0	0	9057	24.53
15/ 2	6557	0	2500	6557	17.76
31/ 2	5357	0	1200	5357	14.51
15/ 3	9818	5461	1000	5357	14.51
31/ 3	9818	0	0	9818	26.59
15/ 4	9818	0	0	9818	26.59
31/ 4	9818	0	0	9818	26.59
15/ 5	14318	4500	0	9818	26.59
31/ 5	13818	0	500	13818	37.42
15/ 6	14568	1250	500	13818	37.42
31/ 6	16368	2300	0	14568	39.46
15/ 7	16368	1000	1500	16368	44.33
31/ 7	14868	0	1500	14868	40.27
15/ 8	13668	0	1200	13668	37.02
31/ 8	11668	0	2000	11668	31.60
TOTAL INTERETS:					469.72

FIN (O/N) ?

PLACEMENT CHEZ L'ÉCUREUIL



```

10 REM PLACEMENT CHEZ L'ECUREUIL
20 REM -----
30 DIM D(24),R(24),M$(12),C(24),T(24)
40 FOR I=1 TO 12:READ M$(I):NEXT I
50 DATA JANVIER,FEVRIER,MARS,AVRIL,MAI,JUIN
60 DATA JUILLET,AOUT,SEPTEMBRE,OCTOBRE,NOVEMBRE,DE
CEMBRE
70 MODE 1:INK 0,0:INK 1,15:BORDER 0
80 INPUT"PREMIERE UTILISATION (O/N)";R$
90 IF R$="0" THEN 200
100 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
110 OPENIN"!ECUREUIL"
120 INPUT#9,M,C
130 FOR I=1 TO 24
140 INPUT#9,D(I),R(I),T(I)
150 NEXT I
160 CLOSEIN
180 REM -----
190 REM DONNEES DU MOIS
200 CLS
210 IF M<>0 THEN 230
220 INPUT"CAPITAL AU 1er JANVIER (AVEC INTERETS DE
L'ANNEE PRECEDENTE) ";C
230 C(0)=C
240 FOR I=1 TO M
250 C(I)=C(I-1)-R(I)+D(I)
260 NEXT I
270 IF M=24 THEN 500
280 M=M+1
290 PRINT
300 PRINT"CAPITAL ACTUEL:";C(M-1)
310 PRINT"DONNEZ LES MOUVEMENTS DU : "
320 IF M/2=INT(M/2) THEN PRINT"15 AU 31 "; ELSE PR
INT"1er AU 14 ";
330 PRINT M$(INT(0.5+M/2))
340 PRINT
350 INPUT"DEPOT ";D(M)
360 INPUT"RETRAIT ";R(M)
370 INPUT"TAUX ";T$:IF T$="" THEN T(M)=T(M-1) ELSE
T(M)=VAL(T$)/100
380 IF INT(C(M-1)+D(M)-R(M))<0 THEN PRINT"IMPOSSIB
LE: CAPITAL < 0":GOTO 290
390 PRINT
400 C(M)=C(M-1)+D(M)-R(M)
480 REM -----
490 REM RESULTATS
500 TT=0
510 CLS:PRINT:PRINT
520 FOR I=1 TO M
530 IF R(I)>D(I) THEN R=R(I)-D(I) ELSE R=0
540 CA=C(I-1)-R
550 T=(CA/24)*T(I)
560 TT=TT+T
570 IF I/2=INT(I/2) THEN PRINT"31/"; ELSE PRINT"15
/";
580 PRINT USING"###";INT(0.5+I/2);
590 PRINT USING":#####";C(I),D(I),R(I),CA;
600 PRINT USING":####.##";T
610 NEXT I

```

```

620 PRINT"-----:-----"
630 PRINT TAB(16);"TOTAL INTERETS:";:PRINT USING"#
###.##";TT:PRINT
640 LOCATE 1,1:PRINT"DATE :CAPIT:DEPOT:RETR.:I SUR
:INTERETS"
650 PRINT"-----:-----:-----:-----"
660 LOCATE 1,25:PRINT"FIN (0/N) ?";
670 R$=INKEY$:IF R$="" THEN 670 ELSE IF R$="N" THE
N 200
680 REM -----
690 REM SAUVEGARDE
700 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R EST PRET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 OPENOUT "!ECUREUIL"
730 PRINT#9,M,C
740 FOR I=1 TO 24
750 PRINT#9,D(I),R(I),T(I)
760 NEXT I
770 CLOSEOUT
780 CLS:PRINT"A LA PROCHAINE QUINZAINE !"
790 END

```

LE PROGRAMME

Lignes 30 à 160 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 400 : données du mois

S'il s'agit de la première utilisation ($M=0$), la ligne 220 saisit le capital au 1^{er} janvier. Les lignes 240 à 260 placent dans le tableau C() le capital du mois (capital du mois précédent – les retraits + les dépôts du mois).

Si le fichier est déjà plein ($M=24$), la ligne 270 renvoie aux résultats sans passer par la saisie.

Le programme demande les dépôts, les retraits et le taux d'intérêts de la quinzaine. Si vous ne donnez aucune valeur pour ce dernier, on prend automatiquement le taux du mois précédent (ligne 370).

Lignes 500 à 670 : résultats

Ces lignes affichent un tableau récapitulatif des dépôts et des retraits et calculent le capital et les intérêts. La ligne 630 donne le total des intérêts annuels.

La ligne 670 renvoie à la saisie si l'on désire entrer les données de la quinzaine suivante avant la sauvegarde.

Lignes 700 à 790 : sauvegarde

On sauve sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D()	liste des dépôts
C	capital au 1 ^{er} janvier
C()	capital par quinzaine
CA	capital pour le calcul des intérêts
I	indice de boucle FOR/NEXT
M	numéro de la quinzaine en cours
MS()	noms des 12 mois
R()	liste des retraits
RS	réponse de l'utilisateur
T	intérêts de la quinzaine
T()	liste des taux par quinzaine
TT	total des intérêts

BUDGET FAMILIAL


A partir du total de vos dépenses et des dépenses par poste que vous lui donnez, AMSTRAD vous indiquera le pourcentage du budget que vous consacrez aux impôts, au loyer, à la voiture, aux vacances, aux vêtements, à la nourriture, aux fêtes, aux loisirs et aux assurances.

Au dernier montant entré, l'ordinateur dessine l'histogramme de votre budget familial en identifiant chaque rectangle. Les pourcentages sont rappelés à droite de l'écran.


Au cas où vous voudriez modifier une somme, vous devez taper sur ENTER. Le questionnaire qui apparaît n'est vide qu'en apparence car il contient toujours les données précédentes.


Il suffit de taper un nouveau montant pour les modifications éventuelles ou de passer au poste suivant en tapant simplement ENTER s'il n'y a rien à changer. Puis, comme dans le premier cas, l'ordinateur redessine l'histogramme corrigé.

C'est un moyen rapide d'avoir une vue d'ensemble sur vos dépenses et vous aurez peut-être des surprises.



BUDGET FAMILIAL





```

10 REM BUDGET FAMILIAL
20 REM -----
30 DATA IMPOTS,LOYER,VOITURE,VACANCES,VETEMENTS,NO
URRITURE
40 DATA FETE,LOISIRS,ASSURANCES,DIVERS
50 FOR I=1 TO 10
60 READ N$(I)
70 NEXT I
80 REM -----
90 REM SAISIE DES DEPENSES
100 MODE 1:INK 0,0:INK 1,15:BORDER 0
110 INPUT"TOTAL DES DEPENSES ";R$
120 IF VAL(R$)<>0 THEN TD=VAL(R$)
130 IF TD=0 THEN 100
140 MAXI=0:DT=0
150 PRINT:PRINT
160 FOR I=1 TO 9
170 PRINT N$(I);
180 INPUT R$
190 IF VAL(R$)<>0 THEN D(I)=VAL(R$)
200 IF D(I)<MAXI THEN 220
210 MAXI=D(I)
220 DT=DT+D(I)
230 IF DT<=TD THEN 260
240 PRINT"TOTAL > AUX DEPENSES"
250 TD=DT
260 NEXT I
270 D(10)=TD-DT
280 IF D(10)>MAXI THEN MAXI=D(10)
290 FOR I=1 TO 10
300 P(I)=INT((D(I)/(MAXI/100)))*3
310 NEXT I
380 REM -----
390 REM HISTOGRAMME
400 MODE 1:PRINT"^";TAB(25)"DEPENSES      %"
410 PLOT 6,390:DRAW 6,50:PLOT 8,390:DRAW 8,50
420 PLOT 6,50:DRAW 500,50:PLOT 6,52:DRAW 500,52
430 FOR I=1 TO 10
440 IF P(I)=0 THEN 500
450 FOR J=60 TO 60+P(I) STEP 2
460 MOVE I*32-10,J:DRAW 5+I*32,J
470 NEXT J
480 MOVE 5+I*32,60:DRAW 10+I*32,65:DRAW 10+I*32,J+
5
490 MOVE I*32-10,J:DRAWR 5,5:DRAWR 15,0:DRAWR -5,-
5
500 LOCATE 22,I+1:PRINT USING"## ";I;
510 PRINT N$(I):LOCATE 36,I+1:PRINT USING": ##";IN
T(D(I)/(TD/100))
520 LOCATE I*2-1,23:PRINT I
530 NEXT I
540 IF INKEYS<>CHR$(13) THEN 540
550 GOTO 100

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

On place dans le tableau **NS()** les libellés des dépenses. Ces derniers se trouvant placés dans les DATAs des lignes 30 et 40, il est possible de les modifier pour d'autres dépenses si vous n'avez pas de voiture ou si vous ne partez jamais en vacances.

Lignes 100 à 310 : saisie des dépenses

La variable **MAX** garde en mémoire la valeur de la plus grande dépense effectuée. Elle servira de référence dans l'histogramme.

La variable **DT** garde la somme totale des 9 dépenses. Les dépenses diverses prendront alors la différence entre le total des dépenses (ce qui est réellement sorti de votre portefeuille) et la somme des 9 autres dépenses.

Dans le cas où DT devient supérieur à TD, on modifie TD pour le réajuster ; les dépenses diverses deviennent obligatoirement nulles.

On calcule pour chaque dépense un pourcentage pour donner une idée de la valeur que l'on vient de rentrer au clavier.

Lignes 400 à 550 : histogramme

Les lignes 400 à 530 affichent l'histogramme et les pourcentages de chaque dépense (lignes 500 et 510). L'ordonnée des dépenses est affichée à la ligne 520.

Dès qu'une touche est enfoncée, on revient à la ligne 100 pour une nouvelle saisie ou pour la modification des dernières données.

LISTE DES VARIABLES :

D()	tableau des dépenses
DT	dépenses totales
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
MAX	valeur maximale dans l'histogramme
NS()	libellé des dépenses
P()	pourcentage de chaque dépense
RS	dernière dépense tapée au clavier
TD	total des dépenses

COUT ANNUEL DE VOTRE VOITURE

Que coûte votre voiture chaque année ? Quelles sommes avez-vous dépensées pour l'entretien et l'essence ? Quel est le prix de revient d'un petit kilomètre ? Autant de questions auxquelles votre AMSTRAD répondra avec plaisir.


Le programme demande tous les mois le nombre de kilomètres au compteur et les dépenses en essence, en entretien et en mécanique ainsi que les remboursements éventuels d'un crédit (entrer "0" si vous n'avez pas de crédit).

Si vous utilisez le programme pour la première fois (ou une autre année), il demande aussi le prix de l'assurance et de la vignette, et le nombre de kilomètres au compteur au 1^{er} janvier.

Avec ces quelques renseignements, AMSTRAD affiche pour chaque mois la distance parcourue, le total des dépenses, le prix du kilomètre et en supplément gratuit le prix annuel du kilomètre.

COUT DE VOTRE VOITURE						
ASSURANCE: 5600				VIGNETTE: 1000		
MOIS	KM	ESS.	MECA	CRED.	TOT.	TOT/K
1	450	210	0	0	760	1.69
2	375	182	120	0	852	2.27
3	290	152	0	0	702	2.42
4	210	0	225	0	775	3.69
5	464	205	0	0	755	1.63
6	212	145	0	0	695	3.28
7	0	0	0	0	0	0
8	0	0	0	0	0	0
9	0	0	0	0	0	0
10	0	0	0	0	0	0
11	0	0	0	0	0	0
12	0	0	0	0	0	0
TOTAL/KM		ESS/F	MECA/F	CR/F	TOT/F	
2001		894	345	0	4539	
PRIX MOYEN DU KM: 2.27						

COUT ANNUEL DE VOTRE VOITURE



```

10 REM COUT ANNUEL DE VOTRE VOITURE
20 REM -----
30 DIM E(12),M(12),K(12),C(12),M$(12)
40 FOR I=1 TO 12:READ M$(I):NEXT I
50 DATA DE JANVIER,DE FEVRIER,DE MARS,D'AVRIL,DE M
AI,DE JUIN
60 DATA DE JUILLET,D'AOUT,DE SEPTEMBRE,D'OCTOBRE,D
E NOVEMBRE,DE DECEMBRE
70 MODE 1:INK 0,0:INK 1,15: BORDER 0
80 INPUT"PREMIERE UTILISATION (O/N)";R$
90 IF R$="0" THEN 300
100 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
110 OPENIN "!VOITURE"
120 INPUT#9,V,A,K
130 FOR I=1 TO 12
140 INPUT#9,E(I),M(I),K(I),C(I)
150 IF M=0 AND K(I)=0 THEN M=I
160 NEXT I
170 CLOSEIN
180 IF M=0 THEN M=12:K(0)=K:GOTO 500
280 REM -----
290 REM DONNEES DU MOIS
300 CLS
310 IF M<>0 THEN 370
320 INPUT"PRIX ANNUEL DE L'ASSURANCE ";A
330 INPUT"PRIX DE LA VIGNETTE ";V
340 INPUT"NOMBRE DE KMS AU 1ER JANVIER ";K
350 M=1
360 PRINT
370 K(0)=K
380 PRINT"ENTREZ LES DONNEES DU MOIS ";M$(M):PRINT
390 INPUT"ESSENCE ";E(M)
400 INPUT"ENTRETIEN/MECANIQUE ";M(M)
410 INPUT"NB DE KILOMETRES AU COMPTEUR ";K(M)
420 IF K(M)<K(M-1) THEN PRINT"KMS DU MOIS PRECEDEN
T: ";K(M-1):GOTO 410
430 INPUT"REBOURSEMENT CREDIT ";C(M)
480 REM -----
490 REM RESULTATS
500 CLS:PRINT TAB(10)"COUT DE VOTRE VOITURE":PRINT
:PRINT
510 PRINT"ASSURANCE: ";A; " VIGNETTE: ";V
520 PRINT:PRINT
530 PRINT"MOIS KM ESS. MECA CRED. TOT. TOT/K"
540 FOR I=1 TO 12
550 PRINT USING"###";I;
560 IF I>M THEN PRINT " 0":GOTO 610
570 T=E(I)+M(I)+C(I)+(A/12)+(V/12):TT=T/(K(I)-K(I-
1))
580 PRINT USING"#####";K(I)-K(I-1),E(I),M(I),C(I)
,T;
590 PRINT USING" ##.##";TT
600 TE=TE+E(I):TK=K(I)-K:TM=TM+M(I):TC=TC+C(I):TA=
TA+T
610 NEXT I
620 PRINT:PRINT"TOTAL/KM ESS/F MECA/F CR/F TOT/F"
:PRINT " ";
630 PRINT USING"#####":TK,TE,TM,TC,TA

```

```
640 PRINT:PRINT"PRIX MOYEN DU KM: ";:PRINT USING"##
.##";TA/TK
650 IF INKEY$<>CHR$(13) THEN 650
680 REM -----
690 REM SAUVEGARDE
700 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R EST PRET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 OPENOUT "!VOITURE"
730 PRINT#9,V,A,K
740 FOR I=1 TO 12
750 PRINT#9,E(I),M(I),K(I),C(I)
760 NEXT I
770 CLOSEOUT
780 CLS:PRINT"AU MOIS PROCHAIN !"
790 END
```

LE PROGRAMME

Lignes 30 à 180 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 300 à 430 : données du mois

Le programme demande le prix de l'essence, de l'entretien, du crédit et le nombre de kilomètres au compteur. Cette dernière donnée est vérifiée avec celle du mois précédent. S'il s'agit de la première utilisation (M=0), les lignes 320 à 340 saisissent le prix de l'assurance et de la vignette, et le nombre de kilomètres au 1^{er} janvier.

Lignes 500 à 650 : résultats

Ces lignes affichent un tableau récapitulatif des dépenses et le prix moyen au kilomètre pour chaque mois et pour l'année entière.

Lignes 700 à 790 : sauvegarde

On sauve sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A	prix de l'assurance
E()	liste des dépenses d'essence
C()	liste des remboursements du crédit
I	indice de boucle FOR/NEXT
K	nombre de kilomètres au 1 ^{er} janvier
K()	liste des kilomètres au compteur
M	mois en cours
M()	liste des dépenses d'entretien et de mécanique
M\$()	liste des noms des mois
R\$	réponse de l'utilisateur
T	total mensuel des dépenses
TA	total annuel des dépenses
TE	total annuel des dépenses d'essence
TC	total annuel des remboursements du crédit
TK	total annuel des kilomètres parcourus
TM	total annuel des dépenses d'entretien et de mécanique
TT	prix moyen du kilomètre du mois
V	prix de la vignette


CALCUL D'IMPOTS

Pour ne pas avoir de mauvaise surprise en recevant votre avis d'imposition, ce programme vous donnera plus de détails sur son calcul.

Comment l'utiliser : vous devez en priorité répondre à quatre questions. Si vous êtes marié (répondre par oui ou non), le nombre de personnes à charge qui ont des revenus, votre nombre de parts et pour finir les charges déductibles (vos achats de valeurs françaises par exemple).


Puis, pour vous, votre conjoint et les personnes à charge vous devez donner le revenu imposable (ne pas déduire les 10 et 20 %) et le taux de la déduction supplémentaire si vous y avez droit (par exemple limeur de cadres de bicyclette du département de la Loire (20 %), brodeur de la région lyonnaise utilisant des métiers pantographes (20 %), journaliste (30 %), speakerine de la télévision française (20 %), scaphandrier (10 %), mannequin d'une grande maison de couture parisienne (10 %), etc.).


Le programme tient compte des mesures fiscales de l'année 84 (minoration, majoration et décote de l'impôt). Vous serez peut-être amené à modifier le programme pour les années à venir. Il faut dans tous les cas modifier les tranches d'imposition situées à la fin du programme, sinon votre impôt risque d'être inférieur à la réalité.



VOUS		
REVENU IMPOSABLE	:	48910
DEDUCTION 10 %	:	-4891
ABATTEMENT 20 %	:	-8803.8
NET IMPOSABLE	:	35215.2
TOTAL NET IMPOSABLE	:	35210
NOMBRE DE PARTS	:	1
QUOTIENT FAMILIAL	:	35210
TRANCHE D'IMPOSITION	:	20 %
DECOTE	:	-844.5
MINORATION 5.0 %	:	-115
IMPOT	:	2195

CALCUL D'IMPOT





```

10 REM CALCUL D'IMPOTS 85 (REVENUS 84)
20 REM -----
30 MODE 1:INK 0,0:INK 1,15:BORDER 0
40 DIM T(14):GOSUB 1000
80 REM -----
90 REM DONNEES DE CALCUL
100 CLS
110 PRINT TAB(10)"CALCUL D'IMPOTS":PRINT:PRINT
120 INPUT"ETES-VOUS MARIE (O/N)";R$
130 IF R$="O" THEN M=2 ELSE M=1
140 INPUT"NOMBRE DE PERSONNES A CHARGE";P
150 IF P>0 THEN M=M+P
160 INPUT"NOMBRE DE PARTS";N
170 IF N<1 THEN 160
180 INPUT"CHARGES DEDUCTIBLES";C
190 FOR I=1 TO M
200 CLS:IF I>2 THEN PRINT D$(3);I-2 ELSE PRINT D$(
I)
210 PRINT
220 INPUT"REVENU IMPOSABLE";R(I)
230 INPUT"DEDUCTION SUPPLEMENTAIRE (O/N)";R$
240 IF R$<>"O" THEN 260
250 INPUT"TAUX";TS(I):TS(I)=TS(I)/100
260 NEXT I
380 REM -----
390 REM CALCUL
400 CLS
410 FOR I=1 TO M
420 IF I>2 THEN PRINT D$(3);I-2 ELSE PRINT D$(I)
430 PRINT"REVENU IMPOSABLE : ";TAB(30);R(I)
440 D1=R(I)*0.1
450 IF D1>S1 THEN D1=S1
460 IF D1<I1 THEN D1=0
470 R(I)=R(I)-D1
480 PRINT"DEDUCTION 10 %      : ";TAB(30);-D1
490 IF TS(I)=0 THEN 540
500 DS=R(I)*TS(I)
510 IF DS>SS THEN DS=SS
520 R(I)=R(I)-DS
530 PRINT"DEDUCTION";TS(I)*100;"%      : ";TAB(30);-DS
540 D2=R(I)*0.2
550 IF D2>S2 THEN D2=S2
560 IF D2<I2 THEN D2=0
570 R(I)=R(I)-D2
580 PRINT"ABATTEMENT 20 %      : ";TAB(30);-D2
590 PRINT"NET IMPOSABLE      : ";TAB(30);R(I):PRINT
600 R=R+R(I)
610 NEXT I
680 REM -----
690 REM RESULTATS
700 R=R-C:R=INT(R/10)*10:QF=R/N
710 FOR I=1 TO 14
720 D=D+T(I-1)*0.05
730 IF QF<=T(I) THEN 760
740 T=T+0.05
750 NEXT I
760 PRINT
770 PRINT"TOTAL NET IMPOSABLE      : ";TAB(30);R
780 PRINT"NOMBRE DE PARTS          : ";TAB(30);N

```

```

790 PRINT"QUOTIENT FAMILIAL      :";TAB(30);INT(QF)
800 PRINT"TRANCHE D'IMPOSITION  :";TAB(30);T*100;"
%"
810 R=(R*T)-(D*N)
820 IF R<4000 AND N=1 THEN DE=4000-R
830 IF R<1600 AND N=1.5 THEN DE=1600-R
840 IF DE>0 THEN PRINT"DECOTE          :";TAB
B(30);-DE:R=R-DE
850 IF R<21521 THEN MI=R*.05:TI=5
860 IF R>21520 AND R<26900 THEN MI=4*(1345-(R*.05
)):TI=(MI/(R/100))
870 IF MI>0 THEN PRINT"MINORATION";USING" #.# %";T
I;:PRINT"      :";TAB(30);-INT(MI)
880 IF R>32280 THEN ME=R*.03:TM=3
890 IF ME>0 THEN PRINT"MAJORATION";TM;"%      : "
;TAB(30);INT(ME)
900 PRINT"IMPOT          :";TAB(30);:IF R+M
E-MI<320 THEN PRINT 0 ELSE PRINT INT(R+ME-MI)
910 IF INKEY$<>CHR$(13) THEN 910
920 RUN
980 REM -----
990 REM TRANCHES
1000 D$(1)="VOUS":D$(2)="VOTRE CONJOINT":D$(3)="PE
RSONNE A CHARGE"
1010 FOR I=1 TO 14
1020 READ T(I)
1030 NEXT I
1040 REM TRANCHES D'IMPOSITION
1050 DATA 14820,15490,18370,29050,37340,46920,5677
0
1060 DATA 65500,109140,150100,177550,201970,228920
,999999
1070 REM MINI-MAXI DES 10,20% ET SUPPL.
1080 I1=1800:S1=54770:S2=99000
1090 SS=50000
1100 RETURN

```

PROGRAMME

Lignes 100 à 260 : données de calcul

Après l'entrée des quatre premières questions, le programme saisit par une boucle FOR/NEXT le revenu et le taux de déduction supplémentaire pour tous les membres de la famille ayant des revenus.

Lignes 400 à 610 : calcul

Pour chaque revenu on déduit les 10 et 20 %, et s'il y a lieu la déduction supplémentaire.

Lignes 700 à 920 : résultats

Après l'addition de tous les revenus et déduction des charges, on recherche la tranche d'imposition dans le tableau T à partir du quotient familial (revenu imposable divisé par le nombre de parts). La réduction d'impôt (variable D) est calculée en additionnant 5 % de toutes les tranches précédentes.

Les lignes 820 à 840 calculent la décote, les lignes 850 à 870 la minoration et les lignes 880 à 890 la majoration. Vous devrez modifier ces lignes les années à venir en fonction des nouvelles mesures fiscales.


Lignes 1000 à 1100 : tranches

Vous devez modifier chaque année les DATAs des lignes 1050 et 1060 avec les nouvelles tranches d'imposition que l'on trouve généralement en page 7 de la notice explicative.

Les variables I1 et S1 correspondent au plancher et au plafond de la déduction de 10 %, S2 à celle des 20 % (pas de plancher) et SS à celle de la déduction supplémentaire.

LISTE DES VARIABLES :

C	montant des charges déductibles
D	déduction de la tranche d'impôt
DE	montant de la décote.
D1	montant de la déduction de 10 %
D2	montant de l'abattement de 20 %
DS	montant de la déduction supplémentaire
DS()	noms des membres de la famille
I	indice de boucle FOR/NEXT
I1	plancher de la déduction de 10 %
M	nombre de personnes ayant des revenus
ME	montant de la majoration exceptionnelle
MI	montant de la minoration exceptionnelle
N	nombre de parts
P	nombre de personnes à charge
QF	quotient familial
R	total des revenus imposables
R()	revenu imposable des membres de la famille
R\$	réponse du contribuable
S1	plafond de l'abattement de 10 %
S2	plafond de l'abattement de 20 %
SS	plafond de la déduction supplémentaire
T	taux de la tranche d'imposition
TI	taux de la minoration exceptionnelle
TM	taux de la majoration exceptionnelle
TS()	taux des déductions supplémentaires



THEME 2 PEDAGOGIE

MUSIQUE

Voilà pour exercer vos talents artistiques un programme de musique.

Après son lancement, deux portées se dessinent sur l'écran ainsi qu'une flèche. Celle-ci se déplace avec les touches droite et gauche et pointe l'endroit où se positionne la note entrée à l'aide des touches de la première rangée du clavier (du DO de la première octave au FA de la seconde. Exemple : 1=DO, 2=RE, 3=MI, etc.).

En dessous des portées plusieurs options vous sont proposées :


- **DURÉE** : en tapant sur B, N ou C vous obtenez des blanches, des noires, ou des croches.
- **ENVELOPPE** : en tapant sur E puis sur une des deux flèches droite ou gauche vous pouvez la faire varier de 1 à 4.
- **TEMPO** : il varie comme l'enveloppe, mais après avoir tapé sur la lettre T (de 1 à 20).


Pour modifier une note il suffit de faire venir la flèche en dessous et d'effectuer le changement.

Pour effacer ou insérer une note, il faut la désigner avec la flèche puis appuyer sur DEL ou COPY.

Quand votre mélodie est entièrement écrite ou pour faire des essais, la touche M la fait **exécuter** par AMSTRAD.

MUSIQUE





```

10 REM MUSIQUE
20 REM -----
30 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1: BORDER 1
40 DIM N(120),D(120),V(11):SYMBOL AFTER 128:GOSUB
1200
50 F$=CHR$(243)+CHR$(242)+"CNBTE"+CHR$(224)+CHR$(1
27)+"M1234567890-"
60 F(1)=148:F(2)=140:F(3)=144
70 LF$=CHR$(10)+CHR$(8)+CHR$(8)
80 FOR I=128 TO 138 STEP 2
90 C$=C$+CHR$(I)+CHR$(I+1)+LF$
100 NEXT I
110 GOSUB 1000
120 I=0:X=3:Y=2
130 D=1:T=10:E=1
140 ENV 2,4,3,1,1,0,10,6,-2,2:ENV 3,15,1,1,1,0,10,
15,-1,1:ENV 4,3,5,1,1,0,2,15,-1,1
280 REM -----
290 REM SAISIE DES NOTES
300 LOCATE X,Y+6:PRINT"^":LOCATE 7,23:PRINT CHR$(F
(D))
310 LOCATE 1,24:PRINT"DUREE:";CHR$(F(D)+1):PRINT"E
NVELOPPE:";E;" TEMPO:";T;
320 A$=INKEY$:IF A$="" THEN 320
330 LOCATE X,Y+6:PRINT" "
340 Z=INSTR(F$,A$):A$="":IF Z=0 THEN 300
350 IF Z>10 THEN Z=Z-10:GOTO 550
360 ON Z GOTO 370,390,410,410,410,420,450,480,510,
540
370 IF I<113 THEN I=I+1:X=X+1:IF X=41 THEN X=3:Y=Y
+7
380 GOTO 300
390 IF I>0 THEN I=I-1:X=X-1:IF X=2 THEN X=40:Y=Y-7
400 GOTO 300
410 D=Z-2:GOTO 300
420 WHILE A$="":A$=INKEY$:WEND:IF A$=CHR$(242) AND
T>1 THEN T=T-1
430 IF A$=CHR$(243) AND T<20 THEN T=T+1
440 GOTO 300
450 WHILE A$="":A$=INKEY$:WEND:IF A$=CHR$(242) AND
E>1 THEN E=E-1
460 IF A$=CHR$(243) AND E<4 THEN E=E+1
470 GOTO 300
480 FOR J=113 TO I STEP -1:N(J+1)=N(J):D(J+1)=D(J)
:NEXT J
490 N(I)=0:D(I)=0:X1=X:Y1=Y:I1=I:GOSUB 1000:X=X1:Y
=Y1:I=I1
500 GOTO 300
510 FOR J=I TO 113:N(J)=N(J+1):D(J)=D(J+1):NEXT J
520 N(113)=0:D(113)=0:X1=X:Y1=Y:I1=I:GOSUB 1000:X=
X1:Y=Y1:I=I1
530 GOTO 300
540 GOSUB 800:GOTO 300
550 N(I)=Z
560 D(I)=D
570 GOSUB 1050:IF X=1 THEN X=3
580 SOUND 1,V(N(I)),D(I)*2*T,12
590 IF I<113 THEN I=I+1 ELSE X=40:Y=16
600 GOTO 300

```

```

780 REM -----
790 REM MUSIQUE
800 FOR J=0 TO I
810 SOUND 1,V(N(J)),D(J)*2*T,12,E
820 NEXT J
830 RETURN
980 REM -----
990 REM DESSIN DE LA PORTEE
1000 X=1:Y=2
1010 FOR I=0 TO 113
1020 GOSUB 1050
1030 NEXT I
1040 RETURN
1050 IF X=1 THEN LOCATE 1,Y:PRINT C$:X=X+2
1060 LOCATE X,Y-1:PRINT " ":LOCATE X,Y+5:PRINT " "
1070 FOR J=Y TO Y+4:LOCATE X,J:PRINT CHR$(152):NEXT
T
1080 IF N(I)=0 THEN 1130
1090 IF N(I)/2=INT(N(I)/2) THEN P=2 ELSE P=0
1100 IF D(I)<2 THEN N=8:GOTO 1120
1110 IF D(I)>2 THEN N=4 ELSE N=0
1120 LOCATE X,Y+4-INT((N(I)-1)/2):PRINT CHR$(140+P
+N):LOCATE X,Y+5-INT((N(I)-1)/2):PRINT CHR$(141+P+
N)
1130 X=X+1:IF X=41 THEN X=1:Y=Y+7
1140 RETURN
1180 REM -----
1190 REM GRAPHIQUES
1200 SYMBOL 128,1,2,4,4,4,255,4,4
1210 SYMBOL 129,128,64,32,32,32,255,32,96
1220 SYMBOL 130,4,4,4,4,5,255,3,6
1230 SYMBOL 131,96,96,192,192,128,255,0,0
1240 SYMBOL 132,14,14,26,50,50,255,99,199
1250 SYMBOL 133,0,0,0,0,0,255,192,224
1260 SYMBOL 134,207,217,209,209,209,255,200,196
1270 SYMBOL 135,240,48,24,24,8,255,136,136
1280 SYMBOL 136,64,32,32,16,15,255,0,0
1290 SYMBOL 137,144,144,160,192,192,255,64,64
1300 SYMBOL 138,0,0,14,31,31,30,15,0
1310 SYMBOL 139,64,64,64,64,64,64,128,0
1320 SYMBOL 140,0,0,0,0,2,255,2,2
1330 SYMBOL 141,2,2,2,62,126,255,126,60
1340 SYMBOL 142,2,2,2,2,2,255,2,62
1350 SYMBOL 143,126,126,126,60,0,255,0,0
1360 SYMBOL 144,0,0,0,0,2,255,2,2
1370 SYMBOL 145,2,2,2,62,66,255,66,60
1380 SYMBOL 146,2,2,2,2,2,255,2,62
1390 SYMBOL 147,66,66,66,60,0,255,0,0
1400 SYMBOL 148,0,0,0,0,6,255,5,5
1410 SYMBOL 149,5,4,4,60,124,255,124,56
1420 SYMBOL 150,6,7,5,5,5,255,4,60
1430 SYMBOL 151,124,124,124,56,0,255,0,0
1440 SYMBOL 152,0,0,0,0,0,255,0,0
1450 FOR I=1 TO 11
1460 READ V(I)
1470 NEXT I
1480 RETURN
1490 DATA 239,213,190,179,159,142,127,119,106,95,8
9

```

LE PROGRAMME

Lignes 30 à 140 : initialisation

Après la définition des caractères graphiques par le sous-programme en 1200, on place dans le tableau F() les pointeurs des caractères représentant les 3 durées et on initialise les quatre enveloppes possibles.

Le sous-programme en 1000 affiche la portée et les clefs.

Lignes 300 à 600 : saisie des notes

La saisie est divisée en 11 parties :

ligne 370	: déplace le curseur à droite
ligne 390	: déplace le curseur à gauche
ligne 410	: modifie la durée de la prochaine note
lignes 420 à 440	: modifient le tempo de la mélodie
lignes 450 à 470	: modifient l'enveloppe de la mélodie
lignes 480 à 500	: insèrent une note sous le curseur
lignes 510 à 530	: suppriment la note sous le curseur
ligne 540	: joue la mélodie (sous-programme en 800)
lignes 550 à 600	: placent une note dans N() et D().

Lignes 800 à 900 : musique

Ces lignes jouent toutes les notes du tableau N() pour les durées contenues dans le tableau D().

Lignes 1000 à 1140 : dessin de la portée

Le sous-programme affiche sur l'écran les portées avec les clefs de sol, ainsi que toutes les notes du tableau N().

Cette partie du programme est aussi utilisée pour supprimer ou ajouter une note (option COPY et DEL).

LISTE DES VARIABLES :

A\$	dernière touche appuyée
C\$	dessin de la clef de sol
D	durée en cours
D()	liste des durées
E	type d'enveloppe en cours
F()	pointeurs des caractères graphiques pour chaque durée
F\$	liste des fonctions
I	position du curseur dans le tableau N()
I1	sauvegarde de I
J	indice de boucle FOR/NEXT
LF\$	1 flèche vers le bas + 2 flèches vers la gauche
N	2 ^e pointeur des graphiques de la note à afficher
N()	liste des notes
P	1 ^{er} pointeur des graphiques de la note à afficher
T	tempo en cours
V()	liste des fréquences de chaque note
X et Y	position du curseur sur l'écran
X1 et Y1	sauvegardes des variables X et Y
Z	numéro de la note en cours


DUREE: 3
ENVELOPPE: 3 TEMPO: 8

GEOGRAPHIE


Comment devenir un as en géographie grâce à votre ordinateur ?

Celui-ci vous dessine d'abord les cartes de la France et de la Corse. Puis il vous demande soit la préfecture d'un département, soit le numéro du département d'une préfecture. A chaque bonne réponse il affiche définitivement la ville, sinon il l'efface et vous la reproposera plus tard.

Vous pouvez compter sur l'infinie patience de votre AMSTRAD. Il s'arrêtera seulement quand toutes les villes seront rétablies sur la carte.



GEOGRAPHIE



```

10 REM GEOGRAPHIE
20 REM -----
30 DIM X(96),Y(96),N$(96),D$(96),G(96)
40 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1:BOARDER 1
50 RANDOMIZE TIME
60 PLOT 212,392
70 FOR I=1 TO 123
80 READ X,Y:DRAW X*2,400-Y*2
90 NEXT
100 PLOT 0,399:DRAWR 380,0:DRAWR 0,-380:DRAWR -380
,0:DRAWR 0,380
110 PLOT 32,100
120 FOR I=1 TO 24
130 READ X,Y:DRAW X*2,400-Y*2
140 NEXT
150 PLOT 0,112:DRAWR 54,0:DRAWR 0,-93:DRAWR -54,0:
DRAWR 0,93
160 FOR I=1 TO 96
170 READ X(I),Y(I),N$(I),D$(I)
180 X(I)=X(I)*2:Y(I)=400-Y(I)*2
190 NEXT I
200 REM -----
290 REM NOM DES VILLES
300 P=INT(RND(1)*96)+1:IF G(P)<>0 THEN 300
310 IF RND(1)>0.5 THEN 370
320 LOCATE 25,10:PRINT CHR$(18);"VILLE:"

330 LOCATE 25,14:PRINT CHR$(18);"DEPARTEMENT:";LOC
ATE 25,15:PRINT CHR$(18);D$(P)
340 GOSUB 600
350 IF R$<>LEFT$(N$(P),LEN(R$)) OR R$="" THEN 300
360 GOTO 420
370 LOCATE 25,10:PRINT CHR$(18);"DEPARTEMENT:"
380 LOCATE 25,14:PRINT CHR$(18);"VILLE:";LOCATE 25
,15:PRINT CHR$(18);N$(P)
390 GOSUB 600
400 IF LEN(R$)=1 THEN R$=""+R$
410 IF R$<>D$(P) THEN 300
420 PLOT X(P),Y(P),1:DRAWR 1,0,1:DRAWR 0,-1,1:DRAW
R -1,0,1:DRAWR 0,1,1
430 G(P)=1:G=G+1
440 LOCATE 4,25:PRINT G;"VILLES TROUVEES";
450 IF G<96 THEN 300
460 LOCATE 2,25:PRINT"VOUS ETES FORT EN GEO ?";
470 IF INKEY$=CHR$(13) THEN RUN ELSE 470
500 REM -----
590 REM SAISIE DES REPONSES
600 R$="":LOCATE 25,11:PRINT CHR$(18);
610 PLOT X(P),Y(P),0:DRAWR 1,0,0:DRAWR 0,-1,0:DRAW
R -1,0,0:DRAWR 0,1,0
620 A$=INKEY$:IF A$="" THEN 680
630 IF A$=CHR$(13) THEN RETURN
640 IF A$=CHR$(127) THEN IF LEN(R$)>0 THEN R$=LEFT
$(R$,LEN(R$)-1):GOTO 670 ELSE 670
650 R$=R$+A$
660 IF LEN(R$)=16 THEN RETURN
670 LOCATE 25,11:PRINT R$;" ";CHR$(8);
680 PLOT X(P),Y(P),1:DRAWR 1,0,1:DRAWR 0,-1,1:DRAW
R -1,0,1:DRAWR 0,1,1
690 GOTO 610

```

980 REM -----
 990 REM FRANCE METROPOLITAINE
 1000 DATA 93,8,91,25,73,33,72,37,74,38
 1010 DATA 68,41,57,39,54,40,52,36,53,34
 1020 DATA 53,32,44,32,45,39,47,41,48,52
 1030 DATA 50,53,50,55,45,54,44,52,38,54
 1040 DATA 37,53,32,56,26,48,19,49,19,52
 1050 DATA 16,50,3,54,2,57,7,57,8,57
 1060 DATA 5,60,8,61,8,62,2,63,6,66
 1070 DATA 7,69,9,70,12,67,25,76,29,74
 1080 DATA 29,77,33,78,33,81,38,84,38,86
 1090 DATA 39,87,38,91,42,99,51,103,52,110
 1100 DATA 50,115,58,125,51,120,44,160,40,165
 1110 DATA 45,166,44,172,48,172,59,177,62,176
 1120 DATA 66,179,79,180,77,177,90,182,93,182
 1130 DATA 95,185,100,188,102,186,105,188,108,188
 1140 DATA 110,186,114,186,116,184,114,181,115,172
 1150 DATA 129,162,131,163,138,164,140,167,143,164
 1160 DATA 158,170,165,169,184,153,185,146,178,147
 1170 DATA 172,142,174,134,168,131,168,127,174,124
 1180 DATA 174,118,170,113,172,109,169,105,169,100
 1190 DATA 162,100,160,105,156,106,159,97,172,79
 1200 DATA 176,79,178,61,186,45,164,41,162,37
 1210 DATA 157,35,154,37,150,35,147,35,138,30
 1220 DATA 138,22,135,24,135,27,129,27,128,19
 1230 DATA 122,20,119,15,115,15,115,12,113,9
 1240 DATA 110,12,107,7,106,4
 1250 REM -----
 1260 REM CORSE
 1270 DATA 16,156,14,155,6,160,5,163,7,165
 1280 DATA 5,165,5,167,8,169,6,173,8,173
 1290 DATA 7,176,10,177,8,180,13,182,13,183
 1300 DATA 15,184,18,177,18,172,20,168,20,159
 1310 DATA 18,156,19,154,18,149,16,150
 1320 REM -----
 1330 REM VILLES
 1340 DATA 127,62,TROYES,10,175,64,COLMAR,68
 1350 DATA 161,63,EPINAL,88,142,66,CHAUMONT,52
 1360 DATA 29,73,VANNES,56,120,73,AUXERRE,89
 1370 DATA 168,73,BELFORT,90,158,75,VESOUL,70
 1380 DATA 88,77,BLOIS,41,61,79,ANGERS,49
 1390 DATA 79,81,TOURS,37,46,83,NANTES,44
 1400 DATA 142,83,DIJON,21,157,83,BESANCON,25
 1410 DATA 103,88,BOURGES,18,115,90,NEVERS,58.
 1420 DATA 93,94,CHATEAUROUX,36,48,95,LA ROCHE S/YO
 N,85
 1430 DATA 150,95,LONS-LE-SAUNIER,39,74,99,POITIERS
 ,86
 1440 DATA 118,99,MOULINS,03,62,103,NIORT,79
 1450 DATA 52,106,LA ROCHELLE,17,139,103,MACON,71
 1460 DATA 145,105,BOURG EN BRESSE,01,159,110,ANNEC
 Y,74
 1470 DATA 112,14,LILLE,59,108,20,ARRAS,62
 1480 DATA 102,30,AMIENS,80,135,30,MEZIERES,08
 1490 DATA 121,36,LAON,02,86,39,ROUEN,76
 1500 DATA 99,39,BEAUVAIS,60,65,43,CAEN,14
 1510 DATA 54,45,ST LO,50,30,56,ST BRIEUC,22
 1520 DATA 10,65,QUIMPER,29,46,66,RENNES,35
 1530 DATA 85,47,EVREUX,27,157,44,METZ,57
 1540 DATA 179,54,STRASBOURG,67,131,48,CHALONS S/MA
 RNE,51

1550 DATA 103, 50, PARIS, 75, 142, 52, BAR LE DUC, 55
1560 DATA 156, 53, NANCY, 54, 70, 60, ALENCON, 61
1570 DATA 90, 60, CHARTRES, 28, 107, 57, MELUN, 77
1580 DATA 58, 66, LAVAL, 53, 73, 68, LE MANS, 72
1590 DATA 97, 70, ORLEANS, 45, 96, 107, GUERET, 23
1600 DATA 96, 113, LIMOGES, 87, 114, 115, CLERMONT-FERRAN
ND, 63
1610 DATA 140, 114, LYON, 69, 134, 122, ST ETIENNE, 42
1620 DATA 157, 118, CHAMBERY, 73, 70, 117, ANGOULEME, 16
1630 DATA 80, 126, PERIGUEUX, 24, 95, 125, TULLE, 19
1640 DATA 104, 133, AURILLAC, 15, 120, 141, MENDE, 48
1650 DATA 159, 140, GAP, 05, 125, 130, LE PUY, 43
1660 DATA 137, 135, PRIVAS, 07, 142, 132, VALENCE, 26
1670 DATA 153, 126, GRENOBLE, 38, 112, 180, PERPIGNAN, 66
1680 DATA 150, 165, MARSEILLE, 13, 91, 174, FOIX, 09
1690 DATA 58, 134, BORDEAUX, 33, 60, 155, MONT-DE-MARSA
N, 40
1700 DATA 62, 166, PAU, 64, 68, 168, TARBES, 65
1710 DATA 103, 169, CARCASSONNE, 11, 158, 168, TOULON, 83
1720 DATA 179, 155, NICE, 06, 126, 160, MONTPELLIER, 34
1730 DATA 77, 159, AUCH, 32, 89, 160, TOULOUSE, 31
1740 DATA 77, 148, AGEN, 47, 89, 151, MONTAUBAN, 82
1750 DATA 100, 154, ALBI, 81, 133, 150, NIMES, 30
1760 DATA 142, 153, AVIGNON, 84, 162, 149, DIGNE, 04
1770 DATA 90, 142, CAHORS, 46, 106, 144, RODEZ, 12
1780 DATA 7, 172, AJACCIO, 2A, 105, 52, CRETEIL, 94
1790 DATA 101, 49, NANTERRE, 92, 104, 48, BOBIGNY, 93
1800 DATA 105, 55, EVRY, 91, 100, 52, VERSAILLES, 78
1810 DATA 100, 45, PONTOISE, 95, 19, 155, BASTIA, 2B

LE PROGRAMME

Lignes 30 à 190 : initialisation

Après le dimensionnement des divers tableaux (voir liste des variables), le programme initialise les nombres aléatoires (ligne 50).

Les lignes 60 à 140 tracent la carte de France et de Corse avec les DATAs des lignes 1000 à 1310.

On place les coordonnées, le nom et le département des villes dans les tableaux X(), Y(), N\$() et D\$() avec les DATAs des lignes 1340 à 1810.

Lignes 300 à 470 : nom des villes

Le nombre aléatoire contenu dans la variable P pointe une des 95 villes du fichier. Si cette ville a déjà été trouvée ($G(P) < > 0$) on en essaye une autre (ligne 300).

Le nombre aléatoire de la ligne 310 détermine si l'on doit demander le nom ou le numéro du département.

La saisie de la réponse s'effectue à l'aide du sous-programme des lignes 600 à 690.

Si la réponse est juste, on place le chiffre 1 dans le tableau G() pour ne pas demander plusieurs fois la même ville.

Lignes 600 à 690 : saisie des réponses

On saisit caractère par caractère la réponse du joueur, ceci pour éviter l'instruction INPUT qui pourrait détruire le dessin si l'on tape trop de lettres ou de chiffres.

On sort du sous-programme quand le joueur a appuyé sur la touche ENTER ou si la réponse contient au moins 16 caractères.

LISTE DES VARIABLES :


A\$	dernière touche appuyée
D\$()	liste des départements
G	nombre de villes déjà trouvées
G()	liste des villes déjà trouvées
I	indice de boucle FOR/NEXT
N\$()	liste des villes
P	pointe la prochaine ville à trouver
R\$	réponse du joueur
X et Y	coordonnées d'un point à afficher
X() et Y()	coordonnées des villes sur l'écran

APPRENDRE A TAPER AU CLAVIER


Si vous désirez apprendre à taper ou améliorer votre vitesse de frappe au clavier, le programme qui suit devrait vous être très utile.

Après le dessin du clavier, AMSTRAD affiche une lettre au hasard sur la touche correspondante. Vous devez bien sûr taper immédiatement cette même lettre pendant le temps très court qui vous est imparti. Chaque réponse correcte vous donne un point. Si vous obtenez plus de 17/20, le niveau de rapidité augmente. Pour les niveaux supérieurs à 5 il n'est plus possible de regarder le clavier, vous devez donc le connaître par cœur !

Paradoxalement, vous devrez taper le programme avant de pouvoir vous exercer !



APPRENDRE A TAPER AU CLAVIER



```

10 REM APPRENDRE A TAPER AU CLAVIER
20 REM -----
30 DIM C$(48),X(48),Y(48)
40 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1: BORDER 1
50 DATA 1,2,3,4,5,6,7,8,9,0,-,^
60 DATA Q,W,E,R,T,Y,U,I,O,P,@,[
70 DATA A,S,D,F,G,H,J,K,L,"",;,]
80 DATA Z,X,C,V,B,N,M,"",.-,/, \
90 DATA " "
100 Y=3:X=2
110 FOR I=1 TO 48
120 READ C$(I)
130 X(I)=X:Y(I)=Y
140 X=X+3
150 IF X>37 THEN Y=Y+3:X=INT(Y/3)+1
160 NEXT I
170 RANDOMIZE TIME
280 REM -----
290 REM DESSIN DU CLAVIER
300 CLS
310 FOR I=1 TO 47
320 MOVE X(I)*16-26,425-Y(I)*16: DRAWR 35,0: DRAWR 0
,-35: DRAWR -35,0: DRAWR 0,35
330 NEXT I
340 MOVER -13,-48: DRAWR -419,0: DRAWR 0,-35: DRAWR 4
19,0: DRAWR 0,35
350 X(48)=20:Y(48)=15
380 REM -----
390 REM APPRENTISSAGE
400 N=0
410 P=0
420 LOCATE 1,20: PRINT "NIVEAU: ";N; "          MOYE
NNE: "
430 FOR K=1 TO 20
440 C=INT(RND(1)*48)+1
450 IF C=48 THEN LOCATE X(C),Y(C): PRINT CHR$(231):
GOTO 470
460 LOCATE X(C),Y(C): PRINT C$(C)
470 FOR I=1 TO 380-N*20
480 A$=INKEY$: IF A$<>" " THEN 500
490 NEXT I
500 IF A$=C$(C) THEN P=P+1
510 LOCATE 30,20: PRINT P; "/20 "
520 FOR I=1 TO 380-N*20: Z$=INKEY$: NEXT I
530 LOCATE X(C),Y(C): PRINT " "
540 NEXT K
550 IF P>17 THEN N=N+1: FOR I=500 TO 50 STEP -5: SOU
ND 1,I,2: NEXT ELSE FOR I=50 TO 500 STEP 5: SOUND 1,
I,2: NEXT
560 IF N>9 THEN N=9
570 GOTO 410

```

LE PROGRAMME

Lignes 30 à 170 : lecture des DATAs

On place dans le tableau C\$() les lettres qui composent le clavier et dans les tableaux X() et Y() leurs coordonnées sur l'écran.

La ligne 170 initialise les nombres aléatoires.

Lignes 300 à 350 : dessin du clavier

Chaque touche du clavier est représentée par un carré sur l'écran. La touche d'espace étant de forme différente, c'est la ligne 340 qui l'affiche.

Lignes 400 à 570 : apprentissage

On affiche vingt fois de suite une lettre au hasard aux coordonnées pointées par les tableaux X et Y.

Pour la barre d'espacement, on affiche le point contenu dans le caractère graphique 231. La ligne 480 saisit la réponse du joueur pendant un temps précis défini par le niveau N. Si on a plus de 17 réponses exactes, on augmente automatiquement le niveau (ligne 550).

LISTE DES VARIABLES :

A\$	réponse tapée au clavier
C	numéro de la lettre choisie au hasard
C\$()	tableau des lettres du clavier
I	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
N	niveau de rapidité
P	nombre de points
X et Y	coordonnées temporaires
X() et Y()	coordonnées des lettres sur l'écran


ALPHABET

Si, depuis que vous avez acheté votre AMSTRAD, vos charmants enfants vous le réclament par curiosité, proposez-leur ce programme qui les occupera pendant que vous lirez la suite de ce livre.


L'alphabet entier se dessine sur l'écran. Il y a une lettre à trouver et elle est désignée par un rectangle noir.

Si le caractère frappé par l'enfant en réponse est le bon, il entend une petite musique bien connue. Dans le cas contraire, c'est une gamme descendante qui se fait entendre.

Il n'y a qu'un seul essai à chaque coup. Par contre, l'ordinateur demandera inlassablement une nouvelle lettre...



ALPHABET



```

10 REM ALPHABET
20 REM -----
30 SYMBOL AFTER 128
40 SYMBOL 128,0,30,18,255,128,128,255,60
50 SYMBOL 129,63,33,33,255,1,1,255,60
60 SYMBOL 130,63,32,47,41,47,32,255,30
70 SYMBOL 131,255,1,61,37,61,1,255,30
80 RANDOMIZE TIME
180 REM -----
190 REM AFFICHE LE JEU
200 MODE 0:INK 0,1:INK 1,24:INK 2,6:PAPER 0:PEN 1:
BORDER 1
210 LOCATE 5,6
220 L=INT(RND(1)*26)+1
230 FOR I=1 TO 26
240 IF I=L THEN PRINT CHR$(143);:GOTO 260
250 PRINT CHR$(64+I);
260 IF I=13 THEN LOCATE 5,10
270 NEXT I
280 LOCATE 1,22:PRINT"QUELLE EST LA LETTREQUI MANQ
UE ?"
290 MOVE 0,109:DRAW 640,109,2
300 P=5:D=15:PEN 2
310 FOR I=5 TO 14 STEP 2
320 LOCATE I,18:PRINT CHR$(130);CHR$(131)
330 NEXT I
380 REM -----
390 REM LE JEU
400 FOR I=1 TO 10
410 AS=INKEY$:IF AS<>" " THEN 490
420 NEXT I
430 LOCATE P,18:PRINT CHR$(130);CHR$(131)
440 P=P-2:IF P<1 THEN P=18
450 D=D-2:IF D<1 THEN D=18
460 LOCATE P,18:PRINT CHR$(128);CHR$(129)
470 LOCATE D,18:PRINT" "
480 GOTO 400
490 IF AS<"A" OR AS>"Z" THEN 410
500 LOCATE 14,23:PRINT AS
510 IF AS<>CHR$(64+L) THEN FOR I=50 TO 500 STEP 4:
SOUND 1,I,2:NEXT:GOTO 540
520 RESTORE 530:FOR I=1 TO 11:READ N:SOUND 1,N,25:
SOUND 1,0,5,0:NEXT I
530 DATA 478,478,478,426,379,426,478,379,426,426,4
78
540 FOR I=1 TO 1000:NEXT I
550 GOTO 200

```

LE PROGRAMME

Lignes 30 à 80 : initialisation

Initialisation des caractères graphiques et des nombres aléatoires d'AMSTRAD.

Lignes 200 à 330 : affichent le jeu

Ces lignes affichent l'alphabet en gros caractères (MODE 0) et un carré noir sur la lettre à trouver. Elles affichent aussi un petit train en bas de l'écran.

Lignes 400 à 550 : le jeu

Les lignes 400 à 420 testent dix fois de suite si une touche a été enfoncée. Si c'est le cas, on va en ligne 490 pour la vérifier. Dans le cas contraire on déplace le train d'une case.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
D	dernière position du train
I	indice de boucle FOR/NEXT
L	lettre à trouver
P	position de départ du train


APPRENDRE A COMPTER

Une idée originale pour apprendre à compter !


Un nombre différent de pommes, cow-boys, téléphones, etc., se dessine à l'écran. Votre bambin doit tout simplement les compter et donner au professeur (le plus patient des professeurs !) le nombre qu'il pense avoir trouvé.

AMSTRAD ne donne pas la réponse, mais indique quand même si elle est exacte.

On recommence ?



APPRENDRE A COMPTER



```

10 REM APPRENDRE A COMPTER
20 REM -----
30 MODE 0
40 DIM N$(21):SYMBOL AFTER 166:GOSUB 300
50 INK 0,1:INK 1,24:PAPER 0:PEN 1:BORDER 1
60 AS=CHR$(10)+STRING$(2,8)
70 RANDOMIZE TIME
80 I=INT(RND(1)*22)
90 CLS
100 N=INT(RND(1)*9)+1
110 FOR J=1 TO N
120 LOCATE X(J),Y(J)
130 PRINT CHR$(166+I*4);CHR$(168+I*4);
140 PRINT AS;
150 PRINT CHR$(167+I*4);CHR$(169+I*4)
160 NEXT J
170 LOCATE 6,20:PRINT"NOMBRE ";
180 IF LEFT$(N$(I),1)="A" OR LEFT$(N$(I),1)="H" TH
EN PRINT"D" ELSE PRINT"DE"
190 PRINT TAB(10-LEN(N$(I))/2);N$(I);" ?"
200 RS=INKEY$:IF RS="" THEN 200
210 R=VAL(RS):IF R=0 THEN 200
220 LOCATE 6,20:PRINT"
"
230 LOCATE 8,20:IF R=N THEN PRINT"BRAVO!" ELSE PRI
NT"NON!";N
240 PRINT STRING$(19,32)
250 IF R=N THEN FOR I%=500 TO 50 STEP -5:SOUND 1,I
%,2:NEXT ELSE FOR I%=50 TO 500 STEP 4:SOUND 1,I%,2
:NEXT
260 GOTO 80
280 REM -----
290 REM GRAPHIQUES
300 SYMBOL 166,7,5,7,1,15,31,51,99
310 SYMBOL 167,111,7,7,3,3,3,15
320 SYMBOL 168,240,176,240,192,248,252,230,227
330 SYMBOL 169,227,224,96,96,96,96,120
340 SYMBOL 170,0,0,7,4,4,31,16,19
350 SYMBOL 171,19,16,16,19,18,18,16,31
360 SYMBOL 172,0,192,224,32,34,250,10,202
370 SYMBOL 173,202,10,10,206,72,72,8,248
380 SYMBOL 174,0,3,13,16,32,32,64,97
390 SYMBOL 175,65,33,32,16,13,3,0,0
400 SYMBOL 176,0,128,96,16,56,72,132,12
410 SYMBOL 177,4,8,8,16,96,128,0,0
420 SYMBOL 178,0,3,12,16,33,35,69,65
430 SYMBOL 179,65,65,33,33,16,12,3,0
440 SYMBOL 180,0,192,48,8,4,116,66,98
450 SYMBOL 181,66,66,4,4,8,48,192,0
460 SYMBOL 182,0,63,196,159,228,9,18,36
470 SYMBOL 183,40,40,36,34,33,32,63,24
480 SYMBOL 184,0,252,35,249,39,144,72,36
490 SYMBOL 185,20,20,100,100,148,4,252,24
500 SYMBOL 186,127,64,84,92,72,64,64,66
510 SYMBOL 187,67,65,64,84,92,72,64,127
520 SYMBOL 188,252,4,84,116,36,4,4,132
530 SYMBOL 189,132,4,4,84,116,36,4,252
540 SYMBOL 190,255,68,68,255,17,17,255,68
550 SYMBOL 191,68,255,17,17,255,68,68,255
560 SYMBOL 192,255,68,68,255,17,17,255,68
570 SYMBOL 193,68,255,17,17,255,68,68,255
580 SYMBOL 194,63,32,36,32,32,36,32,63
590 SYMBOL 195,32,32,32,32,36,32,63,0
600 SYMBOL 196,252,4,36,4,4,36,4,252
610 SYMBOL 197,4,36,4,4,4,4,252,0
620 SYMBOL 198,3,6,10,18,34,74,130,146

```

```

630 SYMBOL 199,13,164,137,133,164,192,255,0
640 SYMBOL 200,255,1,73,1,73,1,73,1
650 SYMBOL 201,255,2,36,8,144,32,192,0
660 SYMBOL 202,1,64,200,48,25,141,95,35
670 SYMBOL 203,83,139,11,3,3,3,3,7
680 SYMBOL 204,32,162,68,136,52,227,128,241
690 SYMBOL 205,202,164,146,168,128,128,128,192
700 SYMBOL 206,0,0,12,30,63,127,127,63
710 SYMBOL 207,30,12,0,0,0,0,0,0
720 SYMBOL 208,7,26,100,132,8,136,144,16
730 SYMBOL 209,24,60,126,255,255,126,60,24
740 SYMBOL 210,0,0,3,1,1,1,1,1
750 SYMBOL 211,7,3,7,31,11,23,44,120
760 SYMBOL 212,128,128,224,64,192,64,192,64
770 SYMBOL 213,240,96,240,252,104,244,26,15
780 SYMBOL 214,63,96,192,135,140,152,248,152
790 SYMBOL 215,24,48,96,192,255,1,3,3
800 SYMBOL 216,252,6,3,225,49,25,31,25
810 SYMBOL 217,24,12,6,3,255,128,192,192
820 SYMBOL 218,0,1,19,31,255,127,63,31
830 SYMBOL 219,31,31,31,63,63,127,31,0
840 SYMBOL 220,192,224,252,255,254,254,252,252
850 SYMBOL 221,252,252,254,254,255,252,128,0
860 SYMBOL 222,0,0,0,0,0,7,31,127
870 SYMBOL 223,127,255,255,255,255,127,62
880 SYMBOL 224,0,7,24,44,206,199,131,129
890 SYMBOL 225,128,192,192,192,128,128,0,0
900 SYMBOL 226,1,7,15,63,255,80,76,69
910 SYMBOL 227,65,65,73,68,35,17,8,7
920 SYMBOL 228,0,192,224,248,254,20,102,70
930 SYMBOL 229,6,6,38,70,138,21,39,197
940 SYMBOL 230,0,0,1,3,7,15,15,31
950 SYMBOL 231,63,63,127,255,3,3,3,0
960 SYMBOL 232,0,0,0,128,192,224,224,240
970 SYMBOL 233,248,248,252,254,128,128,128,0
980 SYMBOL 234,3,3,1,7,15,11,19,11
990 SYMBOL 235,6,3,2,2,3,2,21,10
1000 SYMBOL 236,128,128,0,192,224,160,192,160
1010 SYMBOL 237,148,168,212,160,192,128,0,0
1020 SYMBOL 238,7,31,7,7,3,31,63,103
1030 SYMBOL 239,127,55,7,6,6,6,6,7
1040 SYMBOL 240,192,240,64,192,128,240,248,216
1050 SYMBOL 241,223,204,192,192,192,192,249
1060 SYMBOL 242,63,127,240,127,63,3,3,3
1070 SYMBOL 243,3,3,3,31,3,15,3,31
1080 SYMBOL 244,252,254,15,254,252,192,192,192
1090 SYMBOL 245,192,192,192,192,192,192,192,192
1100 SYMBOL 246,0,0,0,15,63,115,247,231
1110 SYMBOL 247,255,255,255,127,31,7,0,0
1120 SYMBOL 248,8,30,55,34,240,252,254,255
1130 SYMBOL 249,255,255,255,255,254,248,0,0
1140 SYMBOL 250,12,14,27,53,106,213,170,213
1150 SYMBOL 251,107,53,26,14,4,0,0,0
1160 SYMBOL 252,0,0,0,128,192,96,240,152
1170 SYMBOL 253,12,6,3,1,0,0,0,0
1180 FOR I=0 TO 21:READ N$(I):NEXT I
1190 FOR I=1 TO 9:READ X(I),Y(I):NEXT I
1200 RETURN
1210 DATA ROBOTS,MACHINES,HORLOGES,PIECES,TELEPHON
ES,CARTES,MURS
1220 DATA DOMINOS,DES,ARBRES,CERISES,TOURS EIFFEL,
CLOCHES,CARTES DE FRANCE
1230 DATA POIRES,CHINOIS,SAPINS,SKIEURS,COW-BOYS
1240 DATA CLEFS,POMMES,DRAPEAUX
1250 DATA 8,8,12,8,10,3,6,3,4,8,14,3,10,13,6,13,14
,13

```

LE PROGRAMME

Lignes 30 à 100 : initialisation

L'appel du sous-programme en 300 place dans le tableau N\$() les noms des objets qui peuvent apparaître à l'écran, définit les caractères graphiques de ces objets et leurs coordonnées sur l'écran (tableaux X et Y).

La variable A\$ contient un LINEFEED (saut de ligne) suivi de 2 BACKSPACE (recul arrière d'un caractère) ce qui permet d'afficher 4 caractères graphiques à un point donné de l'écran avec une seule instruction LOCATE.

On place dans les variables N et I des nombres aléatoires pour définir respectivement le nombre et le type de caractères graphiques à afficher.

Lignes 110 à 260 : le jeu

Après l'affichage des N dessins, on saisit la réponse du joueur dans la variable R\$. Si la réponse est juste, on affiche le message "BRAVO", dans le cas contraire le message "NON !".

LISTE DES VARIABLES :

A\$	chaîne de caractères : 1 LF + 2 BS
I	numéro de l'objet à afficher
J	indice de boucle FOR/NEXT
N	nombre d'objets à afficher
N\$()	noms des objets
R	valeur décimale de R\$
R\$	réponse du joueur
X() et Y()	coordonnées des objets sur l'écran

TABLES DE MULTIPLICATION

Attention : programme réservé aux moins de dix ans !

Vos enfants vont enfin pouvoir réviser tranquillement leurs tables de multiplication.

L'ordinateur demande juste la table que vous voulez obtenir (jusqu'à 22).

Aucun doute, AMSTRAD ne se trompe jamais !





TABLE DE 12

1	x12	=	12
2	x12	=	24
3	x12	=	36
4	x12	=	48
5	x12	=	60
6	x12	=	72
7	x12	=	84
8	x12	=	96
9	x12	=	108
10	x12	=	120
11	x12	=	132
12	x12	=	144

TABLES DE MULTIPLICATION



```
10 REM TABLES DE MULTIPLICATION
20 REM -----
30 MODE 0:INK 0,1:INK 1,24:BORDER 1
40 LOCATE 3,10:PRINT"QUELLE TABLE ";
50 INPUT T$
60 T=VAL(T$)
70 IF T<1 OR T>22 THEN 30
180 REM -----
190 REM TABLE
200 CLS
210 FOR I=1 TO 12
220 PRINT
230 PRINT TAB(8);
240 PRINT USING"## x";I;
250 PRINT USING"## =";T;
260 PRINT USING"###";I*T
270 NEXT I
280 MOVE 200,0:DRAWR 400,0:DRAWR 0,399:DRAWR -400,
0:DRAWR 0,-399
290 LOCATE 1,13:PRINT "TABLE"
300 PRINT"DE ";
310 PRINT USING"##";T
320 MOVE 0,220:DRAWR 150,0
330 DRAWR 50,-25:DRAWR -50,-26
340 DRAWR -150,0:DRAWR 0,51
350 IF INKEY$<>CHR$(13) THEN 350
360 RUN
```

LE PROGRAMME

Lignes 30 à 70 : saisie

Ces lignes saisissent la table de votre choix. La ligne 70 vérifie que votre réponse est comprise entre 1 et 22.

Lignes 200 à 360 : édition de la table

On affiche avec une boucle FOR/NEXT les 12 premières multiplications de la table choisie.

Les lignes 290 à 340 dessinent le cadre et le numéro de la table.

LISTE DES VARIABLES :


I	indice de boucle FOR/NEXT
T	numéro de la table
T\$	réponse du joueur

LES QUATRE OPERATIONS


Vous n'êtes pas très fort en calcul mental ? AMSTRAD vous sera d'un grand secours.

Il vous propose une des quatre opérations composées de nombres tirés au hasard. Vous devez entrer votre solution comme si vous l'écriviez sur une feuille de papier (de gauche à droite, les chiffres se décalent automatiquement sur l'écran).

Si vous avez du mal à résoudre ces problèmes, reportez-vous au programme précédent !



LES QUATRE OPERATIONS



```

10 REM LES QUATRE OPERATIONS
20 REM -----
30 O$(1)="ADDITION":O$(2)="SOUSTRACTION"
40 O$(3)="MULTIPLICATION":O$(4)="DIVISION"
50 S$(1)="+":S$(2)="-"
60 S$(3)="x"
70 MODE 0
80 RANDOMIZE TIME
90 CLS
100 O=INT(RND(1)*4)+1
110 A=INT(RND(1)*99)+1
120 B=INT(RND(1)*99)+1
130 IF O>2 AND B>9 THEN 120
140 IF O=4 AND A/B<>INT(A/B) THEN 110
150 IF O=2 AND A<B THEN 110
160 ON O GOTO 170,180,190,200
170 R=A+B:GOTO 300
180 R=A-B:GOTO 300
190 R=A*B:GOTO 300
200 R=A/B:GOTO 400
280 REM -----
290 REM OPERATIONS + - *
300 PRINT TAB(10-LEN(O$(0))/2);O$(0)
310 LOCATE 11,9:PRINT USING"##";A
320 LOCATE 7,12:PRINT S$(0);" ";
330 PRINT USING"##";B
340 LOCATE 7,15:PRINT STRING$(6,154)
350 LOCATE 7,18:PRINT"="
360 GOTO 500
380 REM -----
390 REM DIVISION
400 PRINT TAB(6);O$(0)
410 LOCATE 6,12:PRINT USING"## ";A,B
420 FOR I=10 TO 20
430 LOCATE 9,I:PRINT CHR$(149)
440 NEXT I
450 LOCATE 9,15:PRINT CHR$(151);STRING$(3,154)
480 REM -----
490 REM SAISIE REPONSE
500 R$=""
510 A$=INKEY$:IF A$="" THEN 510
520 IF A$<"0" OR A$>"9" THEN 510
530 R$=R$+A$
540 IF LEN(R$)>LEN(STR$(R))-1 THEN 700
550 LOCATE 10,18:PRINT USING"###";VAL(R$)
560 IF VAL(R$)=R THEN 710
570 IF VAL(R$)>R THEN 700
580 GOTO 510
680 REM -----
690 REM GAGNE ET PERDU
700 FOR I=1 TO 1000 STEP 3:SOUND 1,I,1:NEXT:GOTO 7
20
710 FOR I=800 TO 1 STEP -3:SOUND 1,I,1:NEXT
720 LOCATE 10,18:PRINT USING"###";R
730 IF INKEY$<>CHR$(13) THEN 730
740 GOTO 90

```

LE PROGRAMME

Lignes 30 à 450 : initialisation et opérations

On place dans le tableau O\$() le nom des quatre opérations et dans le tableau S\$() leurs signes.

La variable O définissant le type d'opération à effectuer (nombre aléatoire) et A et B les opérandes, on les affiche après le calcul du résultat R (lignes 170 à 200).

Lignes 500 à 740 : saisie réponse et résultats

La ligne 510 saisit votre réponse et la ligne 550 l'affiche par un PRINT USING.

Si votre réponse est juste, la ligne 700 envoie une gamme montante, sinon la ligne 710 envoie une gamme descendante.

LISTE DES VARIABLES :

A	première opérande
A\$	dernière touche appuyée
B	deuxième opérande
I	indice de boucle FOR/NEXT
O	opération choisie
O\$()	noms des opérations
R	résultat de l'opération
R\$	réponse du joueur
S\$()	signes des opérations

QUESTIONS/REPONSES

Le jeu des questions-réponses est composé de deux programmes. Le premier permet de constituer un fichier de mots liés par une question. L'exemple ci-dessous est celui des réalisateurs de cinéma. Il demande en priorité le nom du fichier. Pour notre exemple nous avons répondu "CINEMA" (seize lettres maximum).

Si vous l'utilisez pour la première fois, AMSTRAD demande la question (toujours pour notre exemple nous avons répondu "Quel est le réalisateur du film").

Puis le programme propose un menu de trois options :


- 1 – **SAISIE** : vous devez donner le mot de la question suivi de la réponse juste, par exemple "Pickpocket" et "Bresson". Quand vous utiliserez le deuxième programme, il demandera alors : "Quel est le réalisateur de Pickpocket", et la réponse devra être "Bresson".
- 2 – **LISTING** : liste sur l'écran la totalité du fichier. Vous pouvez arrêter cette liste en appuyant sur n'importe quelle touche (idem pour continuer).
- 3 – **SAUVEGARDE** : sauve le fichier sur cassette.

Le deuxième programme est le jeu proprement dit. Il demande comme le premier le nom du fichier. Puis il propose au hasard une des X questions du fichier.

Quand vous désirez arrêter, tapez le mot "FIN" comme réponse, les résultats s'affichent automatiquement.

Ces deux programmes permettent de créer une multitude de fichiers dans des domaines très variés : littérature, peinture, histoire, géographie, musique, sport, animaux, plantes, langues étrangères, informatique, etc., et bien d'autres que vous ne manquerez pas de découvrir.

CREATION QUESTIONS/REPONSES



```

10 REM CREATION FICHER QUESTIONS/REPONSES
20 REM -----
30 DIM Q$(1000,2)
40 E$=STRING$(39,32)+CHR$(13)
50 MODE 1:INK 0,0:INK 1,15:PAPER 0:PEN 1:BORDER 0
60 INPUT"NOM DU FICHER ";N$
70 IF LEN(N$)>15 THEN 60
80 INPUT"PREMIERE UTILISATION (O/N)";R$
90 IF R$="0" THEN 170
100 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
110 OPENIN "!" +N$
120 INPUT#9,Q$
130 IF EOF THEN CLOSEIN:GOTO 200
140 N=N+1
150 INPUT#9,Q$(N,1),Q$(N,2)
160 GOTO 130
170 INPUT"LA QUESTION ";Q$
180 REM -----
190 REM MENU
200 CLS:PRINT TAB(18);"MENU":PRINT
210 PRINT TAB(13);"1 SAISIE 1":PRINT TAB(13);"
2 LISTING 2":PRINT TAB(13);"3 SAUVEGARDE 3"
220 PRINT:PRINT TAB(14);"VOTRE CHOIX ?";
230 R$=INKEY$:IF R$="" THEN 230
240 R=VAL(R$):IF R<1 OR R>3 THEN 230
250 ON R GOTO 300,400,500
280 REM -----
290 REM SAISIE
300 CLS:PRINT TAB(10);"SAISIE D'UNE QUESTION":PRIN
T
310 PRINT:PRINT:PRINT"QUESTION: ";Q$
320 LOCATE 10,10:PRINT E$;"MOT PROPOSE : ";:LINE I
NPUT A$
330 IF A$="" THEN 200
340 N=N+1:Q$(N,1)=A$
350 LOCATE 10,12:PRINT E$;"REPONSE : ";:LINE I
NPUT Q$(N,2)
360 GOTO 300
380 REM -----
390 REM LISTING
400 CLS
410 FOR I=1 TO N
420 PRINT Q$(I,1);" --> ";Q$(I,2)
430 IF INKEY$<>" " THEN WHILE INKEY$="" :WEND
440 NEXT I
450 IF INKEY$="" THEN 450
460 GOTO 200
480 REM -----
490 REM SAUVEGARDE
500 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R EST PRET POUR L'ENREGISTREMENT"
510 LINE INPUT R$
520 OPENOUT "!" +N$
530 PRINT#9,Q$
540 FOR I=1 TO N
550 PRINT#9,Q$(I,1);", ";Q$(I,2)
560 NEXT I
570 CLOSEOUT
580 CLS:PRINT"AU REVOIR !"
590 END

```

PROGRAMME "CREATION QUESTIONS/REPONSES"

Lignes 30 à 170 : initialisation

Le programme demande en priorité le nom du fichier puisque l'on peut créer plusieurs jeux "Questions/réponses".

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement, sinon l'utilisateur doit donner la phrase qui correspond à la question.

Lignes 200 à 250 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 300 à 360 : saisie

Le programme saisit le mot proposé puis la réponse. Si le mot proposé est une chaîne vide (A\$=""), on retourne au menu.

Lignes 400 à 460 : listing

On affiche à l'écran par une boucle FOR/NEXT l'intégralité du fichier. La ligne 430 teste le clavier pour permettre à l'utilisateur d'arrêter l'édition à son gré.


Lignes 500 à 590 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A\$	dernier mot proposé
ES	caractères d'effacement d'une ligne
I	indice de boucle FOR/NEXT
N	nombre de questions dans le fichier
N\$	nom du fichier
Q\$	la question
Q\$(,)	liste des mots proposés et des réponses
R	valeur numérique de R\$
RS	choix de l'utilisateur

JEU QUESTIONS/REPONSES



```

10 REM JEU QUESTIONS/REPONSES
20 REM -----
30 DIM Q$(1000,2)
40 ES=STRING$(39,32)+CHR$(13)
50 MODE 1:INK 0,0:INK 1,15:PAPER 0:PEN 1:BORDER 0
60 INPUT"NOM DU FICHER ";N$
70 IF LEN(N$)>15 THEN 60
80 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
90 OPENIN "!" +N$
100 INPUT#9,Q$
110 IF EOF THEN CLOSEIN:GOTO 200
120 N=N+1
130 INPUT#9,Q$(N,1),Q$(N,2)
140 GOTO 110
180 REM -----
190 REM QUESTIONS
200 CLS
210 Q=INT(RND(1)*N)+1
220 LOCATE 1,10:PRINT Q$;" ";Q$(Q,1);" ?"
230 LOCATE 1,14:LINE INPUT"REPONSE :";R$
240 IF R$="FIN" THEN 400
250 P=P+1
260 IF R$=Q$(Q,2) THEN 290
270 LOCATE 1,14:PRINT ES;"REPONSE :";Q$(Q,2)
280 GOTO 310
290 LOCATE 1,18:PRINT"REPONSE JUSTE"
300 B=B+1
310 IF INKEY$="" THEN 310
320 GOTO 200
380 REM -----
390 REM FIN
400 CLS
410 LOCATE 8,10:PRINT B;"REPONSES JUSTE SUR";P
420 PRINT:PRINT:PRINT:END

```

```

PICKPOCKET --> BRESSON
UN TAXI MAUVE --> BOISSET
CARTOUCHE --> BROCA
HOTEL DU NORD --> CARNE
MOURIR D'AIMER --> CAYATTE
LE CHEVAL D'ORGUEIL --> CHABROL
JEUX INTERDIIS --> CLEMENT
LE SALAIRE DE LA PEUR --> CLOUZOT
LA BELLE ET LA BETE --> COCTEAU
A BOUT DE SOUFLE --> GODARD
UN HOMME ET UNE FEMME --> LELOUCH
ZAZIE DANS LE METRO --> MALLE
LE CERCLE ROUGE --> MELVILLE
LA NUIT AMERICAINE --> TRUFFAUT
LE CLAN DES SICILIENS --> VERNEUIL
ANNIE HALL --> ALLEN
LA RUEE VERS L'OR --> CHAPLIN
APOCALYPSE NOW --> COPPOLA
LA MORT AUX TROUSSES --> HITCHCOCK
LA CHARGE FANTASTIQUE --> WALSH
KAGEMUSHA --> KUROSAWA

```

LE PROGRAMME "JEU QUESTIONS/REPONSES"

Lignes 30 à 140 : lecture du fichier

Après la saisie du nom, le programme lit le fichier sur la cassette.

Lignes 200 à 320 : question

Le programme propose la question pointée par la variable Q (nombre aléatoire).

La ligne 230 saisit la réponse du joueur. S'il s'agit du mot "FIN", le programme saute aux lignes 400 pour la fin du jeu.

En cas de réponse juste, la ligne 300 incrémente le nombre de points B.

Lignes 400 à 420 : fin

Ces lignes affichent le nombre de questions posées et le nombre de réponses justes.

LISTE DES VARIABLES :

B	nombre de réponses justes
E\$	caractères d'effacement d'une ligne
I	indice de boucle FOR/NEXT
N	nombre de questions dans le fichier
N\$	nom du fichier
P	nombre de questions posées
Q	pointe la prochaine question
Q\$	la question
Q\$ (,)	liste des mots proposés et des réponses
R\$	réponse du joueur

CONJUGAISONS

Vos enfants ont-ils du mal avec leurs conjugaisons ?


Pour les aider, voilà un programme qui leur permettra d'obtenir le listing des conjugaisons des verbes réguliers des deux premiers groupes pour les quatre premiers temps.


Pour cela, il suffit de donner à votre AMSTRAD le verbe désiré à l'infinitif pour obtenir sa conjugaison au présent, à l'imparfait, au passé simple et au futur simple.

Si par erreur un verbe d'un autre groupe est demandé, l'ordinateur répond "verbe inconnu".

VERBE MANGER	
PRESENT	PASSE SIMPLE
JE MANGE	JE MANGEAI
TU MANGES	TU MANGEAS
IL MANGE	IL MANGEA
ELLE MANGE	ELLE MANGEA
NOUS MANGEONS	NOUS MANGEAMES
VOUS MANGEZ	VOUS MANGEATES
ILS MANGENT	ILS MANGERENT
ELLES MANGENT	ELLES MANGERENT
IMPARFAIT	FUTUR SIMPLE
JE MANGEAIS	JE MANGERAI
TU MANGEAIS	TU MANGERAS
IL MANGEAIT	IL MANGERA
ELLE MANGEAIT	ELLE MANGERA
NOUS MANGIONS	NOUS MANGERONS
VOUS MANGIEZ	VOUS MANGEREZ
ILS MANGERAIENT	ILS MANGERONT
ELLES MANGERAIENT	ELLES MANGERONT

CONJUGAISONS





```

10 REM CONJUGAISONS
20 REM -----
30 DIM A$(56)
40 GOSUB 400
50 MODE 1:INK 0,1:INK 1,24: BORDER 1
60 INPUT"VERBE ";V$
70 IF V$="" THEN 60
80 IF RIGHT$(V$,2)="ER" THEN G=1:GOTO 100
90 IF RIGHT$(V$,2)="IR" THEN G=2 ELSE PRINT"VERBE
INCONNU ...":GOTO 60
100 Z$=LEFT$(V$,1):A$(1)="JE"
110 IF Z$="A" OR Z$="E" OR Z$="I" OR Z$="O" OR Z$=
"U" OR Z$="Y" THEN A$(1)="J'"
120 CLS:PRINT TAB(10);"VERBE ";V$
130 V$=LEFT$(V$,LEN(V$)-2):V1$=LEFT$(V$,LEN(V$)-1)
140 IF G=2 THEN U$="I" ELSE U$=""
150 X=1:Y=3
160 FOR T=1 TO 4
170 LOCATE X,Y:PRINT T$(T):Y=Y+1
180 P=G*6+T*12-9
190 FOR I=1 TO 8
200 Y=Y+1:LOCATE X,Y
210 X$=LEFT$(A$(P),1)
220 IF G=1 AND RIGHT$(V$,1)="G" AND (X$="A" OR X$=
"O") THEN PRINT A$(I);" ";V$;"E";A$(P):GOTO 260
230 IF G=1 AND T=1 AND RIGHT$(V$,1)="Y" AND (I<5 OR
I>6) THEN PRINT A$(I);" ";V1$;"I";A$(P):GOTO 260
240 IF G=1 AND T=4 AND RIGHT$(V$,1)="Y" THEN PRINT
A$(I);" ";V1$;"I";A$(P):GOTO 260
250 PRINT A$(I);" ";V$;U$;A$(P)
260 IF I<>3 AND I<>7 THEN P=P+1
270 NEXT I
280 Y=Y+3:IF Y>24 THEN X=X+20:Y=3
290 NEXT T
300 IF INKEY$<>CHR$(13) THEN 300
310 GOTO 50
380 REM -----
390 REM LECTURE DES DATAS
400 FOR I=1 TO 4
410 READ T$(I)
420 NEXT I
430 FOR I=1 TO 56
440 READ A$(I)
450 NEXT I
460 RETURN
470 DATA PRESENT, IMPARFAIT, PASSE SIMPLE, FUTUR SIMP
LE
480 DATA JE, TU, IL, ELLE, NOUS, VOUS, ILS, ELLES
490 DATA E, ES, E, ONS, EZ, ENT
500 DATA S, S, T, SSONS, SSEZ, SSENT
510 DATA AIS, AIS, AIT, IONS, IEZ, AIENT
520 DATA SSAIS, SSAIS, SSAIT, SSIONS, SSIEZ, SSAIENT
530 DATA AI, AS, A, AMES, ATEs, ERENT
540 DATA S, S, T, MES, TES, RENT
550 DATA ERAI, ERAS, ERA, ERONS, EREZ, ERONT
560 DATA RAI, RAS, RA, RONS, REZ, RONT

```

LE PROGRAMME

Lignes 30 à 110 : saisie du verbe

Le verbe à étudier est saisi dans la variable V\$. Si la terminaison à l'infinitif est différente de ER ou IR, on affiche le message "Verbe inconnu".

Dans le cas où la première lettre de celui-ci est une voyelle, on remplace le JE de A\$(1) par J'.

Lignes 120 à 310 : édition

Les verbes en YER prennent soit un I soit un Y dans leurs terminaisons, et cela suivant le temps étudié. On place donc, dans la variable V1\$, le verbe à éditer en enlevant le dernier caractère. Les verbes du deuxième groupe (G=2) comportant toujours un I pour premier caractère de leur terminaison, on place le caractère I dans la variable U\$.

On édite avec une boucle FOR/NEXT (variable T) le verbe au présent, à l'imparfait, au passé simple et au futur simple. Pour chaque temps, on calcule le pointeur P du tableau des terminaisons (ligne 180).

La ligne 220 tient compte des verbes en GER. On ajoute donc la lettre E devant la terminaison dans le cas où un "O" ou un "A" suivent le G (exemple : mangons → mangeons).

Les lignes 230 et 240 tiennent compte, elles, des verbes en YER. On utilise à ce moment-là la variable V1\$ citée plus haut.

Lignes 400 à 560 : lecture des DATAs

On place dans le tableau T\$ la liste des pronoms et dans A\$ celle des terminaisons qui se trouvent dans les DATAs des lignes 470 à 560.

LISTE DES VARIABLES :

A\$()	liste des terminaisons
G	groupe du verbe à étudier
I	indice de boucle FOR/NEXT
P	pointeur du tableau des terminaisons
T	indice de boucle FOR/NEXT (temps à éditer)
T\$()	liste des pronoms
U\$	U\$="I" si verbe du deuxième groupe
V\$	verbe à éditer
V1\$	verbe à éditer sans le dernier caractère
X\$	contient le 1 ^{er} caractère de la terminaison
Z\$	premier caractère du verbe à éditer
X et Y	coordonnées sur l'écran du verbe à éditer

SUITES LOGIQUES

Sans vouloir se comparer à un test poussé de QI, ce programme de suites logiques peut constituer un bon entraînement au raisonnement.

Il consiste à trouver le nombre découlant d'une suite. Pour cela, il propose trois nombres dans un ordre donné, obéissant à un algorithme utilisant l'addition, la soustraction ou la multiplication.

Exemple : suite 1, 14, 92


Vous devez donner votre première solution. Si elle est fautive, AMSTRAD vous donne le quatrième élément de la suite et vous demande votre deuxième solution. Si elle est encore fautive, l'ordinateur affiche la "bonne solution".

Le jeu a trois niveaux de difficulté : il commence au niveau 1 et passe directement aux niveaux 2 puis 3 si votre première solution est bonne.

Si vous ne trouvez le bon résultat qu'au deuxième coup, le niveau de jeu ne change pas. Dans le cas où vous ne trouvez pas du tout la solution, le jeu redescendra automatiquement au niveau précédent.

Je n'explique pas ici la méthode utilisée pour programmer la suite logique ; cela rendrait le jeu trop facile ; les tricheurs peuvent toujours décortiquer le programme !

SUITES LOGIQUES



```

10 REM SUITES LOGIQUES
20 REM -----
30 MODE 1:INK 0,0:INK 1,15:BORDER 0
40 N=1:A(1)=1
50 RANDOMIZE TIME
80 REM -----
90 REM LE JEU
100 I=INT(RND(1)*10)+1:J=INT(RND(1)*10)+1
110 K=INT(RND(1)*10)+1:L=INT(RND(1)*10)+1
120 E=E+1
130 ON N GOTO 210,200,300
180 REM -----
190 REM NIVEAU 1 & 2
200 IF K>5 THEN J=-J
210 FOR U=2 TO 5
220 A(U)=A(U-1)*I+J
230 NEXT U
240 GOTO 500
280 REM -----
290 REM NIVEAU 3
300 IF K>5 THEN J=-J
310 IF L>5 THEN 360
320 FOR U=2 TO 5
330 A(U)=A(U-1)*I+J
340 NEXT U
350 GOTO 500
360 FOR U=2 TO 5
370 A(U)=(A(U-1)+J)*I
380 NEXT U
480 REM -----
490 REM LA SUITE
500 FOR U=1 TO 2
510 CLS:PRINT"PARTIE: ";E,"NIVEAU: ";N:PRINT
520 FOR V=1 TO 2+U
530 PRINT"LE CHIFFRE";V;"EST ";A(V)
540 NEXT V:PRINT
550 PRINT"LE CHIFFRE";3+U;"EST ";:INPUT S
560 IF S=A(3+U) THEN 700
570 NEXT U
580 CLS:PRINT"LA SOLUTION: ":PRINT
590 FOR U=1 TO 5
600 PRINT U;"=",A(U)
610 NEXT U
620 P=P-1-N
630 IF N>1 THEN N=N-1
640 GOTO 800
680 REM -----
690 REM GAGNE
700 PRINT:PRINT"VOUS AVEZ TROUVE A L'ESSAI";U
710 IF U=2 THEN 750
720 P=P+1+N
730 IF N<3 THEN N=N+1
740 GOTO 800
750 P=P+N
780 REM -----
790 REM FIN
800 PRINT
810 PRINT"POINTS: ";P,"NB DE PARTIES: ";E
820 PRINT:PRINT"VOUS ETES MAINTENANT AU NIVEAU";N
830 IF INKEY$="" THEN 830
840 GOTO 100

```

LE PROGRAMME

Lignes 30 à 130 : initialisation

La suite logique étant placée dans le tableau A(1) à A(5), A(1) est mis dès le début du jeu à 1.

Les lignes 100 et 110 placent dans les variables I, J, K et L quatre nombres aléatoires qui détermineront la suite logique.

La ligne 130 renvoie sur les calculs de suites, suivant le niveau de jeu (variable N).

Lignes 200 à 380 : calcul de la suite logique

Ces lignes calculent pour chaque niveau respectif la suite logique correspondante (5 chiffres au total). Puis chacune de ces lignes renvoie en 500 pour afficher les trois premiers chiffres.

Lignes 500 à 610 : le jeu

Le programme saisit vos deux réponses par une boucle FOR/NEXT (variable U).

Si votre proposition est bonne, la ligne 560, après un test, envoie en ligne 700 pour un message de félicitations.

Dans le cas contraire, la même suite est affichée avec un chiffre supplémentaire.

Si vous n'avez pas trouvé, les lignes 580 à 610 affichent la bonne solution.


Lignes 620 à 840 : gagné-perdu

Les lignes 620 à 750 calculent votre nombre de points suivant le niveau et le nombre d'essais qu'il vous a fallu pour trouver la bonne solution. Si vous n'avez pas trouvé du tout, votre capital diminue de la même manière.


Les lignes 800 à 840 affichent les résultats de la dernière partie.

LISTE DES VARIABLES :

A() contient la suite logique
E nombre de parties
I, J, K et L nombres aléatoires pour le calcul de la suite
N niveau du jeu
P nombre de points
S solution proposée
U indice de boucle FOR/NEXT
V indice de boucle FOR/NEXT



```
PARTIE: 3            NIVEAU: 3  
LE CHIFFRE 1 EST 1  
LE CHIFFRE 2 EST 5  
LE CHIFFRE 3 EST 29  
LE CHIFFRE 4 EST 173  
LE CHIFFRE 5 EST ? 1037  
VOUS AVEZ TROUVE A L'ESSAI 2  
POINTS: 8            NB DE PARTIES: 3  
VOUS ETES MAINTENANT AU NIVEAU 3
```



THEME 3 CUISINE ET TRICOT

FICHER DE RECETTES

Comment préparer vos tournedos pour ce soir ? Quelle recette choisir pour utiliser les tomates de votre jardin ? Le mieux est de demander à votre AMSTRAD.

Le programme propose les options suivantes :

- **SAISIE** : vous devez entrer tous les renseignements d'une recette : nom, ingrédients et préparation.

Si vous ne voulez plus entrer de recette, tapez uniquement sur "ENTER" à la question "NOM:".

- **MODIFICATION** : si vous désirez modifier une recette, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à apporter, tapez "Ø" en réponse à la question "NUMERO DE LA RECETTE:".


- **SUPPRESSION** : si vous désirez enlever une recette du fichier, cette option permet de la supprimer définitivement. Comme pour la modification, tapez "Ø" pour revenir au menu.

- **RECHERCHE** : cette partie recherche le mot que vous lui donnez dans une des trois données d'une recette. Cela vous permet de retrouver les recettes dont le nom contient le mot "Bœuf", ou les recettes utilisant comme ingrédient des "Tomates". Il n'est pas nécessaire de donner un nom entier : le mot "Bœuf" suffit à retrouver la recette du "Bœuf en Daube", du "Bœuf Bourguignon" ou du "Bœuf à la Flamande".

- **LISTING** : éditez simplement sur l'écran le fichier dans son intégralité.

- **SAUVEGARDE** : le programme sauve le fichier sur cassette.

FICHER DE RECETTES



```

10 REM FICHER DE RECETTES
20 REM -----
30 DIM D$(3,100)
40 L$(1)="NOM":L$(2)="INGREDIENTS":L$(3)="RECETTE"
50 MODE 1:INK 0,0:INK 1,15:BORDER 0
60 INPUT"PREMIERE UTILISATION (O/N)":R$
70 IF R$="O" THEN 200
80 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
90 OPENIN"!RECETTE"
100 INPUT#9,N
110 FOR I=1 TO N
120 FOR J=1 TO 3
130 LINE INPUT#9,D$(J,I)
140 NEXT J,I
150 CLOSEIN
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE RECETTES: ";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INKEY$:IF R$="" THEN 290
300 R=VAL(R$):IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE
400 CLS:PRINT"SAISIE D'UNE RECETTE":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:LINE INPUT
R$:GOTO 200
420 N=N+1
430 FOR I=1 TO 3
440 PRINT L$(I);": ";:LINE INPUT D$(I,N)
450 IF I=1 AND D$(1,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 3:PRINT I:L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$=INKEY$:IF R$="" THEN 640
650 R=VAL(R$):IF R<1 OR R>3 THEN 200
660 PRINT L$(R);": ";:LINE INPUT D$:PRINT
670 FOR I=1 TO N
680 IF INSTR(D$(R,I),D$)=0 THEN 740
690 F=1:PRINT USING"### ";I:PRINT"*";D$(1,I);"*"
700 FOR J=2 TO 3
710 FOR K=1 TO LEN(D$(J,I)) STEP 32
720 PRINT" ";MID$(D$(J,I),K,32)
730 NEXT K,J:PRINT
740 NEXT I:PRINT
750 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
760 IF INKEY$="" THEN 760 ELSE 600
880 REM -----
890 REM MODIFICATION

```

```

900 CLS:PRINT"MODIFICATION":PRINT
910 INPUT"NUMERO DE LA RECETTE";NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 3
940 PRINT I;L$(I);": ";D$(I,NU)
950 NEXT I
960 PRINT:INPUT"MODIFICATION DU";R
970 IF R<1 OR R>3 THEN 200
980 PRINT L$(R);": ";;:LINE INPUT D$(R,NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS:PRINT"SUPPRESSION":PRINT
1110 INPUT"NUMERO DE LA RECETTE";NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 3
1150 PRINT I;L$(I);": ";D$(I,NU)
1160 NEXT I:PRINT
1170 INPUT"VOUS ETES SUR DE SUPPRIMER
    CETTE RECETTE";R$
1180 IF R$<>"0" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 3
1220 D$(J,I)=D$(J,I+1)
1230 NEXT J,I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS:F=0
1310 FOR I=1 TO N
1320 PRINT USING"### ";I;:PRINT"*";D$(1,I);"*":F=F
+1
1330 FOR J=2 TO 3
1340 FOR K=1 TO LEN(D$(J,I)) STEP 32
1350 PRINT" ";MID$(D$(J,I),K,32):F=F+1
1360 NEXT K,J:PRINT:F=F+1
1370 IF F>15 THEN PRINT:PRINT"APPUYEZ SUR ENTER PO
UR LA SUITE";:LINE INPUT R$:CLS:F=0
1380 NEXT I
1390 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE ME
NU";
1400 IF INKEY$="" THEN 1400 ELSE 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORD
ER EST PRET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPENOUT"!RECETTE"
1530 PRINT#9,N
1540 FOR I=1 TO N
1550 FOR J=1 TO 3
1560 PRINT#9,D$(J,I)
1570 NEXT J,I
1580 CLOSEOUT
1590 CLS:PRINT"A LA PROCHAINE !"
1600 END

```

LE PROGRAMME

Lignes 30 à 150 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'une opération

On saisit par une boucle FOR/NEXT les trois données du fichier. Si le nom de la recette est une chaîne vide, on revient au menu (ligne 450).

Lignes 600 à 760 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$).

Les lignes 670 à 740 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1400 : listing


On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1370 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
LS()	libellés des données
N	nombre de recettes dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$



```
NOM : TOURNEDOS

1 *TOURNEDOS*
4 TOURNEDOS, 4 CROUTONS, 100G DE
BEURRE
FAIRE DORER LES CROUTONS DANS LA
MOITIE DU BEURRE. FAIRE GRILLER
LES TOURNEDOS DANS LE RESTE DU
BEURRE. CUIRE DE CHAQUE COTE 5 M
N. SALER, PLACER CHAQUE TOURNEDO
S SUR UN CROUTON AVANT DE SERVIR
.

2 *TOURNEDOS A LA LIMOUSINE*
4 TOURNEDOS, 4 BEAUX CEPES, 4 CR
OUTONS RISSOLES AU BEURRE.
FAIRE CUIRE 5 MN DE CHAQUE COTE
ET ENLEVER. FAIRE FRIRE LES CEPE
S DANS L'HUILE; ASSAISONNER D'AI
L ECHALOTES, SEL ET POIVRE. SER
VIR EN COURONNE TOURNEDOS, CROUT
ONS ET CEPES.
```

RECETTES DE CUISINE

Voici le moyen infaillible de faire des gâteaux extraordinaires tous différents les uns des autres.

Très simple ! Vous tapez RUN et en quelques secondes votre ordinateur affiche à l'écran une recette inédite de gâteau au chocolat et aux noix.


Vous n'avez plus qu'à réunir tous les ingrédients dont la liste vous est donnée et à suivre les instructions de fabrication.

Dès que votre gâteau sera cuit, vous pourrez le déguster avec la boisson choisie par AMSTRAD, vin ou cidre...


Un petit conseil cependant : pour déjouer les farces éventuelles de votre ordinateur, vérifiez que la combinaison de tous les ingrédients de la recette soit vraisemblable.

De toute façon, il suffit d'appuyer sur ENTER pour obtenir une nouvelle recette.

Bon appétit !



RECETTE DE CUISINE



```

10 REM CUISINE
20 REM -----
30 DIM C$(24),Q(24),U(24),A(24)
40 DATA FARINE,100,2,3,SUCRE,50,2,2,EAU,10,4,2,LAI
T,10,4,2,OEUFS,1,1,3
50 DATA VANILLE,.5,5,2,BLANC OEUF,1,1,3,HUILE,1,3,
2,BEURRE,50,2,2,LEVURE,.5,5,2
60 DATA RAISINS S.,50,2,2,RHUM,.5,6,1,COGNAC,.5,6,
1,NOIX,50,2,2,NOISETTES,50,2,2
70 DATA CHOCOLAT,50,2,2,JAUNE D'OEUF,1,1,2,OEUF NE
IGE,1,1,3,POMMES,200,2,2
80 DATA CERISES,100,2,2,SEL,1,7,2,AMANDES,50,2,2,F
RUIFS C.,50,2,2,CREME,25,2,2
90 DATA ,GRAM.,CUIL.,CLS,SACHET,VERRE,PINC.
100 DATA VIN ROUGE,VIN BLANC,CIDRE,CHAMPAGNE,JUS D
'ORANGE,VIN CUIT
110 MODE 1:INK 0,1:INK 1,24:BORDER 1
120 RANDOMIZE TIME
180 REM -----
190 REM LECTURE DES INGREDIENTS
200 FOR I=1 TO 24
210 READ C$(I),Q(I),U(I),A(I)
220 NEXT I
230 FOR I=1 TO 7
240 READ U$(I)
250 NEXT I
260 FOR I=1 TO 6
270 READ B$(I)
280 NEXT I
380 REM -----
390 REM ELEMENTS DE BASE
400 CLS:PRINT TAB(14);"RECETTE":PRINT:PRINT
410 NE=1:GOSUB 1000
420 NE=2:GOSUB 1000
430 NL=INT(RND(1)*2)+1
440 FOR I=1 TO NL
450 NE=INT(RND(1)*3)+3
460 IF I=1 THEN 480
470 IF NE=L(1) THEN 450
480 GOSUB 1000
490 L(I)=NE
500 NEXT I
580 REM -----
590 REM ELEMENTS DU GATEAU
600 NS=INT(RND(1)*3)+3
610 FOR I=1 TO NS
620 NE=INT(RND(1)*19)+6
630 IF I=1 THEN 670
640 FOR J=1 TO I-1
650 IF NE=S(J) THEN 620
660 NEXT J
670 GOSUB 1000
680 S(I)=NE
690 NEXT I
780 REM -----
790 REM RECETTE
800 PRINT:PRINT:PRINT "MELANGER FARINE, SUCRE";
810 IF NL=1 THEN 840
820 PRINT", ";C$(L(1));" ET ";C$(L(2));
830 GOTO 850

```

```

840 PRINT " ET ";C$(L(1));
850 PRINT " POUR OBTENIR LA PATE. ";
860 PRINT " AJOUTER ";C$(S(1));
870 FOR I=2 TO NS-1
880 PRINT ", ";C$(S(I));
890 NEXT I
900 PRINT " ET ";C$(S(NS));". ";
910 PRINT " BEURRER UN MOULE ET Y METTRE LA PATE OB
TENU. ";
920 PRINT " PLACER CE MOULE DANS UN FOUR A";100+(10
*INT(RND(1)*6));"DEGRES. ";
930 PRINT " LAISSER CUIRE";30+(5*INT(RND(1)*7));"MI
NUTES. ";
940 PRINT " DEGUSTER CE GATEAU AVEC DU ";B$(INT(RND
(1)*6)+1);". ";
950 IF INKEY$<>CHR$(13) THEN 950
960 GOTO 400
980 REM -----
990 REM CALCUL DES INGREDIENTS
1000 QU=Q(NE)*INT(RND(1)*A(NE)+1)
1010 IF QU=0.5 THEN 1040
1020 PRINT TAB(5);C$(NE);TAB(21);": ";QU;U$(U(NE))
1030 RETURN
1040 PRINT TAB(5);C$(NE);TAB(21);": 1/2 ";U$(U(NE)
)
1050 RETURN

```

LE PROGRAMME

Lignes 30 à 280 : lecture des ingrédients

Tous les éléments des DATAs des lignes 40 à 100 sont placés dans leurs tableaux respectifs (voir identification des variables).

Lignes 400 à 500 : éléments de base

La farine et le sucre sont les deux éléments indispensables à la pâte. On appelle donc toujours le sous-programme des lignes 1000 à 1050 pour ces deux éléments.

On ajoute, pour compléter cette pâte, un ou deux éléments liquides. Le nombre de liquides est déterminé par la variable NL. Dans le cas où deux éléments liquides sont utilisés, on vérifie qu'il ne s'agit pas deux fois des mêmes.

Lignes 600 à 690 : éléments du gâteau

Puis c'est au tour des éléments solides. On utilise le même procédé que pour les liquides. Chaque solide est placé dans le tableau S pour la vérification des doublons. Ce tableau sera aussi utilisé pour l'édition du texte de la recette (lignes 860 à 880).

Lignes 800 à 960 : recette

Chaque élément liquide et solide est repris dans le texte de la recette. Deux nombres aléatoires donnent la température et le temps de cuisson du gâteau. Puis on affiche un des éléments du tableau des boissons.

Lignes 1000 à 1050 : sous-programme

La variable NE correspond au numéro de l'élément à éditer. La quantité à utiliser est comprise entre une base (tableau Q) et cette base multipliée par le nombre aléatoire maximum (tableau A). Dans le cas où l'on obtient la valeur 0.5, la ligne 1040 affiche le message "1/2".

LISTE DES VARIABLES :


A()	valeur maximale aléatoire pour chaque ingrédient
B\$()	liste des boissons
C\$()	liste des composants
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	liste des éléments liquides
NE	numéro de l'élément à éditer
NL	nombre de liquides
NS	nombre de solides
Q()	liste des quantités de base de chaque élément
QU	quantité à utiliser
S()	liste des solides
U()	liste des unités
U\$()	nom des unités

MENUS ALEATOIRES


Si vous manquez d'idées de menu pour le repas de ce soir, AMSTRAD vous propose une infinité de combinaisons, composées bien sûr de vos plats préférés.

Nul besoin de longues explications. Exécutez le programme, un menu s'affiche automatiquement. Il suffit d'appuyer sur "ENTER" pour obtenir d'autres repas.

Bon appétit !



MENUS ALEATOIRES



```

10 REM MENUS ALEATOIRES
20 REM -----
30 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1:BORDER 1
40 DIM E$(158)
50 RANDOMIZE TIME
60 FOR I=1 TO 158
70 READ E$(I)
80 NEXT I
180 REM -----
190 REM LES MENUS
200 CLS:PRINT
210 PRINT TAB(14);"MENU DU JOUR"
230 LOCATE 17,6:PRINT"ENTREE:"
240 E=INT(RND(1)*46)+1
250 LOCATE 20-LEN(E$(E))/2,8:PRINT E$(E)
260 LOCATE 18,11:PRINT"PLAT:"
270 E=INT(RND(1)*76)+47
280 IF RIGHT$(E$(E),1)="*" THEN LOCATE 20-LEN(E$(E))/2,13:PRINT LEFT$(E$(E),LEN(E$(E))-1):GOTO 320
290 LOCATE 20-LEN(E$(E))/2,13:PRINT E$(E)
300 E=INT(RND(1)*12)+123
310 LOCATE 20-LEN(E$(E))/2,14:PRINT E$(E)
320 LOCATE 16,17:PRINT"DESSERT:"
330 E=INT(RND(1)*24)+135
340 LOCATE 20-LEN(E$(E))/2,19:PRINT E$(E)
350 IF INKEY$<>CHR$(13) THEN 350
360 GOTO 200
980 REM -----
990 REM LES ENTREES
1000 DATA SOUPE JULIENNE, SOUPE A L'OIGNON, SOUPE PO
    IREAUX/POMMES DE TERRE
1010 DATA SARDINES, SAUMON, MAQUEREAUX VIN BLANC, HAR
    ENGS, THON MAYONNAISE, COQUILLES ST JACQUES
1020 DATA PATE DE CAMPAGNE, PATE DE FOIE, PATE DE CA
    NARD, TERRINE DE LIEVRE, RILLETES
1030 DATA JAMBON CRU, JAMBON CUIT, MUSEAU VINAIGRETT
    E, SAUCISSON SEC, SAUCISSON A L'AIL, ANDOUILLE
1040 DATA SALADE MIXTE, SALADE DE CONCOMBRES, SALADE
    DE MAIS, OEUFS EN GELEE, OEUFS MAYONNAISE, OEUFS A L
    A COQUE, SALADE DE TOMATES, MELON AU PORTO, RADIS, RAD
    IS NOIR, ARTICHAUTS VINAIGRETTE
1050 DATA CHOU ROUGE, CHOU-FLEUR, BOUCHEES A LA REIN
    E, QUICHE LORRAINE, TARTE AUX OIGNONS, PIZZA, CREPES A
    U JAMBON, CREPES AUX CHAMPIGNONS
1060 DATA ASPERGES VINAIGRETTE, SALADE D'ENDIVES, PO
    IREAUX VINAIGRETTE, CAROTTES RAPEES, FONDS D'ARTICH
    AUTS, LENTILLES VINAIGRETTE, CHAMPIGNONS A LA GRECQU
    E
1070 REM LES PLATS
1080 DATA BIFTECK FRITES*, BOEUF BOURGUIGNON*, BOEUF
    EN DAUBE*, BIFTECK HACHE PUREE*, POT-AU-FEU*, ENTREC
    OTE FRITES*, TOURNEDOS
1090 DATA FOIE GRILLE PUREE*, LANGUE DE BOEUF JARDI
    NIERE*, ENTRECOTES, GRAS-DOUBLE PETITS POIS*
1100 DATA ESCALOPES CHAMPIGNONS*, ESCALOPES A LA CR
    EME*, COTELETTES DE VEAU, ROTI DE VEAU, VEAU MARENGO*
    , RIS DE VEAU*, ROGNONS DE VEAU PUREE*
1110 DATA TETE DE VEAU, FOIE DE VEAU
1120 DATA COTELETTES DE MOUTON, EPAULE DE MOUTON, GI

```

GOT DE MOUTON FLAGEOLETS*, RAGOUT DE MOUTON*, NAVARI
N*
1130 DATA BROCHETTES D'AGNEAU RIZ*, EPAULE D'AGNEAU
, BLANQUETTE DE VEAU*
1140 DATA COTELETTES DE PORC, ROTI DE PORC, JAMBONNE
AU, ANDOUILLETES FRITES*, BOUDIN, BOUDIN BLANC, PIEDS
DE PORC, SAUCISSES AU CHOU*
1150 DATA POULET ROTI, POULET CURRY RIZ*, POULE AU P
OT, POULE AU RIZ*, COQ AU VIN*, OIE ROTI, CONFIT D'OIE
*
1160 DATA DINDE AUX MARRONS*, ROTI DE DINDONNEAU, CA
NARD A L'ORANGE*, CANETONS ROTIS
1170 DATA CIVET DE LIEVRE, CIVET DE LAPIN
1180 DATA FILET DE BARBUE, CABILLAUD FRIT, DAURADE G
RILLEE, LIMANDE, LOTTE, MERLANS FRITS, BRANDADE DE MOR
UE*, RAIE BEURRE NOIR
1190 DATA FILETS DE SOLE, SOLE MEUNIERE, TURBOT, MOUL
ES MARINIERE RIZ*, TRUITES AUX AMANDES*
1200 DATA QUENELLÉS DE POISSON*, ESCARGOTS DE BOURG
OGNE*, CUISSÉS DE GRENOUILLES*
1210 DATA TOMATES FARCIES*, CHOUCRUTE GARNIE*, CASS
OULET*, AUBERGINES FARCIES*, COUSCOUS*, PAELLA*
1220 DATA OMELETTE AUX CHAMPIGNONS*, OMELETTE AUX P
OMMES DE TERRE*, ENDIVES AU JAMBON*, LASAGNES*, CANNE
LLONIS*
1230 REM LEGUMES
1240 DATA PUREE, POMMES DE TERRE, HARICOTS VERTS, PET
ITS POIS, TOMATES PROVENCALES, EPINARDS, SALSIFIS, LEN
TILLES
1250 DATA PATES, SPAGHETTIS, NOUILLES, RIZ
1260 DESSERT
1270 DATA GATEAU DE RIZ, BEIGNETS AUX POMMES, CREME
AU CAMEL, MOUSSE AU CHOCOLAT, PUDDING
1280 DATA CHARLOTTE, FLAN, CREPES, CAKE, CLAFOUTIS, TAR
TE, PAIN PERDU, QUATRE-QUARTS, CREME GLACEE, PATISSERI
E
1290 DATA BANANES FLAMBEES, MELON, SALADE DE FRUITS,
GATEAU MAISON, FRUITS DE SAISON
1300 DATA FROMAGE, CAMEMBERT, FROMAGE BLANC, YAOURT

LE PROGRAMME

Lignes 30 à 80 : initialisation

Le programme lit dans les DATAs des lignes 1000 à 1300 tous les éléments qui vont composer vos menus. Cette liste n'est pas exhaustive. Le lecteur peut ajouter des éléments dans les DATAs entre les lignes 1060 et 1070 pour les entrées, entre les lignes 1220 et 1230 pour les plats, entre les lignes 1250 et 1260 pour les légumes et après la ligne 1300 pour les desserts. Il faut bien sûr modifier le nombre 158 de la ligne 60 (il y a 158 éléments pour l'instant) mais aussi la ligne 240 (il y a 46 entrées, la première est en 1), la ligne 270 (il y a 76 plats, le premier est en 47), la ligne 300 (il y a 12 légumes, le premier est en 123) et pour finir la ligne 330 (il y a 24 desserts, le premier est en 135).

Tous les plats qui n'ont pas besoin d'un légume doivent être suivis d'une étoile.

Lignes 200 à 360 : les menus

Le programme affiche quatre types d'éléments au hasard. Si le nom du plat se termine par une étoile, on ne tient pas compte des légumes pour éviter les "couscous nouilles" et les "choucroutes garnies spaghettis".

LISTE DES VARIABLES :

E	pointe l'élément choisi au hasard
ES()	liste des éléments
I	indice de boucle FOR/NEXT

GESTION D'UN CONGELATEUR

Vous avez du mal à gérer le contenu de votre congélateur ? AMS-TRAD va vous donner un petit coup de main !

Il vous demande en priorité si le fichier a déjà été créé. Si c'est le cas, tapez uniquement sur la touche "ENTER" pour relire les données. Puis le programme affiche un menu pour les options suivantes :

- 1 – **SAISIE** : vous devez entrer tous les renseignements sur l'aliment que vous venez de congeler. Les dates doivent être saisies sous la forme année, mois et jour pour faciliter les recherches de l'option 4, et comporter 6 chiffres (exemple : le 09/03/84 s'écrit ici 840309).

Si vous n'avez plus rien à rentrer, appuyez sur "ENTER" à la question "Date de congélation :".

- 2 – **MODIFICATION** : si vous désirez modifier un élément, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à ajouter, tapez "0" en réponse à la question "Numéro de l'élément :".

- 3 – **SUPPRESSION** : dès qu'un aliment est sorti du congélateur, vous pouvez le supprimer par cette option.


Comme pour la modification, tapez 0 pour revenir au menu.

- 4 – **RECHERCHE** : elle permet d'afficher tous les aliments qui seront périmés entre deux dates précises, de connaître toutes les sortes de crêpes qui vous attendent impatiemment dans votre congélateur, ou tous les petits plats que vous y avez mis durant l'année 84. Le simple mot "Crêpe" permet de retrouver aussi bien les crêpes au jambon que celles aux champignons.

- 5 – **LISTING** : éditez simplement sur l'écran le fichier dans son intégralité.

- 6 – **SAUVEGARDE** : le programme sauve le fichier sur cassette ou sur disquette.

GESTION DU CONGELATEUR



```

10 REM GESTION DU CONGELATEUR
20 REM -----
30 DIM D$(3,100)
40 L$(1)="DATE DE CONGELATION(AAMMJJ)":L$(2)="DATE
DE PEREMPTION (AAMMJJ)":L$(3)="NATURE DU PRODUIT"
50 MODE 1:INK 0,0:INK 1,15:BORDER 0
60 INPUT"PREMIERE UTILISATION (O/N)";R$
70 IF R$="O" THEN 200
80 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
90 OPENIN "!CONGEL"
100 INPUT#9,N
110 FOR I=1 TO N
120 FOR J=1 TO 3
130 LINE INPUT#9,D$(J,I)
140 NEXT J,I
150 CLOSEIN
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB D'ELEMENTS: ";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INKEY$:IF R$="" THEN 290
300 R=VAL(R$):IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE
400 CLS:PRINT"SAISIE D'UN ELEMENT":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:LINE INPUT
R$:GOTO 200
420 N=N+1
430 FOR I=1 TO 3
440 PRINT L$(I);":":LINE INPUT D$(I,N)
450 IF I=1 AND VAL(D$(1,N))=0 THEN N=N-1:GOTO 200
460 IF I<3 AND LEN(D$(1,N))<>6 THEN 440
470 NEXT I
480 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 3:PRINT I:L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$=INKEY$:IF R$="" THEN 640
650 R=VAL(R$):IF R<1 OR R>3 THEN 200
660 IF R=3 THEN 700
670 INPUT"DATE DEBUT (AAMMJJ) ";DD
680 INPUT"DATE FIN (AAMMJJ) ";DF
690 GOTO 710
700 PRINT L$(R);":":LINE INPUT D$:PRINT
710 PRINT
720 FOR I=1 TO N
730 IF R=3 THEN IF INSTR(D$(R,I),D$)=0 THEN 800 EL
SE 750
740 IF VAL(D$(R,I))<DD OR VAL(D$(R,I))>DF THEN 800
750 F=1:PRINT USING"### ";I;
760 FOR J=1 TO LEN(D$(3,I)) STEP 35
770 PRINT TAB(5);MID$(D$(3,I),J,35)
780 NEXT J

```

```

790 PRINT"      CONGELE LE ";D$(1,I);" PERIME LE ";D
$(2,I);:PRINT
800 NEXT I:PRINT
810 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
820 IF INKEY$="" THEN 820 ELSE 600
880 REM -----
890 REM MODIFICATION
900 CLS:PRINT"MODIFICATION":PRINT
910 INPUT"NUMERO DE L'ELEMENT";NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 3
940 PRINT I;L$(I);": ";D$(I,NU)
950 NEXT I
960 PRINT:INPUT"MODIFICATION DU";R
970 IF R<1 OR R>3 THEN 200
980 PRINT L$(R);": ";:LINE INPUT D$(R,NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS:PRINT"SUPPRESSION":PRINT
1110 INPUT"NUMERO DE L'ELEMENT";NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 3
1150 PRINT I;L$(I);": ";D$(I,NU)
1160 NEXT I:PRINT
1170 INPUT"VOUS ETES SUR DE SUPPRIMER
CET ELEMENT";R$
1180 IF R$<>"0" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 3
1220 D$(J,I)=D$(J,I+1)
1230 NEXT J,I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS:F=0
1310 FOR I=1 TO N
1320 PRINT USING"### ";I;
1330 FOR J=1 TO LEN(D$(3,I)) STEP 35
1340 PRINT TAB(5);MID$(D$(3,I),J,35):F=F+1
1350 NEXT J
1360 PRINT"      CONGELE LE ";D$(1,I);" PERIME LE ";
D$(2,I):PRINT:F=F+2
1370 IF F>16 THEN PRINT:PRINT"APPUYEZ SUR ENTER PO
UR LA SUITE";:LINE INPUT R$:CLS:F=0
1380 NEXT I
1390 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE ME
NU";
1400 IF INKEY$="" THEN 1400 ELSE 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORD
ER
EST PRET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPENOUT "!CONGEL"
1530 PRINT#9,N
1540 FOR I=1 TO N
1550 FOR J=1 TO 3
1560 PRINT#9,D$(J,I)
1570 NEXT J,I
1580 CLOSEOUT
1590 CLS:PRINT"A LA PROCHAINE CONGELATION !"
1600 END

```

LE PROGRAMME

Lignes 30 à 150 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes dispatchent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 480 : saisie d'une opération

On saisit par une boucle FOR/NEXT les trois données du fichier. Si aucune date de congélation n'a été entrée, on revient au menu (ligne 450).

Lignes 600 à 820 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$ pour la nature du produit, et date de début DD et date de fin DF pour une période).

Les lignes 720 à 800 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$ ou comprise entre les deux dates DD et DF.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1400 : listing


On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1370 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

DS	nature du produit à rechercher
DS(,)	tableau des données du fichier
DD	date de début des recherches
DF	date de fin des recherches
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre de produits dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
RS	choix de l'utilisateur
R	valeur numérique de R\$



TRICOT

Indispensable à toute tricoteuse, ce petit programme permet de connaître rapidement et sans calcul le nombre de mailles ou de rangs à monter et à tricoter, et d'adapter ainsi à toutes les tailles les explications des tricots aux mesures standardisées.

Il suffit de rentrer le nombre de mailles et de rangs d'un échantillon tricoté de 10 cm sur 10 cm, les mesures diverses du dos et du devant (largeur à la base, largeur après diminutions des emmanchures, hauteur sous les bras, hauteur totale) et les mesures des manches (largeur poignet, largeur à l'épaule, hauteur à l'épaule, hauteur totale).

Dès que tous ces renseignements sont entrés dans votre AMS-TRAD, le programme affiche le nombre de rangs à monter et le nombre de mailles à tricoter de toutes les parties du tricot.


MONTAGE DOS OU DEVANT

- MONTER 124 MAILLES, SOIT 48 CMS.
- TRICOTER 136 RANGS, SOIT 40 CMS.
- A 40 CMS, RABATTRE DE CHAQUE COTE 10 MAILLES, EN PLUSIEURS FOIS, POUR LES EMMANCHURES.
- TRICOTER ENCORE 61 RANGS, SOIT 18 CMS.
- A 58 CMS, FERMER SELON L'ENCOLURE.

MONTAGE DES MANCHES

- MONTER 57 MAILLES, SOIT 22 CMS.
- TRICOTER 146 RANGS, SOIT 43 CMS. EN AUGMENTANT REGULIEREMENT CHAQUE COTE DE 18 MAILLES.
- A 43 CMS, RABATTRE LES 93 MAILLES RESTANTES REGULIEREMENT, SOIT 46 MAILLES DE CHAQUE COTE.
- A 56 CMS, FERMER LE TRICOT.

TRICOT




```

10 REM TRICOT
20 REM -----
30 MODE 1:INK 0,1:INK 1,24:BORDER 1:PRINT"ECHANTIL
LON 10 CMS SUR 10 CMS":PRINT
40 INPUT"NOMBRE DE MAILLES";M1
50 INPUT"NOMBRE DE RANGS";R1
60 M2=M1/10
70 R2=R1/10
80 REM -----
90 REM MESURES DEVANT ET DOS
100 CLS:PRINT"MESURES DEVANT ET DOS":PRINT
110 INPUT"LARGEUR DOS/DEVANT A LA BASE";L1
120 INPUT"LARGEUR POITRINE";L2
130 INPUT"HAUTEUR SOUS LES BRAS";H1
140 INPUT"HAUTEUR TOTALE DOS/DEVANT";H2
150 N1=INT(M2*L1)
160 N2=INT(M2*L2)
170 N3=INT(R2*H1)
180 N4=INT(R2*H2)
190 D=INT((N1-N2)/2)
280 REM -----
290 REM MESURES MANCHES
300 CLS:PRINT"MESURES MANCHES":PRINT
310 INPUT"LARGEUR POIGNET";P1
320 INPUT"LARGEUR MANCHE A L'EPAULE";E1
330 INPUT"HAUTEUR MANCHE A L'EPAULE";H3
340 INPUT"HAUTEUR TOTALE MANCHE";H4
350 P2=INT(M2*P1)
360 E2=INT(M2*E1)
370 N5=INT(R2*H3)
380 N6=INT(R2*H4)
390 A=INT((E2-P2)/2)
480 REM -----
490 REM RESULTATS DOS ET DEVANT
500 CLS:PRINT" MONTAGE DOS OU DEVANT":PRINT
510 PRINT"-MONTER";N1;"MAILLES, SOIT";L1;"CMS."
520 PRINT"-TRICOTER";N3;"RANGS, SOIT";H1;"CMS."
530 PRINT"-A";H1;"CMS, RABATTRE DE CHAQUE COTE"
540 PRINT D;"MAILLES, EN PLUSIEURS FOIS,"
550 PRINT" POUR LES EMMANCHURES."
560 N7=N4-N3
570 C=H2-H1
580 PRINT"-TRICOTER ENCORE";N7;"RANGS, SOIT"
590 PRINT C;"CMS."
600 PRINT"-A";H2;"CMS, FERMER SELON L'ENCOLURE."
680 REM -----
690 REM RESULTATS MANCHES
700 PRINT:PRINT:PRINT" MONTAGE DES MANCHES":PRI
NT
710 PRINT"-MONTER";P2;"MAILLES, SOIT";P1;"CMS."
720 PRINT"-TRICOTER";N5;"RANGS, SOIT";H3;"CMS."
730 PRINT" EN AUGMENTANT REGULIEREMENT"
740 PRINT" CHAQUE COTE DE";A;"MAILLES."
750 N8=(P2+(A*2))
760 N9=INT(N8/2)
770 PRINT"-A";H3;"CMS, RABATTRE LES";N8;"MAILLES"
780 PRINT" RESTANTES REGULIEREMENT, SOIT";N9
790 PRINT" MAILLES DE CHAQUE COTE."
800 PRINT"-A";H4;"CMS, FERMER LE TRICOT."

```

LISTE DES VARIABLES :

A	augmentation des manches de chaque côté
C	cms entre l'épaule et le total (dos ou devant)
D	diminutions des emmanchures de chaque côté
E1	largeur à l'épaule
E2	nombre de mailles à l'épaule
H1	hauteur sous le bras
H2	hauteur totale dos ou devant
H3	hauteur à l'épaule
H4	hauteur totale
I	indice de boucle FOR/NEXT
L1	largeur dos ou devant
L2	largeur à la poitrine
M1	nombre de mailles pour 10 cm
M2	nombre de mailles pour 1 cm
N1	nombre de mailles largeur dos ou devant
N2	nombre de mailles largeur poitrine
N3	nombre de rangs hauteur sous bras
N4	nombre de rangs hauteur totale dos ou devant
N5	nombre de rangs à l'épaule
N6	nombre de rangs hauteur totale
N7	nombre de rangs entre l'épaule et le total dos ou devant
N8	nombre de mailles à l'épaule de la manche
N9	nombre de mailles à rabattre de chaque côté de la manche
P1	largeur poignet
P2	nombre de mailles au poignet
R1	nombre de rangs pour 10 cm
R2	nombre de rangs pour 1 cm



THEME 4 LES JEUX NATIONAUX

LE TIERCE : PRONOSTICS

Vous jouez au tiercé et vous vous lancez dans des calculs fastidieux pour faire la synthèse des avis de vos pronostiqueurs favoris ?

Ne vous trompez plus dans vos numéros ; demandez tout simplement l'aide de votre AMSTRAD !


Entrez le nombre de chevaux de la course, le nombre de pronostics en votre possession et le nombre de chevaux minimum commun à tous les pronostics.


Un court temps de réflexion et l'ordinateur vous donne les chevaux qui ont le plus de chance d'arriver dans les quatre premiers, ainsi que le total des points obtenus dans chaque pronostic d'après l'ordre de classement.

Il vous révélera aussi quatre tiercés possibles d'après ses résultats.

Alors pourquoi ne pas essayer... au moins une fois !

PRONOSTICS		
PLACE	CHEVAL	POINTS
1	10	12
2	12	7
3	7	7
4	13	5
5	1	5
6	25	4
VOUS POUVEZ JOUER LE :		
10	12	7
12	7	13
7	13	10
13	10	12





```

10 REM PRONOSTICS
20 REM -----
30 DIM P(30),R(30)
40 MODE 1:INK 0,1:INK 1,24:BORDER 1
50 INPUT"NOMBRE DE CHEVAUX";NC
60 IF NC>30 THEN 40
70 FOR I=1 TO NC
80 R(I)=I
90 NEXT I
100 PRINT
110 INPUT"NB DE PRNOSTICS";N
120 PRINT
130 INPUT"NB DE CHX/PRONO.";CP
140 CLS
150 FOR I=1 TO N
160 PRINT"PRONOSTICS NO";I
170 FOR J=CP TO 1 STEP-1
180 PRINT"NO DU CHEVAL";CP+1-J;
190 INPUT NP
200 IF NP>NC THEN 180
210 P(NP)=P(NP)+J
220 NEXT J
230 PRINT:PRINT:PRINT
240 NEXT I
280 REM -----
290 REM TRI
300 FOR I=1 TO NC-1
310 FOR J=I+1 TO NC
320 IF P(J)<P(I) THEN 390
330 A=P(J)
340 P(J)=P(I)
350 P(I)=A
360 A=R(J)
370 R(J)=R(I)
380 R(I)=A
390 NEXT J
400 NEXT I
480 REM -----
490 REM RESULTATS
500 CLS:PRINT TAB(13);"PRONOSTICS":PRINT:PRINT
510 PRINT"      PLACE      CHEVAL POINTS"
520 FOR I=1 TO NC
530 IF P(I)=0 THEN 550
540 PRINT TAB(7);I;TAB(15);R(I);TAB(23);P(I)
550 NEXT I
560 PRINT:PRINT
570 PRINT"      VOUS POUVEZ JOUER LE :":PRINT
580 FOR I=1 TO 4
590 FOR J=0 TO 2
600 IF J+I>4 THEN 630
610 PRINT TAB(7+J*8);R(J+I);
620 GOTO 640
630 PRINT TAB(7+J*8);R(J+I-4);
640 NEXT J
650 PRINT
660 NEXT I
670 IF INKEY$<>CHR$(13) THEN 670
680 RUN

```

LE PROGRAMME

Lignes 30 à 240 : saisie des paramètres

Après la saisie du nombre de chevaux, du nombre de pronostics et du nombre de chevaux par pronostic, le programme vous demande les numéros des chevaux, dans l'ordre proposé par vos journaux préférés.

On place alors dans le tableau P les points obtenus, le premier cheval obtenant le maximum (7 points s'il y a 7 chevaux par pronostic).

A l'initialisation, le tableau R contient les numéros des chevaux dans l'ordre de départ ($R(1)=1$, $R(2)=2$, ...).

Lignes 300 à 400 : tri

Cette partie du programme trie les deux tableaux en plaçant les chevaux par ordre croissant des points obtenus.

Le tableau R se trouve alors dans l'ordre d'arrivée (présumé !).

Il s'agit ici d'un tri par la méthode Bubble Sort. Il consiste à comparer le premier élément du tableau P à tous les autres éléments, en effectuant une inversion chaque fois que ceux-ci sont plus petits. Le plus petit élément est ainsi amené en début de table. On effectue la même chose avec le deuxième élément, et ainsi de suite jusqu'au classement complet.

Lignes 500 à 680 : résultats

Le programme affiche à l'écran la liste des chevaux dans l'ordre avec leurs numéros et le nombre de points qu'ils ont obtenus, pour peu que ce dernier soit supérieur à 1.

Les lignes 580 à 660 ne considèrent que les quatre premiers et proposent quatre solutions pour votre tiercé en effectuant des combinaisons.

LISTE DES VARIABLES :

A	utilisée pour le tri des tableaux
CP	nombre de chevaux par pronostic
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de pronostics
NC	nombre de chevaux
NP	numéro du cheval
P()	points de chaque cheval
R()	ordre d'arrivée pronostiqué

LE LOTO : GRILLES ALEATOIRES

Seriez-vous de ces sceptiques qui ne jouent pas au loto parce que l'on n'y gagne jamais ?


Ou seriez-vous au contraire un de ces acharnés qui croient au hasard pur ?

Dans ce cas pourquoi ne pas tenter autre chose que votre date de naissance.

Voilà un programme de loto qui vous imprimera une grille avec un choix de six à dix numéros (pour les amateurs de grilles multiples), juste le temps de feuilleter la suite de votre livre.

	10	20	30	40
1	11	21	31	41
2	12	22	32	42
3	13	23	33	43
4	14	24	34	44
5	15	25	35	45
6	16	26	36	46
7	17	27	37	47
8	18	28	38	48
9	19	29	39	49

LOTO ALEATOIRE



```

10 REM LOTO (ALEATOIRE)
20 REM -----
30 DIM A(49)
40 MODE 1:INK 0,2:INK 1,25:BORDER 2
50 RANDOMIZE TIME
60 INPUT "NOMBRE DE CHIFFRES A TIRER";N
70 IF N<6 OR N>10 THEN 40
80 REM -----
90 REM TIRAGE
100 FOR I=1 TO N
110 A=INT(RND(1)*49)+1
120 IF A(A)=1 THEN 110
130 A(A)=1
140 NEXT I
180 REM -----
190 REM RESULTAT
200 MODE 1:TAG:MOVE 270,399:PRINT "LOTO";
210 MOVE 194,20:DRAW 400,20:DRAW 400,374:DRAW 194,
374:DRAW 194,20
220 FOR I=0 TO 9
230 FOR J=0 TO 40 STEP 10
240 IF I+J=0 THEN 290
250 X=200+J/10*40:Y=340-I*35
260 MOVE X+2,Y+20:PRINT USING"##";I+J;
270 MOVE X,Y:DRAW X+35,Y:DRAW X+35,Y+30:DRAW X,Y+3
0:DRAW X,Y
280 IF A(I+J)=1 THEN MOVE X,Y:DRAW X+35,Y+30:MOVE
X,Y+30:DRAW X+35,Y
290 NEXT J
300 NEXT I
310 IF INKEY$<>CHR$(13) THEN 310
320 RUN

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

Après le dimensionnement du tableau A(), on saisit au clavier le nombre de chiffres que l'on veut obtenir (6 pour les grilles simples, 7 à 10 pour les grilles multiples).

La ligne 70 refuse toutes autres valeurs.

Lignes 100 à 140 : tirage

Les lignes 100 à 140 remplissent le tableau A() avec des chiffres 1 pour les cases qui viennent d'être tirées au hasard. La ligne 120 vérifie le contenu de A() pour s'assurer que le nombre n'a pas déjà été tiré.

Lignes 200 à 320 : résultat

Après le dessin du cadre (ligne 210), on affiche avec deux boucles FOR/NEXT imbriquées les 49 chiffres du loto. Les chiffres qui ont été choisis sont barrés comme sur un bulletin normal.

LISTE DES VARIABLES :

A	nombre aléatoire
A()	liste des chiffres choisis
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de chiffres à tirer
X et Y	coordonnées sur l'écran du chiffre qui va être affiché

LE LOTO : STATISTIQUES

Pourquoi ne pas jouer au Loto "scientifiquement" ?

Il faut pour cela garder tous les tirages depuis la création de ce jeu. En effet, un chiffre tiré 500 fois a beaucoup plus de chance de sortir qu'un chiffre déjà tiré 600 fois.

Lors de la première utilisation, le programme demande les tirages précédents des 49 chiffres. Si vous ne les connaissez pas, vous pouvez les trouver dans une des nombreuses brochures spécialisées sur le Loto.

Cette formalité étant remplie, donnez le tirage de la semaine et le nombre de chiffres dont vous avez besoin (6 pour la formule simple, de 7 à 10 pour les multiples). Après le tri, le bulletin s'affiche, les chiffres qui ont le plus de chance de sortir étant barrés.


Pour finir, AMSTRAD vous invite à sauver ces précieux renseignements pour le prochain tirage sur cassette.

LISTE DES TIRAGES

(1) =	1235	(18) =	1152	(35) =	1284
(2) =	1106	(19) =	1295	(36) =	956
(3) =	1089	(20) =	1104	(37) =	1401
(4) =	1167	(21) =	1046	(38) =	1351
(5) =	1207	(22) =	1074	(39) =	1073
(6) =	1408	(23) =	972	(40) =	1262
(7) =	987	(24) =	983	(41) =	1406
(8) =	1002	(25) =	1035	(42) =	1149
(9) =	867	(26) =	1145	(43) =	1091
(10) =	1011	(27) =	1034	(44) =	1379
(11) =	1237	(28) =	1107	(45) =	1276
(12) =	1138	(29) =	1412	(46) =	1234
(13) =	1156	(30) =	1398	(47) =	1099
(14) =	1076	(31) =	1103	(48) =	1200
(15) =	988	(32) =	1356	(49) =	1385
(16) =	1401	(33) =	1394		
(17) =	1326	(34) =	1105		

APPUYEZ SUR ENTER QUAND LE DATACORDER
EST PRET POUR L'ENREGISTREMENT

LOTO STATISTIQUES



```

10 REM LOTO (STATISTIQUES)
20 REM -----
30 DIM A(49),B(49),C(49),L(49)
40 MODE 1:TAGOFF:INK 0,0:INK 1,15: BORDER 0
60 INPUT"PREMIERE UTILISATION (O/N)";R$
70 IF R$="O" THEN GOSUB 1000:GOTO 200
80 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
90 OPENIN "!LOTO"
100 FOR I=1 TO 49
110 INPUT#9,L(I)
120 NEXT I
130 CLOSEIN
180 REM -----
190 REM LOTO DE LA SEMAINE
200 CLS
210 PRINT"ENTREZ LE LOTO DE LA SEMAINE":PRINT
220 FOR I=1 TO 7
230 IF I=7 THEN PRINT"COMPLEMENTAIRE ";:GOTO 250
240 PRINT"CHIFFRE";I;
250 INPUT C
260 IF C<1 OR C>49 THEN 230
270 L(C)=L(C)+1
280 NEXT I
380 REM -----
390 REM CHOIX DE LA SEMAINE
400 CLS:INPUT"NOMBRE DE CHIFFRES A TIRER ";N
410 IF N<6 OR N>10 THEN 400
420 FOR I=1 TO 49
430 C(I)=L(I):B(I)=I:A(I)=0
440 NEXT I
450 F=0
460 FOR I=1 TO 48
470 IF C(I)<=C(I+1) THEN 510
480 A=C(I):C(I)=C(I+1):C(I+1)=A
490 A=B(I):B(I)=B(I+1):B(I+1)=A
500 F=1
510 NEXT I
520 IF F=1 THEN 450
530 FOR I=1 TO N
540 A(B(I))=1
550 NEXT I
580 REM -----
590 REM RESULTAT
600 MODE 1:TAG:MOVE 270,399:PRINT "LOTO";
610 MOVE 194,20:DRAW 400,20:DRAW 400,374:DRAW 194,
374:DRAW 194,20
620 FOR I=0 TO 9
630 FOR J=0 TO 40 STEP 10
640 IF I+J=0 THEN 690
650 X=200+J/10*40:Y=340-I*35
660 MOVE X+2,Y+20:PRINT USING"##";I+J;
670 MOVE X,Y:DRAW X+35,Y:DRAW X+35,Y+30:DRAW X,Y+30:
DRAW X,Y
680 IF A(I+J)=1 THEN MOVE X,Y:DRAW X+35,Y+30:MOVE
X,Y+30:DRAW X+35,Y
690 NEXT J
700 NEXT I
710 TAGOFF:LOCATE 7,25:INPUT"UNE AUTRE SEMAINE (O/
N) ";R$

```

```
720 IF R$="0" THEN 200
780 REM -----
790 REM SAUVEGARDE
800 CLS:PRINT"LISTE DES TIRAGES"
810 X=1:Y=3
820 FOR I=1 TO 49
830 LOCATE X,Y:PRINT(";I;")=";L(I)
840 Y=Y+1
850 IF I/17=INT(I/17) THEN X=X+13:Y=3
860 NEXT I
870 LOCATE 1,21:PRINT"APPUYEZ SUR ENTER QUAND LE D
ATACORDER EST PRET POUR L'ENREGISTREMENT"
880 LINE INPUT R$
890 OPENOUT "!LOTO"
900 FOR I=1 TO 49
910 PRINT#9,L(I)
920 NEXT I
930 CLOSEOUT
940 CLS:PRINT"A LA SEMAINE PROCHAINE !"
950 END
980 REM -----
990 REM ENTREE DU NOMBRE DE TIRAGES
1000 CLS:INPUT"VOULEZ-VOUS ENTRER LES TIRAGES
DES ANNEES PRECEDENTES (O/N) ";R$
1010 IF R$<>"0" THEN 1070
1020 CLS:PRINT" ENTREZ LES TIRAGES":PRINT
1030 FOR I=1 TO 49
1040 PRINT I;"A DEJA ETE TIRE ";
1050 INPUT L(I)
1060 NEXT I
1070 RETURN
```

LE PROGRAMME

Lignes 30 à 130 : initialisation

S'il s'agit de la première utilisation on exécute le sous-programme en 1000 pour entrer les tirages des années précédentes.

Dans le cas contraire, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 280 : loto de la semaine

Le programme demande les sept chiffres du dernier tirage. On incrémente en conséquence le tableau L().

Lignes 400 à 550 : choix de la semaine

Après la sauvegarde du tableau L() dans le tableau C() (lignes 420 à 440), les lignes 450 à 520 le trient en ordre croissant.

Les lignes 530 à 550 placent dans le tableau A() les N nombres qui ont été choisis, c'est-à-dire ceux qui ont été le moins souvent tirés.

Lignes 600 à 720: résultats

Les lignes 620 à 700 affichent une grille de loto en barrant les chiffres qui ont le plus de chance de sortir.

La ligne 720 renvoie à la saisie si l'on désire entrer le tirage d'une autre semaine avant la sauvegarde.

Lignes 800 à 950 : sauvegarde


Le programme affiche sur l'écran les 49 nombres du loto avec leurs tirages respectifs, puis sauve sur cassette le fichier dès qu'une touche a été appuyée.

Lignes 1000 à 1070 : entrée du nombre de tirages

Ces lignes saisissent les tirages des années précédentes pour chacun des nombres de 1 à 49.

LISTE DES VARIABLES :

A()	liste des nombres choisis
B()	ordre des 49 nombres après le tri
C	chiffres de la semaine
C()	sauvegarde du tableau L() pour le tri
F	drapeau. Si F=1 le tableau n'est pas encore trié.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	liste du nombre de tirages
N	nombre de chiffres à tirer
R\$	réponse du joueur
X et Y	coordonnées du prochain nombre à afficher



THEME 5 LE TEMPS

ARBRE GENEALOGIQUE

Comment déterminer les liens de parenté entre tous les membres d'une famille en vue d'élaborer un arbre généalogique ?

C'est ce que AMSTRAD va faire pour vous.

Trois options vous sont présentées :

1 – SAISIE : vous pouvez entrer le numéro d'une personne suivi de son nom, son sexe (M ou F), puis le numéro d'un autre membre de la famille en relation avec celui que vous désirez créer. Pour ce dernier vous saisirez tout de suite après le type de relation qu'il a avec le personnage en cours de création (E pour enfant, P pour parent, F pour frère et sœur, et M pour un mariage).

Par exemple, on crée la personne X : si Y, référence du lien avec X est l'enfant de X, vous devez donner la lettre E pour enfant.


A la 1^{re} création vous serez obligé de donner un numéro non identifié comme référence, et vous rentrerez ses coordonnées à la deuxième saisie.

2 – RECHERCHE : en tapant le numéro d'une personne, vous obtenez ses liens avec tous les autres membres de la famille.

3 – SAUVEGARDE : sauve le fichier sur cassette.

Pour changer d'option, il suffit de taper "0" comme numéro et l'on revient au menu.

ARBRE GENEALOGIQUE



```

10 REM ARBRE GENEALOGIQUE
20 REM -----
30 MODE 2
40 DIM N$(40), S$(40), P(40,40), T$(40,40), G$(5,8,2),
A$(40)
50 GOSUB 1700
60 INK 0,0:INK 1,15:PAPER 0:PEN 1:BORDER 0
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="O" THEN 300
90 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
100 OPENIN "!ARBRE"
110 INPUT#9,N
120 FOR I=1 TO N
130 INPUT#9,N$(I),S$(I)
140 FOR J=1 TO N
150 INPUT#9,P(J,I),T$(J,I)
160 NEXT J,I
170 CLOSEIN
280 REM -----
290 REM MENU
300 CLS
310 PRINT"1 SAISIE PERSONNE"
320 PRINT"2 RECHERCHE"
330 PRINT"3 SAUVEGARDE"
340 PRINT"VOTRE CHOIX ? ";
350 V=VAL(INKEY$):IF V<1 OR V>3 THEN 350
360 ON V GOTO 400,700,1300
380 REM -----
390 REM SAISIE
400 GOSUB 1500
410 PRINT:INPUT"NUMERO DE SAISIE ";NU
420 IF NU=0 THEN 300
430 IF NU>N+1 THEN 410
440 IF NU>40 THEN PRINT"FICHER PLEIN":GOTO 410
450 IF N$(NU)<>" " THEN PRINT N$(NU):GOTO 500
460 INPUT"NOM";N$(NU)
470 IF N$(NU)=" " THEN 400
480 INPUT"SEXE";S$(NU)
490 IF S$(NU)<>"M" AND S$(NU)<>"F" THEN 400
500 I=NU:N=N+1
510 INPUT"LIAISON AVEC ";L:IF L=0 THEN 400
520 IF P(L,NU)=L THEN PRINT"JE LE SAIS DEJA":GOTO
510
530 P(L,NU)=L
540 INPUT"TYPE (E,P,F,M)";T$(L,NU)
550 IF T$(L,NU)<>"E" AND T$(L,NU)<>"P" AND T$(L,NU)
)<>"F" AND T$(L,NU)<>"M" THEN 540
560 P(NU,L)=NU:T$(NU,L)=T$(L,NU)
570 IF T$(NU,L)="E" THEN T$(NU,L)="P":GOTO 590
580 IF T$(NU,L)="P" THEN T$(NU,L)="E"
590 GOTO 400
680 REM -----
690 REM RECHERCHE
700 GOSUB 1500
710 PRINT:INPUT"NUMERO DE RECHERCHE ";NU
720 IF NU=0 OR N$(NU)=" " THEN 300
730 FOR I=1 TO N:A$(I)=" ":NEXT
740 CLS:PRINT N$(NU); " A POUR: "
750 FOR I=1 TO N

```

```

760 IF T$(I,NU)="" THEN 800
770 A$(I)=T$(I,NU)
780 IF T$(I,NU)="E" THEN A$(P(I,NU))="P":GOTO 800
790 IF T$(I,NU)="P" THEN A$(P(I,NU))="E"
800 NEXT I
810 FOR K=1 TO N
820 FOR I=1 TO N
830 IF I=NU OR N$(I)="" OR A$(I)<>" THEN 880
840 FOR J=1 TO N
850 IF A$(J)="" THEN 870
860 IF P(J,I)=J THEN A$(I)=T$(J,I)+A$(J):GOTO 880
870 NEXT J
880 NEXT I
890 NEXT K
900 FOR I=1 TO N
910 IF A$(I)="" THEN 1230
920 FOR J=1 TO LEN(A$(I))
930 IF MID$(A$(I),J,2)="FF" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1):IF J>1 THEN J=J-2:GOTO 1000 ELSE J=J-1:GOTO 1000
940 IF MID$(A$(I),J,3)="EFE" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
950 IF MID$(A$(I),J,3)="PFP" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
960 IF MID$(A$(I),J,2)="FE" THEN A$(I)=LEFT$(A$(I),J-1)+RIGHT$(A$(I),LEN(A$(I))-J)
970 IF MID$(A$(I),J,2)="EM" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
980 IF MID$(A$(I),J,2)="PF" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1):J=J-1:GOTO 1000
990 IF MID$(A$(I),J,2)="MP" THEN A$(I)=LEFT$(A$(I),J-1)+RIGHT$(A$(I),LEN(A$(I))-J):J=J-1
1000 NEXT J
1010 V=0:F=0:M=0:C=0
1020 FOR J=1 TO LEN(A$(I))
1030 IF MID$(A$(I),J,1)="E" THEN V=V-1
1040 IF MID$(A$(I),J,1)="P" THEN V=V+1
1050 IF MID$(A$(I),J,2)="EF" THEN F=1
1060 IF MID$(A$(I),J,2)="FP" THEN F=1
1070 IF MID$(A$(I),J,2)="MF" THEN F=1
1080 IF MID$(A$(I),J,2)="FM" THEN F=1
1090 IF MID$(A$(I),J,1)="M" THEN M=2
1100 IF MID$(A$(I),J,2)="PE" AND MID$( " "+A$(I),J,1)<>"E" AND MID$(A$(I)+" ",J+2,1)<>"P" THEN M=2
1110 IF MID$(A$(I),J,2)="EP" AND MID$( " "+A$(I),J,1)<>"P" AND MID$(A$(I)+" ",J+2,1)<>"E" THEN F=1
1120 IF MID$(A$(I),J,3)="EFP" THEN C=3
1130 NEXT J
1140 IF C=3 AND M=2 THEN C=0
1150 IF S$(I)="M" THEN S=1 ELSE S=2
1160 IF V=0 THEN PRINT G$(1+F+M+C,4,S);:GOTO 1220
1170 FOR J=V TO SGN(V) STEP -SGN(V)
1180 IF J<>V THEN PRINT "-";
1190 IF J+4<1 OR J+4>7 THEN PRINT"ARRIERE";:GOTO 1210
1200 PRINT G$(1+F+M+C,J+4,S);
1210 NEXT J
1220 PRINT " ";N$(I)
1230 NEXT I
1240 IF INKEY$=CHR$(13) THEN 700 ELSE 1240

```

```

1280 REM -----
1290 REM SAUVEGARDE
1300 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORD
ER EST PRET POUR L'ENREGISTREMENT"
1310 LINE INPUT R$
1320 OPENOUT "!ARBRE"
1330 PRINT#9,N
1340 FOR I=1 TO N
1350 PRINT#9,N$(I);", ";S$(I)
1360 FOR J=1 TO N
1370 PRINT#9,P(J,I);", ";T$(J,I)
1380 NEXT J,I
1390 CLOSEOUT
1400 CLS:PRINT"A LA PROCHAINE NAISSANCE !"
1410 END
1480 REM -----
1490 REM AFFICHE LES NOMS
1500 CLS
1510 FOR I=1 TO N
1520 IF N$(I)="" THEN 1630
1530 PRINT "(";I;" " ";N$(I);" ";
1540 FOR J=1 TO N
1550 IF T$(J,I)<>"" THEN 1570
1560 NEXT J:GOTO 1630
1570 M=0
1580 IF S$(I)="M" THEN S=1 ELSE S=2
1590 IF T$(J,I)="P" THEN V=1:GOTO 1620
1600 IF T$(J,I)="E" THEN V=-1:GOTO 1620
1610 IF T$(J,I)="F" THEN V=0 ELSE V=0:M=2
1620 PRINT G$(1+M,V+4,S);" DE ";N$(J)
1630 NEXT I
1640 RETURN
1680 REM -----
1690 REM DONNEES GENEALOGIQUES
1700 FOR K=1 TO 5
1710 FOR I=1 TO 7
1720 READ G$(K,I,1),G$(K,I,2)
1730 NEXT I
1740 NEXT K
1750 RETURN
1760 DATA ARRIERE,ARRIERE,PETIT,PETITE,FILS,FILLE,
FRERE,SOEUR
1770 DATA PERE,MERE,GRAND,GRAND,ARRIERE,ARRIERE
1780 DATA ARRIERE,ARRIERE,PETIT,PETITE,NEVEU,NIECE
,FRERE,SOEUR
1790 DATA ONCLE,TANTE,GRAND,GRAND,ARRIERE,ARRIERE
1800 DATA ARRIERE,ARRIERE,PETIT,PETITE,GENDRE,BRU,
EPOUX,EPOUSE
1810 DATA BEAU-PERE,BELLE-MERE,GRAND,GRAND,ARRIERE
,ARRIERE
1820 DATA ARRIERE,ARRIERE,PETIT,PETITE,NEVEU (PAR
ALLIANCE),NIECE (PAR ALLIANCE),BEAU-FRERE,BELLE SO
EUR
1830 DATA ONCLE (PAR ALLIANCE),TANTE (PAR ALLIANCE
),GRAND,GRAND,ARRIERE,ARRIERE
1840 DATA ARRIERE,ARRIERE,ARRIERE,ARRIERE,PETIT-CO
USIN,PETITE-COUSINE,COUSIN,COUSINE
1850 DATA GRAND-COUSIN,GRANDE-COUSINE,ARRIERE,ARRI
ERE,ARRIERE,ARRIERE

```

LE PROGRAMME

Lignes 30 à 170 : initialisation

On fait appel au sous-programme en 1700 pour placer dans le tableau G\$() toutes les données généalogiques (père, mère, frère, sœur, fils, fille, etc.).

Les tableaux N\$(), S\$(), etc., (ligne 40) correspondent aux noms, sexes, etc., des personnes dans le fichier. Si vous trouvez que ce n'est pas assez, il est possible de modifier leurs indices maximum (ici 40) par une autre valeur supérieure. Il faut pour cela modifier tous les chiffres 40 de la ligne 40 et celui de la ligne 440. Donnez des valeurs raisonnables, sinon il ne restera plus de place en mémoire pour exécuter le programme. Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 300 à 360 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur (variable V).

Lignes 400 à 590 : saisie

On fait d'abord appel au sous-programme en 1500 (affiche les noms et les liaisons avec d'autres personnes).

Puis le programme demande le numéro que l'on veut créer ou modifier. Si on répond 0, la ligne 420 renvoie au menu. Dans le cas contraire, les lignes 460 et 480 saisissent le nom et le sexe.

Les lignes 510 à 590 saisissent les liaisons avec d'autres personnes. On vérifie en ligne 520 si le renseignement donné n'est pas déjà dans le fichier. Par exemple si l'on dit : X est l'enfant de Y, et que l'on sait déjà que Y est le parent de X on refuse l'entrée.

Ces lignes créent aussi toutes les liaisons inverses. Par exemple : si X est l'enfant de Y, on crée dans l'enregistrement de Y la liaison Y est le parent X (c'est le tableau S\$() – le sexe – qui détermine s'il s'agit du fils ou de la fille, du père ou de la mère).

Lignes 700 à 1240 : recherches

Après l'appel du sous-programme en 1500, on demande le numéro du nom que l'on veut rechercher. La boucle FOR/NEXT des lignes 750 à 800 place dans le tableau A\$(,) toutes les liaisons directes avec le nom à rechercher (frère, père, etc.). Les lignes 810 à 890 placent dans le tableau A\$(,) les liaisons indirectes en utilisant les liaisons directes déjà dans le tableau (le père du père, le frère du père, etc.).

Ce sont les lignes 900 à 1230 qui déterminent les liaisons finales de la personne recherchée. La variable V donne le niveau dans le tableau G\$(,), la variable F un lien entre frères, la variable M un mariage et la variable C concerne les cousins.

Les lignes 1170 à 1220 affichent les liaisons complètes. Si V=2 par exemple, on affiche à la suite les mots "grand" puis "père". Si V=3, les mots "arrière", "grand" puis "père". Si V=-2, les mots "petit" et "fils", et ainsi de suite. Les variables F, M et C font passer à d'autres niveaux dans le tableau G\$(,). Si F=1 et V=-1 par exemple, on affiche le mot "neveu" à la place de "fils" (c'est le fils du frère). La combinaison des variables F, M et C permet d'utiliser toutes les solutions du tableau G\$(,).

Lignes 1300 à 1410 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

Lignes 1500 à 1640 : affichent les noms

Ce sous-programme affiche tous les noms et les liaisons correspondantes.

LISTE DES VARIABLES :

A\$()	liste des noms des liaisons (enfant, parent, etc.), pour la recherche
C	C=3 si on a trouvé une liaison entre cousins
F	F=1 si on a trouvé une liaison entre frères
G\$(),	liste des données généalogiques
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
L	dernière liaison saisie
M	M=2 si on a trouvé une liaison entre époux
N	nombre de personnes dans le fichier
NU	numéro de la personne à rechercher
N\$()	liste des noms
P	P=1 si toutes les liaisons avec NU n'ont pas été trouvées
P(),	liste des liaisons avec les autres personnes (numéros)
R\$	réponse de l'utilisateur
S	S=1 pour masculin, S=2 pour féminin
S\$()	liste des sexes
T\$(),	liste des liaisons avec les autres personnes (enfant, parent, etc.).
V	niveau généalogique

```
( 1 ) DUPONT JEAN PERE DE DUPONT ANAIS  
( 2 ) DUPONT ANAIS FILLE DE DUPONT JEAN  
( 3 ) DUPONT JACQUES FRERE DE DUPONT ANAIS  
( 4 ) DUPONT JEANNE EPOUSE DE DUPONT JEAN  
( 5 ) DURANT MICHEL EPoux DE DUPONT ANAIS  
( 6 ) DURANT LEON FILS DE DURANT MICHEL
```

NUMERO DE RECHERCHE : 6

```
DURANT LEON A POUR:  
GRAND-PERE DUPONT JEAN  
MERE DUPONT ANAIS  
ONCLE DUPONT JACQUES  
GRAND-MERE DUPONT JEANNE  
PERE DURANT MICHEL
```


ASTROLOGIE

Votre AMSTRAD va vous permettre de mieux étudier votre entourage grâce à son programme d'astrologie occidentale.


Dès que vous aurez rentré la date de naissance d'un proche (JJ, MM), il vous affichera son signe zodiacal.

Il vous exposera aussi en quelques lignes les traits généraux qui caractérisent les natifs de ce signe.

A propos, de quel signe est votre AMSTRAD ?



ASTROLOGIE



```

10 REM ASTROLOGIE
20 REM -----
30 DIM J(12),S$(12):GOSUB 600
40 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1:BORDER 1
50 INPUT"JOUR ET MOIS DE NAISSANCE";J,M
60 IF M=1 THEN 80
70 FOR I=1 TO M-1:N=N+J(I):NEXT I
80 N=N+J
180 REM -----
190 REM CALCUL DU SIGNE
200 FOR I=1 TO 13
210 S=S+1
220 READ P:C=C+P
230 IF N<C THEN S=S-1:GOTO 260
240 NEXT I
250 STOP
260 IF S=0 THEN S=12
380 REM -----
390 REM SIGNE ASTROLOGIQUE
400 CLS:PRINT TAB(15)"ASTROLOGIE":PRINT:PRINT:PRINT
410 PRINT" VOUS ETES DU SIGNE ";
420 A$=RIGHT$(S$(S),2):IF A$="NS" OR A$="UX" THEN
PRINT"DES ";:GOTO 440
430 IF A$="CE" OR A$="GE" THEN PRINT"DE LA "; ELSE
PRINT"DU ";
440 PRINT S$(S)
450 PRINT
460 PRINT:RESTORE 660
470 IF A$="NS" OR A$="UX" THEN PRINT" LES ";S$(S);
" SONT:":GOTO 490
480 IF A$="CE" OR A$="GE" THEN PRINT" LA ";S$(S);"
EST:" ELSE PRINT"LE ";S$(S);" EST:"
490 FOR I=1 TO S:READ A$:NEXT I
500 PRINT:PRINT A$
510 IF INKEY$=CHR$(13) THEN RUN ELSE 510
580 REM -----
590 REM DONNEES
600 FOR I=1 TO 12:READ J(I):NEXT I
610 DATA 31,28,31,30,31,30,31,31,30,31,30,31
620 FOR I=1 TO 12:READ S$(I):NEXT I
630 DATA VERSEAU,POISSONS,BELIER,TAUREAU,GEMEAUX,C
ANCER,LION,VIERGE,BALANCE,SCORPION,SAGITTAIRE,CAPR
ICORNE
640 RETURN
650 DATA 20,30,30,30,31,31,32,31,31,30,30,30,29
660 DATA "SIGNE D'AIR. ALTRUISTE ET SENSIBLE, IL C
HERCHE A SE REALISER DANS LE DOMAINE MORAL OU SOCI
AL. INDEPENDANT, SOUVENT ANTICONFORMISTE, MAIS TOU
JOURS LUCIDE, IL EST A LA RECHERCHE D'UN ABSOLU DA
NS SA VIE."
670 DATA "SIGNE D'EAU. SECRET ET IMAGINATIF, IL SE
VEUT PARTIE INTEGRANTE DE L'UNIVERS. IL EST EMOTI
F, IMPRESSIONNABLE, SENSIBLE A TOUT CE QUI TOUCHE
LES AUTRES. SA GENEROSITE L'ENTRAINE VERS LES GRAN
DES CAUSES HUMANITAIRES."
680 DATA "SIGNE DE FEU. IL EST LE TYPE MEME DE L'H
OMME OU DE LA FEMME D'ACTION. PLEIN DE VITALITE ET
DE TONUS, IL AIME SE DEPENSER PHYSIQUEMENT ET SE
SURPASSER DANS L'EFFORT."

```

690 DATA "SIGNE DE TERRE. IL POSSEDE UNE PERSONNALITE PROFONDE ET PASSIONNEE. C'EST UN ETRE CHARNEL ET SENSUEL. IL FAIT PREUVE D'UNE ACTIVITE TENACE QUI LE Pousse A ALLER JUSQU'AU BOUT DE CE QU'IL ENT REPREND."

700 DATA "SIGNE D'AIR. LE NATIF DE CE SIGNE POSSEDE UNE GRANDE VIVACITE DE TEMPERAMENT QUI EN FAIT UN ETRE DOUE. SA PERSONNALITE MULTIPLE (SIGNE DOUBLE) LUI DONNE UNE GRANDE FACULTE D'ADAPTATION ET BEAUCOUP DE RAPIDITE DANS LE TRAVAIL."

710 DATA "SIGNE D'EAU, IL EST DOTE D'UNE PERSONNALITE TRANQUILLE POSSEDANT LE CALME DE L'ESPRIT, INCHARME SUBLIL ET UNE GRANDE DOUCEUR DE TEMPERAMENT. IL AIME FLANER, REVER, SE REFUGIER DANS L'IMAGINAIRE."

720 DATA "SIGNE DE FEU. D'UN TEMPERAMENT MAGNIFIQUE ET GENEREUX, IL DETESTE PASSER INAPERCU ET POSSEDE LE GOUT DE SE REALISER PLEINEMENT DANS SA VIE SOCIALE. VOLONTAIRE, C'EST L'HOMME OU LA FEMME DES GRANDES ENTREPRISES."

730 DATA "SIGNE DE TERRE. DE NATURE METHODIQUE ET METICULEUSE, LE NATIF DE LA VIERGE RECHERCHE LE CALME DE L'ESPRIT ET LA SECURITE. IL FAIT CEPENDANT PREUVE DE CURIOSITE ET DE VIVACITE D'ESPRIT EN PROFONDEUR."

740 DATA "SIGNE D'AIR. OUVERT, SENSIBLE, AFFECTUEUX, LE NATIF DE LA BALANCE EST UN ETRE A LA RECHERCHE DE SON EQUILIBRE. IL SE CARACTERISE PAR SON CHARME, SA SUBLILITE ET SON SENS INTUITIF."

750 DATA "SIGNE D'EAU. C'EST UN PASSIONNE COMBATIF. ALTRUISTE ET TRAVAILLEUR ACHARNE, LE NATIF DE CE SIGNE POSSEDE UNE RESISTANCE PHYSIQUE CONSIDERABLE."

760 DATA "SIGNE DE FEU. LE SAGITTAIRE POSSEDE UNE PERSONNALITE BRILLANTE SOUTENUE PAR UN PROFOND EQUILIBRE ET UN TRES GRAND SERIEUX. IL EPROUVE LE BESOIN DE FRANCHIR SES LIMITES TANT MATERIELLES QUE SPIRITUELLES."

770 DATA "SIGNE DE TERRE. C'EST UN FLEGMATIQUE OU UN PASSIONNE A FROID. REFLECHI ET METHODIQUE, IL EST PROFOND, GRAVE ET DISCIPLINE. UNE FOIS SA DECISION PRISE, IL EST DIFFICILE DE LE FAIRE CHANGER D'AVIS."

LE PROGRAMME

Lignes 30 à 80 : initialisation

Après l'appel du sous-programme en 600 qui place dans le tableau J() le nombre de jours par mois et dans le tableau S\$() les douze signes du zodiaque, le programme saisit le jour et le mois de naissance et calcule le nombre de jours vécus depuis le 1^{er} janvier.

Lignes 200 à 260 : calcul du signe

Les DATAs de la ligne 650 donnent le nombre de jours d'écart entre chaque signe. La boucle FOR/NEXT des lignes 200 à 240 compare le total de ces jours avec ceux du nombre de jours vécus (variable N) pour déterminer le signe du zodiaque (variable S).

Lignes 400 à 510 : signe astrologique

Ces lignes affichent le signe du zodiaque pointé par S, ainsi que le commentaire correspondant (DATAs des lignes 660 à 770).

Les lignes 420, 430, 470 et 480 rétablissent l'orthographe s'il s'agit d'un signe au pluriel ou au féminin.

LISTE DES VARIABLES :

A\$	deux dernières lettres du signe
C	total du nombre de jours d'écart entre les signes
I	indice de boucle FOR/NEXT
J	jour de naissance
J()	nombre de jours par mois
M	mois de naissance
N	nombre de jours vécus depuis le 1 ^{er} janvier
P	dernier écart entre deux signes
S	numéro du signe
S\$()	liste des douze signes du zodiaque

ASTROLOGIE CHINOISE


Pour approfondir la connaissance des gens qui vous entourent, voilà un programme d'astrologie chinoise qui complète celui d'astrologie occidentale.

La saisie d'une date de naissance par votre AMSTRAD lui fait calculer le signe chinois de cette année et l'élément qui la caractérise.

En effet, le signe astrologique dépend de l'année de naissance (l'année chinoise bien sûr, différente par sa durée d'une année occidentale). Ce signe est complété par un élément associé au yin et au yang (année positive et négative). Ainsi, 1983 était l'année du cochon d'eau vive (du 13 février 83 au 2 février 84).

Maintenant vous saurez tout sur votre AMSTRAD !





```

10 REM ASTROLOGIE CHINOISE
20 REM -----
30 DIM J(12), S$(12):GOSUB 700
40 MODE 1:INK 0,1:INK 1,24:PAPER 0:PEN 1: BORDER 1
50 INPUT"JOUR,MOIS,ANNEE DE NAISSANCE";J,M,A
60 IF A>1899 THEN A=A-1900
70 IF A<0 OR A>99 THEN 40
80 N=A*365+INT(ABS((A-1)/4))
90 IF A/4=INT(A/4) AND A<>0 THEN J(2)=29 ELSE J(2)
=28
100 IF M=1 THEN 120
110 FOR I=1 TO M-1:N=N+J(I):NEXT I
120 N=N+J
180 REM -----
190 REM CALCUL DU SIGNE
200 FOR I=1 TO 101
210 S=S+1
220 READ P:C=C+T(P)
230 IF N<C THEN S=S-1:GOTO 260
240 NEXT I
250 STOP
260 E=S:S=S MOD 12:E=E MOD 10
270 IF S=0 THEN S=12
280 IF E=0 THEN E=10
380 REM -----
390 REM SIGNE ASTROLOGIQUE
400 CLS:PRINT TAB(11)"ASTROLOGIE CHINOISE":PRINT:P
RINT
410 PRINT" VOUS ETES DU SIGNE DU ";S$(S)
420 PRINT
430 PRINT A+1900;"EST UNE ANNEE ";E$(E);" ";Y$(E)
440 PRINT
450 PRINT" VOUS ETES DONC UN ";S$(S);" ";E$(E)
460 PRINT:RESTORE 890
470 PRINT" LE ";S$(S);" EST:"
480 FOR I=1 TO (S-1)*10:READ A$:NEXT I
490 PRINT:PRINT" - ";
500 FOR I=1 TO 5
510 READ A$:PRINT A$;" - ";
520 NEXT I
530 PRINT:PRINT:PRINT" MAIS AUSSI:":PRINT:PRINT" -
";
540 FOR I=1 TO 5
550 READ A$:PRINT A$;" - ";
560 NEXT I
570 PRINT:PRINT
580 IF INKEY$=CHR$(13) THEN RUN ELSE 580
680 REM -----
690 REM DONNEES
700 FOR I=1 TO 12:READ J(I):NEXT I
710 DATA 31,28,31,30,31,30,31,31,30,31,30,31
720 FOR I=1 TO 12:READ S$(I):NEXT I
730 DATA RAT,BOEUF,TIGRE,CHAT,DRAGON,SERPENT,CHEVA
L,BOUC,SINGE,COQ,CHIEN,COCHON
740 FOR I=1 TO 10 STEP 2:READ E$(I):E$(I+1)=E$(I):
NEXT I
750 DATA DE METAL,D'EAU,DE BOIS,DE FEU,DE TERRE
760 FOR I=1 TO 10:READ Y$(I):NEXT I
770 DATA FROID,CHAUD,NATURELLE,VIVE,,ALLUME,LATENT
,ALLUME,INCULTE,CULTIVEE

```

780 T(0)=353:T(1)=354:T(2)=355:T(3)=383:T(4)=384:T
 (5)=385:T(6)=31
 790 RETURN
 800 DATA 6,4,1,2,3,1,2,4,1,2,4,1
 810 DATA 4,1,1,4,1,2,4,2,4,1,1,4
 820 DATA 1,2,4,1,2,4,1,3,1,2,4,2
 830 DATA 1,4,1,4,1,1,4,2,1,5,1,1
 840 DATA 4,1,4,1,2,4,1,2,4,1,3,2
 850 DATA 1,4,2,1,4,2,0,4,2,4,1,2
 860 DATA 4,1,1,4,1,4,1,2,4,2,1,4
 870 DATA 1,4,1,1,4,2,2,4,1,1,3,2
 880 DATA 4,1,2,4,1
 890 DATA ENERGIQUE, METICULEUX, SOCIABLE, HUMORISTE, S
 ENTIMENTAL
 900 DATA AGITE, JOUEUR, INQUIET, ENNUYEUX, SOUPCONNEUX
 910 DATA TRAVAILLEUR, FIER, ELOQUENT, ORIGINAL, TENACE
 920 DATA LENT, AUTORITAIRE, VINDICATIF, JALOUX, LOURDA
 UD
 930 DATA GENEREUX, COURAGEUX, SENSIBLE, PROFOND, ACTIF
 940 DATA VANITEUX, QUERELLEUR, TEMERAIRE, EMPORTE, ENT
 ETE
 950 DATA DISCRET, RAFFINE, VERTUEUX, AMBITIEUX, INGENI
 EUX
 960 DATA PEDANT, SUSCEPTIBLE, DISTANT, SECRET, TORTUEU
 X
 970 DATA SCRUPULEUX, PERSPICACE, CHANCEUX, AUTONOME, I
 NFLUENT
 980 DATA INFLEXIBLE, EXIGEANT, IRRITABLE, INFATUE, INS
 ATISFAIT
 990 DATA SAGE, CULTIVE, INTUITIF, ELEGANT, PHILOSOPHE
 1000 DATA PRESOMPTUEUX, POSSESSIF, FLEGMATIQUE, EXTRA
 VAGANT, VOLAGE
 1010 DATA AIMABLE, HABILE, INDEPENDANT, SENSUEL, ENJOU
 E
 1020 DATA EGOISTE, FAIBLE, IMPATIENT, INSENSIBLE, IMPI
 TOYABLE
 1030 DATA CREATIF, INTELLIGENT, DOUX, SIMPLE, ALTRUIST
 E
 1040 DATA CAPRICIEUX, INEXACT, INDISCRET, DEPENDANT, I
 NDISCIPLINE
 1050 DATA SPIRITUEL, SUBTIL, ADROIT, PASSIONNE, FASCIN
 ANT
 1060 DATA RUSE, INFIDELE, MALHONNETE, VANITEUX, DISSIM
 ULATEUR
 1070 DATA FRANC, SINCERE, AMUSANT, ENTHOUSIASTE, AVENT
 UREUX
 1080 DATA MANIAQUE, MEFIANT, DEPENSIER, VANTARD, ACERB
 E
 1090 DATA LOYAL, ATTENTIF, LUCIDE, FIDELE, MODESTE
 1100 DATA CRITIQUE, PESSIMISTE, MORALISATEUR, RESERVE
 , CYNIQUE
 1110 DATA INDULGENT, CONSCIENCIEUX, DECISIF, PACIFIQU
 E, AIMANT
 1120 DATA NAIF, CREDULE, ENTETE, DESARME, TERRE A TERR
 E

LE PROGRAMME

Lignes 30 à 120 : initialisation

Après l'appel du sous-programme en 700 qui place dans le tableau J() le nombre de jours par mois, dans S\$() les douze signes, dans E\$() les 5 éléments et dans Y\$() le yin et le yang, le programme saisit le jour, le mois et l'année de naissance et calcule le nombre de jours vécus depuis le 1^{er} janvier 1900.

Lignes 200 à 280 : calcul du signe

Les DATAs des lignes 800 à 880 donnent le nombre de jours de chaque année chinoise (totalement différent des années occidentales de 365 ou 366 jours) en se référant au tableau T() pour ne pas surcharger les données dans le programme. La boucle FOR/NEXT des lignes 200 à 240 compare le total de ces jours avec ceux du nombre de jours vécus (variable N) pour déterminer le signe chinois (variable S). Les signes se répétant tous les douze ans, on ne prend que le modulo 12 de S (modulo = reste de la division). Il en est de même pour les éléments (variable E) qui se répètent tous les dix ans.

Lignes 400 à 580 : signe astrologique

Ces lignes affichent le signe chinois pointé par S, l'élément pointé par E avec le yin et le yang correspondant (yang = positif, yin = négatif), ainsi que des commentaires sur le signe (DATAs des lignes 890 à 1120).

LISTE DES VARIABLES :

A	année de naissance
A\$	commentaires sur le signe à afficher
C	total du nombre de jours des années chinoises
E\$()	liste des cinq éléments
I	indice de boucle FOR/NEXT
J	jour de naissance
J()	nombre de jours par mois
M	mois de naissance
N	nombre de jours vécus depuis le 1 ^{er} janvier 1900
P	nombre de jours dans l'année chinoise (pointe le tableau T)
S	numéro du signe
S\$()	liste des douze signes chinois
T()	nombre de jours possibles dans une année chinoise
Y\$()	liste des yin et des yang

ESPERANCE DE VIE

Vous ne le savez peut-être pas, mais les voyantes modernes ne vous répondent plus à la légère quand vous leur demandez combien de temps vous vivrez.

Elles calculent consciencieusement votre espérance de vie grâce à un ordinateur discrètement relié à leur boule de cristal.

Lequel ordinateur utilise le test très sérieux qui vous est proposé dans les pages suivantes. Il a été mis au point par un médecin anglais d'après les renseignements d'ordre privé demandés par les assureurs-vie à leurs clients.

Ce test est basé sur le principe d'une espérance de vie correspondant à chaque âge (71 ans si l'on a 30 ans, 73 ans si l'on a 50 ans...). Mais des critères négatifs (abus du tabac, de l'alcool...) ou positifs (bon sommeil, longévité des parents) peuvent rallonger ou raccourcir cette espérance de vie de plusieurs années. Bien sûr, cela reste du domaine des statistiques, chaque histoire humaine étant unique et imprévisible ! Mais on peut méditer...


ESPERANCE DE VIE


QUESTION 4

1-VOUS AVEZ UN PARENT DECEDE D'UNE
CRISE CARDIAQUE AVANT 50 ANS
2-D'UNE EMBOLIE OU ASTERIOSCLEROSE
AVANT 50 ANS
3-DECEDE ENTRE 50 ET 60 ANS
4-DECEDE DE DIABETE AVANT 60 ANS
5-DECEDE DE CANCER DE L'ESTOMAC AVANT
60 ANS
6-AUCUN

REPONDEZ PAR UN CHIFFRE ENTRE 1 ET 6 ?

ESPERANCE DE VIE





```

10 REM ESPERANCE DE VIE
20 REM -----
30 CLS:PRINT TAB(12);"ESPERANCE DE VIE":PRINT:PRINT
40 INPUT"VOTRE AGE";AGE
50 IF AGE<15 OR AGE>80 THEN 30
60 IF AGE<41 THEN E=70+(AGE-15)*(1.8/25)
70 IF AGE>40.AND AGE<60 THEN E=72.6+(AGE-41)*(3.4/
18)
80 IF AGE>59 THEN E=76.3+(AGE-60)*(9.4/20)
180 REM -----
190 REM INTERPRETATION DES QUESTIONS
200 FOR I=1 TO 14
210 CLS:PRINT TAB(12);"ESPERANCE DE VIE"
220 READ N
230 FOR J=1 TO N
240 READ A(J),Q$(J)
250 NEXT J
260 PRINT:PRINT"QUESTION ";I:PRINT
270 FOR J=1 TO N
280 A$=STR$(J)+"-"+Q$(J):PRINT A$
290 NEXT J
300 IF N>1 THEN 360
310 PRINT:PRINT"REPONDEZ PAR OUI OU NON ?"
320 R$=INKEY$
330 IF R$<>"0" AND R$<>"N" THEN 320
340 IF R$="0" THEN E=E+A(1)
350 GOTO 400
360 PRINT:PRINT"REPONDEZ PAR UN CHIFFRE ENTRE 1 ET
";N;"?"
370 R$=INKEY$
380 R=VAL(R$):IF R<1 OR R>N THEN 370
390 E=E+A(R)
400 NEXT I
480 REM -----
490 REM RESULTATS
500 CLS:PRINT TAB(12);"ESPERANCE DE VIE":PRINT:PRINT
510 PRINT TAB(8);"VOTRE ESPERANCE DE VIE EST":PRINT
520 PRINT TAB(16);"DE";INT(E);"ANS"
530 IF INKEY$<>CHR$(13) THEN 530
540 RUN
580 REM -----
590 REM DONNEES DES QUESTIONS
600 DATA 1,3,"ETES-VOUS UNE FEMME"
610 DATA 1,1,"UN DE VOS GRANDS-PARENTS A VECU
80 ANS OU PLUS"
620 DATA 4,4,"VOTRE MERE A VECU PLUS DE 80 ANS"
630 DATA 2,"VOTRE PERE A VECU PLUS DE 80 ANS"
640 DATA 6,"LES DEUX",0,"AUCUN"
650 DATA 6,-4
660 DATA "VOUS AVEZ UN PARENT DECEDE D'UNE
CRISE CARDIAQUE AVANT 50 ANS"
670 DATA -4,"D'UNE EMBOLIE OU ASTERIOSCLEROSE
AVANT 50 ANS"
680 DATA -2,"DECEDE ENTRE 50 ET 60 ANS"
690 DATA -3,"DECEDE DE DIABETE AVANT 60 ANS"
700 DATA -2,"DECEDE DE CANCER DE L'ESTOMAC AVANT
60 ANS"

```

710 DATA 0, "AUCUN"
720 DATA 4, -9, "VOUS FUMEZ PLUS DE 40 CIGARETTES PA
R JOUR"
730 DATA -7, "ENTRE 20 ET 40 CIGARETTES"
740 DATA -2, "MOINS DE 20 CIGARETTES"
750 DATA 0, "VOUS NE FUMEZ PAS"
760 DATA 1, 2
770 DATA "VOUS AVEZ DES RELATIONS SEXUELLES 1
OU 2 FOIS PAR SEMAINE"
780 DATA 1, 2, "VOUS EFFECTUEZ UN EXAMEN MEDICAL
COMPLET TOUS LES ANS"
790 DATA 1, -2, "VOUS DEPASSEZ VOTRE POIDS NORMAL
(VOIR PROGRAMME 'POIDS IDEAL')"
800 DATA 1, -2, "VOUS DORMEZ PLUS DE 10 HEURES PAR
NUIT OU MOINS DE 5 HEURES"
810 DATA 3, 3
820 DATA "VOUS BUVEZ MODEREMENT (1/2 LITRE DE
VIN OU 4 VERRS DE BIERE)"
830 DATA 0, "VOUS NE BUVEZ PAS"
840 DATA -8, "VOUS BUVEZ BEAUCOUP"
850 DATA 1, 3, "VOUS PRENEZ DE L'EXERCICE TROIS FOIS
PAR SEMAINE"
860 DATA 3, 3, "VOUS ETES EX-UNIVERSITAIRE"
870 DATA 2, "VOUS ETES BACHELIER"
880 DATA 0, "NI L'UN NI L'AUTRE"
890 DATA 2, 1, "VOUS ETES CITADIN"
900 DATA -1, "VOUS ETES CAMPAGNARD"
910 DATA 5, 1, "VOUS ETES MARIE OU VIVEZ
MARITALEMENT"
920 DATA -9, "VOUS ETES SOLITAIRE"
930 DATA -7, "VOUS ETES VEUF"
940 DATA -4, "FEMME SEPARÉE OU DIVORCÉE"
950 DATA -3, "FEMME VEUVE"

LE PROGRAMME

Lignes 30 à 80 : initialisation

L'espérance de vie initiale (variable E) est calculée suivant une courbe statistique pour des âges allant de 15 à 80 ans. Pour suivre le plus possible cette courbe, on effectue un découpage en trois tranches : de 15 à 40 ans, de 41 à 59 ans et de 60 à 80 ans.

Lignes 200 à 400 : interprétation des questions

Toutes les questions sont placées dans les DATAs des lignes 600 à 950. La première donne le nombre de réponses possibles (variable N), et les N DATAs qui suivent les questions et le nombre d'années à retrancher ou à ajouter.

Dans le cas d'une seule question (réponse par OUI ou NON), le tableau A est pris en considération uniquement si la réponse est OUI.

Lignes 500 à 540 : résultats


Ces lignes affichent à l'écran votre espérance de vie (variable E).

Si vous obtenez un nombre d'années inférieur à votre âge réel, vous avez sûrement menti quelque part !

LISTE DES VARIABLES :

A\$	dernière touche appuyée
AGE	âge actuel de l'utilisateur
A()	années à ajouter ou à retrancher suivant la réponse aux questions
E	espérance de vie
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de réponses possibles
Q\$()	tableau des questions
R	réponse de l'utilisateur (numérique)
R\$	réponse de l'utilisateur (alpha-numérique)

BIORYTHME



```

10 REM BIORYTHME
20 REM -----
30 DIM M$(12),M(12)
40 FOR I=1 TO 12:READ M$(I),M(I):NEXT I
50 DATA JANVIER,31,FEVRIER,28,MARS,31,AVRIL,30,MAI
,31,JUIN,30
60 DATA JUILLET,31,AOUT,31,SEPTEMBRE,30,OCTOBRE,31
,NOVEMBRE,30,DECEMBRE,31
70 MODE 1:INK 0,0:INK 1,26:INK 2,2:INK 3,6:PAPER 0
:PEN 1:BORDER 0
180 REM -----
190 REM SAISIE DES DATES
200 INPUT"VOTRE PRENOM";P$
210 PRINT"JOUR, MOIS, ANNEE DE NAISSANCE"
220 INPUT J,M,A
230 PRINT"MOIS, ANNEE DU BIORYTHME"
240 INPUT V,W
250 IF W<A THEN PRINT"IMPOSSIBLE":GOTO 210
260 IF W=A AND M>=V THEN PRINT"IMPOSSIBLE":GOTO 21
0
280 REM -----
290 REM CALCULS
300 R=0
310 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
320 R=M(M)-J+1
330 IF A=W THEN D=V-1 ELSE D=12
340 FOR I=M+1 TO D:R=R+M(I):NEXT I
350 IF A=W THEN 410
360 FOR I=A+1 TO W-1
370 R=R+365:IF I/4=INT(I/4) THEN R=R+1
380 NEXT I
390 IF W/4=INT(W/4) THEN M(2)=29 ELSE M(2)=28
400 FOR I=1 TO V-1:R=R+M(I):NEXT I
410 PRINT R
480 REM -----
490 REM DESSIN DU BIORYTHME
500 CLS
510 PRINT P$;" NE LE";J;M$(M);A
520 PRINT"BIORYTHME POUR ";M$(V);W
530 PRINT"^"
540 MOVE 6,360:DRAW 6,50,1:MOVE 8,360:DRAW 8,50,1
550 MOVE 6,200:DRAW 520,200,1
560 FOR J=0 TO (M(V)-1)*16
570 P=200+INT(120*SIN((R+(J/16))*2*PI/23))
580 E=200+INT(120*SIN((R+(J/16))*2*PI/28))
590 I=200+INT(120*SIN((R+(J/16))*2*PI/33))
600 PLOT J+20,P,1:PLOT J+20,E,2:PLOT J+20,I,3
610 IF J>=M(V) THEN 640
620 LOCATE J+2,23:PRINT MID$(STR$(J+1),2,1)
630 LOCATE J+2,24:PRINT MID$(STR$(J+1)+" ",3)
640 NEXT J
650 LOCATE 35,4:PRINT"+":LOCATE 35,21:PRINT"- "
660 PEN 1:LOCATE 35,12:PRINT"P"
670 PEN 2:LOCATE 35,13:PRINT"E"
680 PEN 3:LOCATE 35,14:PRINT"I":PEN 1
690 IF INKEY$<>CHR$(13) THEN 690
700 RUN

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

Après dimensionnement des tableaux M\$() et M(), on y place le nom et le nombre de jours de chaque mois que l'on lit dans les DATAs des lignes 50 et 60.

Lignes 200 à 260 : saisie des dates

On saisit dans ces lignes tous les renseignements nécessaires à l'élaboration du biorythme.

Lignes 300 à 410 : calcul

On calcule le nombre de jours vécus par l'intéressé, en tenant compte des années bissextiles évidemment.

Lignes 500 à 700 : édition

Avec les renseignements collectés ci-dessus, on peut éditer le biorythme. Après les axes, on dessine point par point les trois courbes en calculant leurs ordonnées par une fonction sinus. L'abscisse correspond aux numéros des jours concernés.


LISTE DES VARIABLES :

A	année de naissance
D	nombre de mois restants dans l'année de naissance
E	ordonnée du point "émotionnel"
I	ordonnée du point "intellectuel"
J	jour de naissance
M	mois de naissance
P	ordonnée du point "physique"
P\$	prénom
PI	valeur de PI
R	nombre de jours vécus
V	mois du biorythme
W	année du biorythme


JOUR DE LA SEMAINE

Etes-vous curieux de savoir si vous êtes né un jeudi ou un dimanche ?

Voilà un petit programme qui vous donnera, outre le jour de la semaine de votre date de naissance, le nombre de jours, d'heures et de minutes que vous avez déjà vécus (nombre parfois impressionnant !).



JOUR DE LA SEMAINE



```

10 REM JOUR DE LA SEMAINE
20 REM -----
30 DIM M$(12),M(12)
40 FOR I=1 TO 12:READ M$(I),M(I):NEXT
50 DATA JANVIER,31,FEVRIER,28,MARS,32,AVRIL,30,MAI
,31,JUIN,30
60 DATA JUILLET,31,AOUT,31,SEPTEMBRE,30,OCTOBRE,31
,NOVEMBRE,30,DECEMBRE,31
70 MODE 0:INK 0,1:INK 1,24: BORDER 1
80 FOR I=1 TO 7:READ JS(I):NEXT
90 DATA LUNDI,MARDI,MERCREDI,JEUDI,VENDREDI,SAMEDI
,DIMANCHE
180 REM -----
190 REM SAISIE DES DATES
200 PRINT"JOUR, MOIS, ANNEE DE NAISSANCE"
210 INPUT J,M,A
220 PRINT"JOUR, MOIS, ANNEE ACTUELS"
230 INPUT U,V,W
240 IF A=W AND V<=M THEN PRINT"IMPOSSIBLE":GOTO 200
0
250 IF A>99 OR W>99 OR A>W THEN 200
280 REM -----
290 REM CALCUL DU NB DE JOURS
300 R=0
310 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
320 R=M(M)-J+1
330 IF A=W THEN D=V-1 ELSE D=12
340 FOR I=M+1 TO D:R=R+M(I):NEXT I
350 IF A=W THEN 410
360 FOR I=A+1 TO W-1
370 R=R+365:IF I/4=INT(I/4) THEN R=R+1
380 NEXT I
390 IF W/4=INT(W/4) THEN M(2)=29 ELSE M(2)=28
400 FOR I=1 TO V-1:R=R+M(I):NEXT
410 R=R+U-1
480 REM -----
490 REM CALCUL DU JOUR DE LA SEMAINE
500 S=(A*365)+INT((A-1)/4)
510 IF A=0 THEN S=0
520 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
530 FOR I=1 TO M-1:S=S+M(I):NEXT I
540 S=S+J-1
550 Y=S-(7*INT(S/7))+1
580 REM -----
590 REM RESULTATS
600 CLS
610 PRINT"VOUS ETES NE(E) LE ":PRINT
620 PRINT JS(Y);J;M$(M):PRINT TAB(7);1900+A
630 PRINT:PRINT:PRINT
640 PRINT"VOUS AVEZ VECU ":PRINT
650 PRINT USING"#####";R;:PRINT" JOURS"
660 PRINT USING"#####";R*24;:PRINT" HEURES"
670 PRINT USING"#####";R*1440;:PRINT" MINUTE
S"
680 PRINT USING"#####";R*86400;:PRINT" SECON
DES"
690 PRINT:PRINT

```

LE PROGRAMME

Lignes 30 à 90 : initialisation

La ligne 40 remplit les tableaux M\$() et M() avec le nom et le nombre de jours par mois avec les DATAs des lignes 50 et 60.

La ligne 80 remplit, elle, le tableau JS() avec le nom des jours de la semaine.

Lignes 200 à 250 : saisie des dates

Les lignes 200 à 230 saisissent successivement le jour, le mois et l'année de naissance, puis le jour, le mois et l'année actuelle.

En cas d'erreur, la ligne 240 affiche le message "IMPOSSIBLE".

Lignes 300 à 410 : calcul du nombre de jours

On calcule d'abord le nombre de jours entre la date de naissance et le 31 décembre de cette même année. Puis le nombre de jours entre l'année de naissance + 1 et l'année actuelle - 1. Pour terminer on ajoute le nombre de jours entre le 1^{er} janvier de l'année actuelle et la date désirée.

Lignes 500 à 550 : calcul du jour de la semaine

On calcule pour cela le nombre de jours entre le 1^{er} janvier 1900 et la date de naissance.

Lignes 600 à 690 : résultats

Les lignes 320 à 500 affichent sur l'écran les résultats des calculs précédents :

- La date de naissance
- Le nombre de jours vécus
- Le nombre d'heures (on multiplie par 24 le nombre de jours)
- Le nombre de minutes (on multiplie par 1440)
- Le nombre de secondes (on multiplie par 86400)

On utilise dans ces lignes l'instruction PRINT USING pour cadrer les chiffres sur l'écran.

LISTE DES VARIABLES :

A	année de naissance
I	indice de boucle FOR/NEXT
J	jour de naissance
J\$()	noms des jours de la semaine
M	mois de naissance
M\$()	noms des mois
M()	nombre de jours par mois
R	nombre de jours vécus
S	nombre de jours entre le 1/1/1900 et la date de naissance
U	jour actuel
V	mois actuel
W	année actuelle
Y	jour de la semaine (exemple : Y=1 → lundi)

CALENDRIER

Comme son titre l'indique, ce programme permet d'obtenir sur l'écran un calendrier complet d'une année du vingtième siècle.

En effet, après avoir tapé RUN, l'ordinateur demande l'année du calendrier que vous souhaitez.


Cette année doit être rentrée sous la forme de ses deux derniers chiffres. Exemple : entrez 84 pour 1984 ; cela limite les demandes de 0 à 99, c'est-à-dire des années 1900 à 1999.


L'ordinateur affiche le calendrier mois par mois dès que vous appuyez sur une touche (appuyez sur "F" pour une autre année). Il précise, à gauche et à droite de chaque mois, le nombre de jours passés et à passer.

De quoi jeter un coup d'œil discret vers 1990 ou 1999 !

243 SEPTEMB. 85 122						
L	M	M	J	V	S	D
						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30						

CALENDRIER





```

10 REM CALENDRIER
20 REM -----
30 DIM M$(12),N(12)
40 DATA JANVIER,31,FEVRIER,28,MARS,31
50 DATA AVRIL,30,MAI,31,JUIN,30,JUILLET,31,AOUT,31
60 DATA SEPTEMB.,30,OCTOBRE,31,NOVEMBRE,30,DECEMBR
E,31
70 FOR I=1 TO 12
80 READ M$(I),N(I)
90 NEXT I
180 REM -----
190 REM CALCUL DU JOUR
200 J=0:B=0
210 MODE 1:INK 0,0:INK 1,15:BORDER 0
220 INPUT"ANNEE ";U
230 IF U>99 THEN 210
240 M=(U*365)+INT((U-1)/4)
250 IF U=0 THEN M=0
260 Y=M-(7*INT(M/7))+1
270 IF INT(U/4)=U/4 THEN B=1
280 REM -----
290 REM EDITION
300 FOR I=1 TO 12
310 NJ=N(I)
320 IF I=2 AND B=1 THEN NJ=29
330 CLS:PRINT:PRINT
340 PRINT TAB(9);J;TAB(18-LEN(M$(I))/2);M$(I);U;TA
B(27);USING"###";365+B-J
350 PRINT:PRINT
360 PRINT TAB(10);" L M M J V S D"
370 PRINT:PRINT:PRINT
380 FOR K=1 TO NJ
390 PRINT TAB((Y-1)*3+10);USING"###";K;
400 Y=Y+1
410 IF Y=8 THEN Y=1:PRINT:PRINT
420 NEXT K
430 J=J+NJ
440 MOVE 125,390:DRAW 480,390:DRAW 480,50:DRAW 125
,50:DRAW 125,390
450 MOVE 130,370:DRAW 475,370:DRAW 475,350:DRAW 13
0,350:DRAW 130,370
460 FOR K=1 TO 8
470 MOVE 88+K*48,330:DRAW 88+K*48,70
480 NEXT K
490 MOVE 136,330:DRAW 471,330
500 MOVE 136,70:DRAW 471,70
510 AS=INKEY$:IF AS="" THEN 510 ELSE IF AS="F" THE
N RUN
520 NEXT I
530 GOTO 200

```

LE PROGRAMME

Lignes 30 à 90 : initialisation

Après dimensionnement des tableaux M\$ et N, on remplit ceux-ci avec la liste des noms des mois de l'année et le nombre de jours par mois.

Lignes 200 à 270 : calcul du jour de la semaine

On saisit l'année du calendrier dans la variable U. Cette variable est alors utilisée pour connaître le numéro du jour de la semaine du 1^{er} janvier de cette année (lundi=1, mardi=2, etc.).

En ligne 270, on regarde s'il s'agit d'une année bissextile ; si c'est le cas, on met la variable B à 1.

Lignes 300 à 530 : édition du calendrier

Pour chaque mois de l'année, on extrait du tableau N le nombre de jours par mois (variable NJ).

S'il s'agit du mois de février (l=2), on vérifie le contenu de la variable B (année bissextile). Si c'est le cas, on place 29 dans la variable NJ au lieu de 28 comme il est indiqué dans le tableau N().

La ligne 340 affiche le nombre de jours écoulés (variable J), le nom du mois (tableau M\$), l'année étudiée (variable U) et le nombre de jours restants ($365 + B - J$).

La variable Y donnant la position du jour suivant, on affiche le numéro de ce jour en utilisant l'instruction TAB(Y).

Quand $Y=8$ (le 8^e jour de la semaine !), on remet cette variable à 1, et ainsi de suite.

En ligne 430, on incrémente le nombre de jours écoulés du nombre de jours que l'on vient d'afficher.

LISTE DES VARIABLES :

A\$	dernière touche enfoncée
B	B=1 si année bissextile
I	indice de boucle FOR/NEXT
J	nombre de jours écoulés
K	indice de boucle FOR/NEXT pour l'édition des jours
K\$	valeur de K sous forme de chaîne de caractères
M	nombre de jours écoulés depuis le 1 ^{er} janvier 1900
M\$()	liste des mois
N()	nombre de jours par mois
NJ	nombre de jours du mois étudié
U	année à étudier
Y	numéro du jour de la semaine (lundi=1, mardi=2, ...)


HORLOGE


Un nouveau rôle pour votre AMSTRAD. Vous pourrez l'installer sur un meuble de votre salon ou pourquoi pas dans votre cuisine !

Entrez l'heure au clavier en laissant un peu d'avance. Dès que le dessin est terminé, appuyez sur n'importe quelle touche pour mettre la pendule en route.

Le compteur interne utilisé est d'une très grande précision. Il permet d'obtenir une variation inférieure à une seconde par jour.

Assurez-vous que vous ne tapez que des chiffres de 0 à 12.





```

10 REM HORLOGE
20 REM -----
30 MODE 0:INK 0,6:INK 1,24: BORDER 6
40 INPUT "H,M,S";H,M,S
50 H=H*5+INT(M/12)
60 DIM X(62),Y(62)
70 GOSUB 500
80 IF INKEY$="" THEN 80
180 REM -----
190 REM MOUVEMENT DES AIGUILLES
200 EVERY 50 GOSUB 220
210 GOTO 210
220 S=S+1
230 IF S<60 THEN 320
240 S=0:M=M+1:IF M<60 THEN 260
250 M=0
260 IF INT(M/12)<>M/12 THEN 310
270 H=H+1
280 IF H<60 THEN 300
290 H=0
300 PLOT 320,200: DRAW 320-60*X(H+1),200-60*Y(H+1),
0
310 PLOT 320,200: DRAW 320-80*X(M+1),200-80*Y(M+1),
0
320 PLOT 320,200: DRAW 320-95*X(S+1),200-95*Y(S+1),
0
330 PLOT 320,200: DRAW 320-60*X(H+2),200-60*Y(H+2),
1
340 PLOT 320,200: DRAW 320-80*X(M+2),200-80*Y(M+2),
1
350 PLOT 320,200: DRAW 320-95*X(S+2),200-95*Y(S+2),
1
360 RETURN
480 REM -----
490 REM DESSIN DE L'HORLOGE
500 CLG
510 PLOT 500,200
520 FOR I=0 TO 6.3 STEP PI/30
530 DRAW 320+180*COS(I),200+180*SIN(I)
540 NEXT I
550 FOR I=-PI TO PI STEP 0.104717
560 PLOT 320-100*SIN(I),200-100*COS(I)
570 X(Z+2)=SIN(I):Y(Z+2)=COS(I)
580 IF INT(Z/5)<>Z/5 THEN 640
590 DRAW 320-110*SIN(I),200-110*COS(I)
600 IF Z=0 THEN 640
610 IF Z/5=12 THEN LOCATE 10.5-4.5*SIN(I),13+8*COS
(I):PRINT RIGHTS(STR$(Z/5),2):GOTO 640
620 LOCATE 10.5-4.5*SIN(I),13+8*COS(I):PRINT RIGHT
$(STR$(Z/5),1)
630 IF Z/5>9 THEN LOCATE 9.5-4.5*SIN(I),13+8*COS(I
):PRINT MIDS(STR$(Z/5),2)
640 Z=Z+1
650 NEXT I
660 X(1)=X(61):Y(1)=Y(61)
670 RETURN

```

LE PROGRAMME

Lignes 30 à 80 : initialisation

La ligne 40 saisit l'heure (heures, minutes et secondes). Pour plus de réalité, la petite aiguille est déplacée toutes les 12 minutes et non toutes les heures. On multiplie donc la variable H par 5 (12 minutes \times 5 = 60 minutes = 1 heure). On l'incrmente aussi du nombre de fractions de 12 minutes contenues dans la variable M.

Lignes 200 à 360 : mouvement des aiguilles

Le compteur défini en ligne 200 fait appel au sous-programme en 220 toutes les secondes. Le compteur S est alors incrémenté de 1.

Les lignes 240 à 290 modifient le contenu de M quand S=60, puis H toutes les 12 minutes.

Les lignes 300 à 350 affichent les trois aiguilles par DRAW grâce aux deux tableaux X() et Y() définis en 500.

Lignes 500 à 670 : dessin de l'horloge

Les lignes 520 à 540 dessinent un cercle sur l'écran, tandis que les lignes 550 à 650 affichent les heures et des tirets pour les secondes. Les coordonnées de ces derniers sont conservées dans les tableaux X() et Y() ce qui facilitera l'affichage des aiguilles.

LISTE DES VARIABLES :


A\$	dernière touche appuyée
H	compteur des heures
I	indice de boucle FOR/NEXT
M	compteur des minutes
S	compteur des secondes
X() et Y()	gardent les coordonnées des aiguilles sur l'écran
Z	compteur des minutes dans l'affichage. On affiche les heures seulement toutes les 5 minutes.

AGENDA ELECTRONIQUE

Pour ne plus oublier de rendez-vous, ou consulter d'un seul coup le planning de votre semaine, utilisez donc ce programme d'agenda.


Il est composé de trois parties :

- **SAISIE** : permet de saisir la date, l'heure et les commentaires sur un rendez-vous ou un fait précis.
- **RECHERCHE** : en donnant une date de début et de fin, le programme affiche tous les enregistrements compris dans cet intervalle.
- **SAUVEGARDE** : sauve le fichier sur cassette.



DATE	HH	COMMENTAIRES
4/	1 10	RDU SA DUPONT
15/	1 16	PASSER A LA BANQUE
23/	1 8	PENSER A PAYER LE GAZ
2/	2 20	DINER AVEC LE PATRON
6/	2 14	RDU SA DURANT
15/	2 16	PRENDRE LES ENFANTS A L'E COLE
16/	2 12	DEJEUNER AVEC MARCEL
5/	3 10	RDU SA DUPONT

AGENDA ELECTRONIQUE



```

10 REM AGENDA ELECTRONIQUE
20 REM -----
30 DIM D(1000),H(1000),C$(1000),J(12)
40 E$=STRING$(39,32)+CHR$(13)
50 FOR I=1 TO 12:READ J(I):NEXT I
60 DATA 31,29,31,30,31,30
70 DATA 31,31,30,31,30,31
80 MODE 1:INK 0,0:INK 1,15:PAPER 0:PEN 1:BORDER 0
90 INPUT"PREMIERE UTILISATION (O/N)";R$
100 IF R$="0" THEN 200
110 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
120 OPENIN"!AGENDA"
130 IF EOF THEN 170
140 N=N+1
150 INPUT#9,D(N),H(N):LINE INPUT#9,C$(N)
160 GOTO 130
170 CLOSEIN
180 REM -----
190 REM MENU
200 CLS:PRINT TAB(18);"MENU":PRINT
210 PRINT TAB(13);"1 SAISIE 1":PRINT TAB(13);"
2 RECHERCHES 2":PRINT TAB(13);"3 SAUVEGARDE 3"
220 PRINT:PRINT TAB(14);"VOTRE CHOIX ?";
230 R$=INKEY$:IF R$="" THEN 230
240 R=VAL(R$):IF R<1 OR R>3 THEN 230
250 ON R GOTO 300,400,800
280 REM -----
290 REM SAISIE
300 CLS:PRINT TAB(10);"SAISIE D'UN RENDEZ-VOUS":PR
INT
310 GOSUB 700
320 N=N+1
330 LOCATE 10,14:PRINT E$;"HEURE : ";:LINE INPUT
H$:H(N)=INT(VAL(H$))
340 IF H(N)<0 OR H(N)>24 THEN 330
350 LOCATE 1,17:LINE INPUT"COMMENTAIRES: ";C$(N)
360 D(N)=M*100+J
370 GOTO 200
380 REM -----
390 REM RECHERCHES
400 CLS:PRINT TAB(15);"RECHERCHES"
410 PRINT:PRINT TAB(14);"DATE DE DEBUT"
420 GOSUB 700
430 DD=M*100+J
440 CLS:PRINT TAB(15);"RECHERCHES"
450 PRINT:PRINT TAB(14);"DATE DE FIN"
460 GOSUB 700
470 DF=M*100+J
480 CLS
490 PRINT"DATE HH COMMENTAIRES"
500 PRINT"---"
510 FOR I=1 TO N
520 IF D(I)<DD OR D(I)>DF THEN 600
530 M=INT((D(I))/100)
540 J=D(I)-M*100
550 PRINT USING"##/";J;
560 PRINT USING"## ";M,H(I);
570 FOR J=0 TO LEN(C$(I))/25
580 PRINT TAB(10);MID$(C$(I),1+J*25,25)
590 NEXT J

```

```

600 NEXT I
610 IF INKEY$<>CHR$(13) THEN 610
620 GOTO 200
680 REM -----
690 REM SAISIE D'UNE DATE
700 LOCATE 10,10:PRINT E$;"JOUR   : ";:LINE INPUT
J$:J=VAL(J$)
710 IF J>31 OR J<1 THEN 700
720 LOCATE 10,12:PRINT E$;"MOIS   : ";:LINE INPUT
M$:M=VAL(M$)
730 IF M>12 OR M<1 THEN 720
740 IF J>J(M) THEN 700
750 RETURN
780 REM -----
790 REM SAUVEGARDE
800 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R   EST PRET POUR L'ENREGISTREMENT"
810 LINE INPUT R$
820 OPENOUT "!AGENDA"
830 FOR I=1 TO N
840 PRINT#9,D(I),H(I),C$(I)
850 NEXT I
860 CLOSEOUT
870 CLS:PRINT"A LA PROCHAINE !"
880 END

```

LE PROGRAMME

Lignes 30 à 170 : initialisation

On place dans le tableau J() le nombre de jours par mois.

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 250 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 300 à 370 : saisie

Pour saisir un rendez-vous, on fait appel au sous-programme en 700 (saisie des dates).

Les lignes 330 et 350 demandent à l'utilisateur l'heure et les commentaires éventuels sur ce rendez-vous.

Lignes 400 à 620 : recherches

Le programme demande les dates de début et de fin pour les recherches. On utilise aussi pour cela le sous-programme de saisie des dates en 700.

Les lignes 510 à 600 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements compris entre la date de début et la date de fin.

Lignes 700 à 750 : saisie d'une date

Ces lignes saisissent au clavier le jour et le mois. Les lignes 710, 730 et 740 vérifient les éventuelles erreurs.

Lignes 800 à 880 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

E\$	caractères d'effacement d'une ligne
C\$()	liste des commentaires
D()	liste des dates
DD	date de début des recherches
DF	date de fin des recherches
H()	liste des heures
H\$	heure du rendez-vous
I	indice de boucle FOR/NEXT
J	jour du rendez-vous (numérique)
J()	nombre de jours par mois
JS	jour du rendez-vous
M	heure du rendez-vous (numérique)
M\$	heure du rendez-vous
N	nombre d'enregistrements dans le fichier
R\$	choix de l'utilisateur
R	valeur numérique de R\$



THEME 6 VOTRE FORME


POIDS IDEAL

Ce programme ne vous donnera pas votre poids idéal sur le plan esthétique, car cela dépend surtout de l'image physique que l'on veut présenter. Mais il calcule le poids idéal dont se servent les assureurs-vie pour juger de la longévité éventuelle de leurs clients selon le rapport poids-taille.

Donnez votre âge, votre taille et votre carrure : AMSTRAD se charge des calculs et vous révélera le poids le plus favorable à votre santé.



POIDS IDEAL



```

10 REM POIDS IDEAL
20 REM -----
30 MODE 1:INK 0,6:INK 1,24: BORDER 6:PAPER 0: PEN 1
40 INPUT "VOTRE AGE ";A
50 IF A<0 OR A>100 THEN 40
60 INPUT "TAILLE (EN CMS) ";T
70 IF T<30 OR T>250 THEN 60
80 INPUT "CARRURE (ETROITE=1,MOY.=2,LARGE=3) ";C
90 IF C<1 OR C>3 THEN 80
100 INPUT "SEXE FEMININ (O/N) ";S$
110 IF S$<>"O" AND S$<>"N" THEN 100
120 IF S$="O" THEN S=0.8 ELSE S=0.9
180 REM -----
190 REM CALCULS
200 MODE 0:PRINT:PRINT:PRINT:PRINT
210 PRINT " VOTRE POIDS IDEAL":PRINT
220 PRINT TAB(7);"EST DE:":PRINT:PRINT
230 PRINT TAB(6);INT((T-100+C+(A/20))*S);"KGS"
240 PRINT:PRINT
250 PRINT TAB(4);"A + OU -";C;"KGS"
260 IF INKEY$<>CHR$(13) THEN 260

```

LE PROGRAMME

Lignes 30 à 120 : saisie des éléments

Le programme saisit successivement l'âge, la taille, la carrure et le sexe du sujet.

Les lignes 50, 70, 90 et 110 vérifient que les réponses soient vraisemblables.

Lignes 200 à 260 : calculs

Ces lignes calculent la fourchette de votre poids idéal en appliquant la formule :

$$\text{taille} - 100 + \text{carrure} + \text{âge}/20$$


le tout multiplié par 0.8 pour les femmes et 0.9 pour les hommes.

La fourchette est donnée par la carrure (entre 1 et 3 kilogrammes).

LISTE DES VARIABLES :

A	âge
C	carrure
\$\$	sexe féminin (oui ou non)
S	multiplicateur suivant le sexe
T	taille en cm

COURBES POIDS/TAILLE



```

10 REM COURBES POIDS/TAILLE
20 REM -----
30 DIM P(104),T(104)
40 MODE 1:INK 0,0:INK 1,15:BORDER 0
50 INPUT"PREMIERE UTILISATION (O/N)";R$
60 IF R$="0" THEN 200
70 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
80 OPENIN "!POIDS"
90 FOR I=0 TO 104
100 INPUT#9,P(I),T(I)
110 IF S=0 AND P(I)=0 THEN S=I
120 NEXT I
130 CLOSEIN
180 REM -----
190 REM DONNEES DE LA SEMAINE
200 CLS
210 IF S=0 THEN PRINT"ENTREZ LES DONNEES A LA NAIS
SANCE":PRINT:GOTO 250
220 PRINT"ENTREZ LES DONNEES DE LA";S;
230 IF S=1 THEN PRINT"ere"; ELSE PRINT"eme";
240 PRINT" SEMAINE":PRINT
250 INPUT"POIDS (EN KILOGRAMMES) ";P(S)
260 IF P(S)>13 THEN 250
270 INPUT"TAILLE (EN CENTIMETRES) ";T(S)
280 IF T(S)>130 THEN 270
290 IF S=0 THEN S=1:GOTO 200
380 REM -----
390 REM COURBES DE LA SEMAINE
400 CLS:PRINT"TAILLE/POIDS":PRINT" CMS KG"
410 LOCATE 12,3:PRINT"^":LOCATE 39,17:PRINT">"
420 MOVE 182,354:DRAW 182,136:DRAW 624,136:MOVE 18
4,354:DRAW 184,136
430 FOR I=0 TO 13
440 LOCATE 1,I+4:PRINT 130-I*10:LOCATE 7,I+4:PRINT
13-I
450 MOVE 184,344-I*16:DRAW 176,344-I*16
460 NEXT I
470 FOR I=1 TO 26
480 A$=RIGHT$(" "+STR$(I*4),3)
490 LOCATE I+12,19:PRINT LEFT$(A$,1):LOCATE I+12,2
0:PRINT MID$(A$,2,1)
500 LOCATE I+12,21:PRINT RIGHT$(A$,1)
510 MOVE I*16+184,138:DRAW I*16+184,130
520 NEXT I
530 LOCATE 30,23:PRINT"SEMAINES"
540 PLOT 182,136+P(0)*16
550 FOR I=1 TO S
560 DRAW 182+I*4,136+P(I)*16
570 NEXT I
580 LOCATE 13+I/4,17-P(S):PRINT"P"
590 PLOT 182,136+(T(0)/10)*16
600 FOR I=1 TO S

```

```
610 DRAW 182+I*4,136+(T(I)/10)*16
620 NEXT I
630 LOCATE 13+I/4,17-(T(S)/10):PRINT"T"
640 LOCATE 1,24:PRINT"UNE AUTRE SEMAINE (O/N) ";
650 R$=INKEY$:IF R$="" THEN 650
660 IF R$="0" THEN S=S+1:GOTO 200
680 REM -----
690 REM SAUVEGARDE
700 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R EST PRET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 OPENOUT "!POIDS"
730 FOR I=0 TO 104
740 PRINT#9,P(I),T(I)
750 NEXT I
760 CLOSEOUT
770 CLS:PRINT"A LA SEMAINE PROCHAINE !"
780 END
```

LE PROGRAMME

Lignes 30 à 130 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 290 : données de la semaine

Le programme demande le poids et la taille de la semaine S. S'il s'agit de la première utilisation (S=0), la ligne 210 affiche le message "données à la naissance".

Lignes 400 à 660 : courbes de la semaine

Ces lignes dessinent sur l'écran les courbes du poids et de la taille de l'enfant de la naissance à la semaine en cours.

Les lignes 400 à 460 affichent l'axe des ordonnées, les lignes 470 à 520 les valeurs des poids et des tailles, et les lignes 530 à 630 tracent l'abscisse et les deux courbes.


La ligne 660 renvoie à la saisie si l'on désire entrer les données de la semaine suivante avant la sauvegarde.

Lignes 700 à 780 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A\$	poids/tailles à afficher (graduations de l'ordonnée)
I	indice de boucle FOR/NEXT
P()	liste des poids
R\$	réponse de l'utilisateur
S	numéro de la semaine (taille du fichier)
T()	liste des tailles



THEME 7

BRICOLAGE

PAPIER PEINT


Ce programme permet de calculer en fonction de certaines caractéristiques le nombre de rouleaux à acheter pour pouvoir tapisser une pièce ainsi que le rendement et le coût de l'opération.

Au départ, l'ordinateur vous demande le prix du rouleau, sa largeur et sa longueur, puis le périmètre et la hauteur de la pièce.


Ces mesures sont à entrer en mètres pour les calculs de l'ordinateur, après lesquels il vous précise le nombre de rouleaux à acheter, le nombre de lés que vous pouvez découper dans chaque rouleau, la longueur des chutes pour les raccords éventuels et le reste inutilisé de papier.

De même il vous donne le coût total de votre achat et vous chiffre les pertes de papier.

Vous pouvez garder AMSTRAD à côté de votre colle et de vos pinces : il sera lui aussi un outil de travail précieux.



PAPIER PEINT



```

10 REM PAPIER PEINT
20 REM -----
30 MODE 1:INK 0,0:INK 1,15:BORDER 0
40 PRINT TAB(14);"PAPIER PEINT"
50 PRINT:PRINT:PRINT
60 INPUT"PRIX DU ROULEAU";PR
70 IF PR<1 THEN PRINT"C'EST PAS CHER. DONNEZ MOI L
'ADRESSE !":GOTO 60
80 INPUT"LARGEUR DU PAPIER";LP
90 IF LP<0.1 THEN PRINT"C'EST DU PAPIER PEINT OU D
U PAPIER TOILETTE ?":GOTO 80
100 INPUT"LONGUEUR DES ROULEAUX";LR
110 IF LR<1 THEN PRINT"VOUS VOUS ETES FAIT ROULER
!":GOTO 100
120 INPUT"PERIMETRE DE LA PIECE";CP
130 IF CP>200 THEN PRINT"VOUS HABITEZ VERSAILLES J
E SUPPOSE ?":GOTO 120
140 IF CP<1 THEN PRINT"VOUS TAPISSEZ VOTRE POULAIL
LER ?":GOTO 120
150 INPUT"HAUTEUR DES MURS";HM
160 IF HM<1 THEN PRINT"C'EST BAS DE PLAFOND CHEZ V
OUS !":GOTO 150
280 REM -----
290 REM CALCULS
300 NL=INT(LR/HM)
310 LC=LR-NL*HM
320 NR=INT((CP/(NL*LP))+0.99999)
330 NTL=INT((CP/LP)+0.99999)
340 ND=NTL-((NR-1)*NL)
350 LD=LR-(ND*HM)
360 PT=NR*PR
370 TC=((NR-1)*LC)+LD
380 R=TC/((NR*LR)/100)
480 REM -----
490 REM RESULTATS
500 CLS
510 PRINT TAB(14);"PAPIER PEINT"
520 PRINT:PRINT:PRINT
530 PRINT"NOMBRE DE ROULEAUX:";NR
540 PRINT"NOMBRE DE LES PAR ROULEAU:";NL
550 PRINT"IL Y A";NR-1;"CHUTES DE";USING"###.## ";
LC;:PRINT"METRES"
560 PRINT"PLUS";USING"###.## ";LD;:PRINT"METRES SU
R LE DERNIER ROULEAU";
570 PRINT"IL RESTE DONC";USING"###.## ";TC;:PRINT"
METRES DE CHUTE":PRINT"POUR LES RACCORDS"
580 PRINT"LE PRIX TOTAL EST DE";PT;"FRANCS"
590 PRINT"ET LES PERTES DE";USING"###.## ";R;:PRINT
"% "
600 PRINT"SOIT";USING"###.## ";PT*R/100;:PRINT"FR
ANCS"
610 IF INKEY$<>CHR$(13) THEN 610
620 RUN

```

LE PROGRAMME

Lignes 30 à 160 : saisie des paramètres

On saisit à la suite le prix, la largeur et la longueur d'un rouleau, puis le périmètre et la hauteur des murs de la pièce à tapisser.

Lignes 300 à 380 : calculs

Cette partie du programme calcule toutes les dimensions des lés, des chutes et la composition du dernier rouleau.

La valeur 0.99999 ajoute un rouleau de plus dans le cas où le dernier est incomplet.

Lignes 500 à 620 : résultats

Toutes les données calculées précédemment sont affichées à l'écran, avec en plus le prix des pertes sur les chutes et le dernier rouleau s'il est incomplet.

LISTE DES VARIABLES :

CP	périmètre de la pièce
HP	hauteur des murs
Lc	longueur de la chute par rouleau
LD	longueur de la chute sur le dernier rouleau
LP	largeur du papier
LR	longueur des rouleaux
ND	nombre de lés sur le dernier rouleau
NL	nombre de lés par rouleau
NR	nombre de rouleaux
NTL	nombre total de lés
PR	prix du rouleau
PT	prix total du papier peint
R	rendement
TC	longueur totale des chutes

DISPOSITION DES MEUBLES

Le mieux pour changer votre appartement est encore d'en modifier la disposition des meubles.

Le programme est composé de deux parties :


- **LE GENERATEUR DE CARACTERES** : vous devez dessiner avec cet utilitaire la forme des meubles en utilisant les quatre flèches du clavier. La touche COPY permet de changer la couleur d'un point élémentaire.

Touches de fonction :


C : change le numéro du caractère en cours.

F : donne accès au deuxième programme (disposition des meubles).

- **DISPOSITION DES CARACTERES SUR L'ECRAN** : les quatre flèches permettent de déplacer le caractère en cours dans toutes les directions et sur toute la surface de l'écran. Comme pour le programme de définition, la touche C permet de changer le numéro du caractère en cours et la touche F de passer à l'autre partie (générateur de caractères).



DISPOSITION DES MEUBLES




```

10 REM DISPOSITION DES MEUBLES
20 REM -----
30 MODE 1:TAGOFF
40 DIM S(16,16,16),XX(16),YY(16),X(16),Y(16)
50 SYMBOL AFTER 128
60 SYMBOL 232,255,255,255,255,255,255,255,255
70 SYMBOL 233,255,128,128,128,128,128,128,128
80 FOR I=128 TO 192:SYMBOL I,0,0,0,0,0,0,0,0:NEXT
I
90 S=1:X=8:Y=8
100 FOR I=1 TO 16
110 X(I)=1:Y(I)=1
120 NEXT I
180 REM -----
190 REM DESSIN DU GENERATEUR
200 MODE 1
210 FOR I=0 TO 16
220 PLOT 0,399-I*16:DRAWR 256,0
230 PLOT I*16,399:DRAWR 0,-256
240 NEXT I
250 LOCATE 1,25:PRINT "CARACTERE No";S;
260 XX=25:YY=1
270 FOR I=1 TO 16
280 LOCATE XX,YY:PRINT "No";I
290 LOCATE XX+1,YY+1:PRINT CHR$(124+I*4);CHR$(125+
I*4)
300 LOCATE XX+1,YY+2:PRINT CHR$(126+I*4);CHR$(127+
I*4)
310 XX(I)=XX+1:YY(I)=YY+1
320 YY=YY+3:IF YY=25 THEN YY=1:XX=XX+8
330 NEXT I
380 REM -----
390 REM AFFICHE UN CARACTERE
400 FOR P=1 TO 4
410 IF P>2 THEN YY=8 ELSE YY=0
420 IF P/2=INT(P/2) THEN XX=8 ELSE XX=0
430 FOR I=1 TO 8:FOR J=7 TO 0 STEP -1
440 LOCATE XX+8-J,YY+I
450 IF S(S,P,I) AND 2^J THEN PRINT CHR$(232) ELSE
PRINT CHR$(233)
460 NEXT J,I,P
470 GOTO 750
580 REM -----
590 REM DEPLACEMENTS
600 IF X<9 THEN P=1:XX=X-1 ELSE P=2:XX=X-9
610 IF Y<9 THEN YY=Y ELSE P=P+2:YY=Y-8
620 A$=INKEY$:IF A$="" THEN 620
630 IF A$="C" THEN GOSUB 900:GOTO 400
640 IF A$="F" THEN 1000
650 IF A$<>CHR$(224) THEN 700
660 IF S(S,P,YY) AND 2^(7-XX) THEN S(S,P,YY)=S(S,P
,YY)-2^(7-XX) ELSE S(S,P,YY)=S(S,P,YY)+2^(7-XX)
670 SYMBOL 123+(S*4)+P,S(S,P,1),S(S,P,2),S(S,P,3),
S(S,P,4),S(S,P,5),S(S,P,6),S(S,P,7),S(S,P,8)
680 LOCATE XX(S),YY(S):PRINT CHR$(124+S*4);CHR$(12
5+S*4)
690 LOCATE XX(S),YY(S)+1:PRINT CHR$(126+S*4);CHR$(
127+S*4)
700 LOCATE X,Y:IF S(S,P,YY) AND 2^(7-XX) THEN PRIN

```

```

T CHR$(232) ELSE PRINT CHR$(233)
710 IF A$=CHR$(242) THEN X=X-1:IF X=0 THEN X=16
720 IF A$=CHR$(243) THEN X=X+1:IF X=17 THEN X=1
730 IF A$=CHR$(240) THEN Y=Y-1:IF Y=0 THEN Y=16
740 IF A$=CHR$(241) THEN Y=Y+1:IF Y=17 THEN Y=1
750 LOCATE X,Y:PRINT CHR$(231)
760 GOTO 600
880 REM -----
890 REM NUMERO DU CARACTERE
900 LOCATE 1,25:PRINT CHR$(18);"No CARACTERE ";
910 INPUT;" ";A$:S=VAL(A$):IF S<1 OR S>16 THEN 900
920 LOCATE 1,25:PRINT CHR$(18);"CARACTERE No";S;
930 RETURN
980 REM -----
990 REM DISPOSITION DES CARACTERES
1000 MODE 0
1010 GOSUB 1140:EVERY 200,0 GOSUB 1140:EI
1020 LOCATE 1,25:PRINT "CARACTERE No";S;
1030 A$=INKEY$:IF A$="" THEN 1030
1040 IF A$="F" THEN EVERY 0,0 GOSUB 1140:DI:GOTO 2
00
1050 IF A$="C" THEN DI:GOSUB 900:EI:GOTO 1030
1060 LOCATE X(S),Y(S):PRINT "  ":LOCATE X(S),Y(S)+1
:PRINT"  "
1070 IF A$=CHR$(242) THEN X(S)=X(S)-2:IF X(S)=-1 T
HEN X(S)=19
1080 IF A$=CHR$(243) THEN X(S)=X(S)+2:IF X(S)=21 T
HEN X(S)=1
1090 IF A$=CHR$(240) THEN Y(S)=Y(S)-2:IF Y(S)=-1 T
HEN Y(S)=23
1100 IF A$=CHR$(241) THEN Y(S)=Y(S)+2:IF Y(S)=25 T
HEN Y(S)=1
1110 LOCATE X(S),Y(S):PRINT CHR$(124+S*4);CHR$(125
+S*4)
1120 LOCATE X(S),Y(S)+1:PRINT CHR$(126+S*4);CHR$(1
27+S*4)
1130 GOTO 1030
1140 FOR I=16 TO 1 STEP -1
1150 LOCATE X(I),Y(I):PRINT CHR$(124+I*4);CHR$(125
+I*4)
1160 LOCATE X(I),Y(I)+1:PRINT CHR$(126+I*4);CHR$(1
27+I*4)
1170 NEXT I
1180 RETURN

```

LE PROGRAMME

Lignes 30 à 120 : initialisation

Les variables X et Y donnent les coordonnées du point de départ, S le numéro du caractère en cours et XX() et YY() la position des caractères dans le générateur (X() et Y() pour la disposition).

Lignes 200 à 330 : dessin du générateur

Affiche avec une boucle FOR/NEXT le dessin du générateur. Les lignes 290 et 300 placent les caractères aux coordonnées XX et YY, et la ligne 280 indique leurs numéros.

Lignes 400 à 470 : affichent un caractère

Ces lignes affichent sur le générateur le dessin grand format du caractère S.

Lignes 600 à 760 : déplacements

Cette partie du programme teste le clavier et déplace le curseur sur l'écran. Les cases changent de couleur si l'on appuie sur la touche COPY. La ligne 660 recalcule immédiatement les valeurs du caractère pour l'instruction SYMBOL.

Lignes 900 à 930 : numéro du caractère en cours

Le programme saisit au clavier le nouveau numéro du caractère.

Lignes 1300 à 1440 : disposition des caractères

On déplace sur l'écran le caractère en cours en testant le clavier. Les coordonnées des seize objets sont conservées dans les tableaux X() et Y().

LISTE DES VARIABLES :

A\$	dernière touche appuyée
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
P	numéro du caractère qui compose un meuble (4 au total)
S	numéro du sprite en cours
S(,,)	dessin binaire des caractères
X et Y	coordonnées du curseur
X() et Y()	coordonnées des caractères dans le générateur
XX et YY	coordonnées à l'intérieur d'un des caractères qui compose un meuble.
XX() et YY()	coordonnées des caractères dans la disposition des meubles.




THEME 8 UTILITAIRES

CRYPTOGRAMMES


Si vous aimez les messages secrets, les informations codées, les encres sympathiques, bref, si vous êtes un cryptologue averti, vous pourrez donner libre cours à votre passion en échangeant des textes indécodables... sauf pour celui qui en connaît la clef !

En effet, AMSTRAD vous demande s'il s'agit d'un message à coder ou à décoder, puis le mot de passe (la clef du message). Vous devez taper un message quelconque, par exemple "IF THEN ELSE" ou "DE NORBERT A MARCEL". Sans cette clef il n'est pas possible de décoder un message.

Yrf xojqcrvozsse !



CRYPTOGRAMMES



```

10 REM CRYPTOGRAMMES
20 REM -----
30 DIM C(26)
40 MODE 1:INK 0,0:INK 1,15:BORDER 0
50 INPUT"CODAGE OU DECODAGE (C/D) ";R$
60 IF R$="D" THEN 400
70 IF R$<>"C" THEN 40
180 REM -----
190 REM CODAGE
200 GOSUB 600
210 CLS
220 PRINT"MESSAGE CODE: "
230 FOR I=1 TO LEN(M$)
240 IF MID$(M$,I,1)<"A" OR MID$(M$,I,1)>"Z" THEN 2
60
250 PRINT CHR$(C(ASC(MID$(M$,I,1))-64)+64);
260 NEXT I
270 IF INKEY$="" THEN 270
280 RUN
380 REM -----
390 REM DECODAGE
400 GOSUB 600
410 CLS
420 PRINT"MESSAGE DECODE: "
430 FOR I=1 TO LEN(M$)
440 IF MID$(M$,I,1)<"A" OR MID$(M$,I,1)>"Z" THEN 4
80
450 FOR J=1 TO 26
460 IF ASC(MID$(M$,I,1))-64=C(J) THEN PRINT CHR$(J
+64);:GOTO 480
470 NEXT J
480 NEXT I
490 IF INKEY$="" THEN 490
500 RUN
580 REM -----
590 REM CREATION CODES
600 CLS
610 INPUT"MOI DE PASSE";P$
620 IF P$="" THEN 600
630 D=1
640 FOR I=1 TO LEN(P$)
650 IF MID$(P$,I,1)<"A" OR MID$(P$,I,1)>"Z" THEN 6
90
660 IF C(ASC(MID$(P$,I,1))-64)<>0 THEN 690
670 C(ASC(MID$(P$,I,1))-64)=D
680 D=D+1
690 NEXT I
700 FOR I=26 TO 1 STEP -1
710 IF C(I)<>0 THEN 740
720 C(I)=D
730 D=D+1
740 NEXT I
750 CLS
760 INPUT"MESSAGE ";M$
770 IF M$="" THEN 750
780 RETURN

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

La ligne 50 saisit le type d'opération à effectuer, codage ou décodage. Les lignes 60 et 70 envoient aux programmes concernés.

Lignes 200 à 280 : codage

Après l'appel du sous-programme de saisie du mot de passe, du message et de la création du tableau de transcription, on code une à une toutes les lettres du message M\$ à l'aide d'une boucle FOR/NEXT. Seules les lettres sont réellement codées, les autres signes sont ignorés.

Lignes 400 à 500 : décodage

Après l'appel du sous-programme de saisie du mot de passe, du message et de la création du tableau de transcription, on décode une à une toutes les lettres du message M\$ à l'aide d'une boucle FOR/NEXT.

Lignes 600 à 780 : création des codes

Dès que le mot de passe est saisi, on place dans le tableau C() la correspondance de chaque lettre (exemple : si C(1)=3, la lettre A devient C; si C(26)=5, la lettre Z devient E) en fonction du mot de passe P\$.

LISTE DES VARIABLES :

C()	tableau de transcription
D	nombre de lettres déjà dans le tableau C()
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
M\$	message à coder ou à décoder
P\$	mot de passe
R\$	choix de l'utilisateur

GESTION DES QSO

Votre AMSTRAD est aussi radio-amateur (il sait décidément tout faire, votre ordinateur !).

Si vous avez déjà utilisé ce programme, vous ne devez pas répondre à la question "Première utilisation" (tapez uniquement sur la touche "ENTER"). Il affiche alors un menu de six options :

1 – SAISIE D'UN QRZ : vous devez entrer le QRZ et le QTH d'un de vos correspondants.

Si vous n'avez plus rien à rentrer, appuyez sur "ENTER" à la question "QRZ:".

2 – SAISIE D'UN QSO : si vous désirez introduire dans le fichier un nouveau QSO, entrez le numéro du QRZ correspondant (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme saisit alors la date de la liaison (on peut mettre 4 QSO par enregistrement). Vous devez taper "Ø" en réponse à la question "Numéro du QRZ:" pour revenir au menu.

3 – SUPPRESSION : en donnant le numéro d'un QRZ, le programme supprime l'enregistrement correspondant dans le fichier.


4 – RECHERCHE : elle permet d'afficher un enregistrement complet à partir du renseignement que vous lui donnez. Par exemple le mot "Cachan" pour obtenir tous les Cibistes de votre connaissance habitant la ville de Cachan.

5 – LISTING : édite simplement sur l'écran le fichier dans son intégralité.

6 – SAUVEGARDE : le programme sauve le fichier sur cassette.

N.B. : le code Q utilisé est celui couramment employé par les Cibistes et différent des normes en vigueur.

GESTION DES QSO



```

10 REM GESTION DES QSO
20 REM -----
30 DIM D$(6,100)
40 L$(1)="QRZ":L$(2)="QTH":L$(3)="QSO DU"
50 MODE 1:INK 0,0:INK 1,15: BORDER 0
60 INPUT"PREMIERE UTILISATION (O/N)";R$
70 IF R$="O" THEN 200
80 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
90 OPENIN"!QSO"
100 INPUT#9,N
110 FOR I=1 TO N
120 FOR J=1 TO 6
130 LINE INPUT#9,D$(J,I)
140 NEXT J,I
150 CLOSEIN
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE QRZ: ";N:PRINT:PRINT
220 PRINT"1 SAISIE QRZ"
230 PRINT"2 SAISIE QSO"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INKEY$:IF R$="" THEN 290
300 R=VAL(R$):IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,800,1000,600,1200,1400
380 REM -----
390 REM SAISIE QRZ
400 CLS:PRINT"SAISIE D'UN QRZ":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:LINE INPUT
  R$:GOTO 200
420 N=N+1
430 FOR I=1 TO 2
440 PRINT L$(I);": ";:LINE INPUT D$(I,N)
450 IF I=1 AND D$(1,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"DONNEE A RECHERCHER: ";:LINE INPUT D$:PR
  INT
620 IF D$="" THEN 200
630 FOR I=1 TO N
640 FOR J=1 TO 6
650 IF INSTR(D$(J,I),D$)<>0 THEN 670
660 NEXT J:GOTO 690
670 PRINT USING"###";I;:FOR K=1 TO 2:PRINT " ";LEF
  T$(D$(K,I)+"",10);:NEXT K:PRINT
680 PRINT TAB(5);:FOR K=3 TO 6:PRINT LEFT$(D$(K,I)
  +",8);":NEXT K:PRINT:F=1
690 NEXT I:PRINT
700 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
710 IF INKEY$="" THEN 710 ELSE 600
780 REM -----
790 REM SAISIE DES QSO

```

```

800 CLS:PRINT"SAISIE DES QSO":PRINT
810 INPUT"NUMERO DU QRZ";NU
820 IF NU<1 OR NU>N THEN 200
830 FOR I=3 TO 6
840 IF D$(I,NU)=" " THEN 870
850 NEXT I
860 PRINT:PRINT"IL N'Y A PLUS DE PLACE";:LINE INPUT AS$:GOTO 880
870 PRINT L$(3);":"::LINE INPUT D$(I,NU)
880 GOTO 800
980 REM -----
990 REM SUPPRESSION
1000 CLS:PRINT"SUPPRESSION":PRINT
1010 INPUT"NUMERO DU QRZ";NU
1020 IF NU<1 OR NU>N THEN 200
1030 PRINT
1040 FOR I=1 TO 3
1050 PRINT I;L$(I);":":D$(I,NU)
1060 NEXT I:PRINT
1070 INPUT"VOUS ETES SUR DE SUPPRIMER CE QRZ";R$
1080 IF R$<>"0" THEN 1000
1090 N=N-1
1100 FOR I=NU TO N
1110 FOR J=1 TO 6
1120 D$(J,I)=D$(J,I+1)
1130 NEXT J,I
1140 GOTO 1000
1180 REM -----
1190 REM LISTING
1200 CLS:F=0
1210 FOR I=1 TO N
1220 PRINT USING"###";I;:FOR K=1 TO 2:PRINT " ";LEFT$(D$(K,I)+",10");:NEXT K:PRINT
1230 PRINT TAB(5);:FOR K=3 TO 6:PRINT LEFT$(D$(K,I)+",8");":":NEXT K:PRINT:F=F+3
1240 IF F>20 THEN PRINT:PRINT"APPUYEZ SUR ENTER POUR LA SUITE";:LINE INPUT R$:CLS:F=0
1250 NEXT I
1260 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE MENU";
1270 IF INKEY$="" THEN 1270 ELSE 200
1380 REM -----
1390 REM SAUVEGARDE
1400 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDER EST PRET POUR L'ENREGISTREMENT"
1410 LINE INPUT R$
1420 OPENOUT"!QSO"
1430 PRINT#9,N
1440 FOR I=1 TO N
1450 FOR J=1 TO 6
1460 PRINT#9,D$(J,I)
1470 NEXT J,I
1480 CLOSEOUT
1490 CLS:PRINT"A LA PROCHAINE !"
1500 END

```

LE PROGRAMME

Lignes 30 à 150 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'un QRZ

On saisit par une boucle FOR/NEXT le QRZ et le QTH. Si le nom de la station est une chaîne vide, on revient au menu (ligne 450).

Lignes 600 à 710 : recherche

Le programme demande d'abord la donnée sur laquelle on désire effectuer la recherche (variable D\$). Puis les lignes 630 à 690 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$.

Lignes 800 à 880 : saisie d'un QSO

Ces lignes saisissent la date d'un QSO dans la limite de quatre.

Lignes 1000 à 1140 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1100 à 1130 décalent le tableau vers le bas.

Lignes 1200 à 1270 : listing


On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1240 dès que l'écran est plein.

Lignes 1400 à 1500 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre de QRZ dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$



```
1 ASPIRINE   PARIS X
  10/01/84  05/06/84  15/10/84
2 BACCARA   BAGNEUX
  12/02/84
3 DRACULA   CACHAN
  05/05/84
4 ALTOS     CACHAN
  31/01/84  06/09/84
5 GAI-LURON PARIS XI
  06/06/84  07/06/84  08/06/84
6 ICARE     CACHAN
  10/09/83  11/12/83  16/04/84  09/10/84
7 SIDERAL   BICETRE
  16/11/83
```

APPUYEZ SUR ENTER POUR LA SUITE

GESTION D'UNE CAVE

Vous désirez un sommelier particulier ? En voici un tout désigné : votre AMSTRAD

Après la rituelle question "Première utilisation", le programme affiche un menu pour les options suivantes :

1 – SAISIE : vous devez entrer tous les renseignements sur un cru que vous avez dans votre cave. Les dates "Avant" et "Après" correspondent aux deux années (ou à une date complète si vous le désirez) entre lesquelles le vin sera le meilleur.

Si vous ne voulez plus entrer de crus, tapez uniquement sur "ENTER" à la question "CRU:".

2 – MODIFICATION : si vous désirez modifier un cru, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5).

Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à ajouter, taper "Ø" en réponse à la question "Numéro du cru:".


3 – SUPPRESSION : si vous avez malheureusement épuisé toutes les bouteilles d'un cru particulier, cette option permet de le supprimer du fichier. Comme pour la modification, taper "Ø" pour revenir au menu.

4 – RECHERCHE : c'est la partie la plus intéressante du programme. Elle permet d'afficher tous les crus à consommer entre deux dates fixes, de connaître toutes les bouteilles de Bordeaux ou de champagne qui vous attendent impatiemment dans votre cave ou celles qui vous restent de l'année 76. Il n'est pas nécessaire de donner un nom entier : le mot "BEAU" suffit à retrouver les crus d'appellation "BEAUjolais" ou les bouteilles de St-Emilion Château "BEAUséjour".

5 – LISTING : éditez simplement sur l'écran le fichier dans son intégralité.

6 – SAUVEGARDE : le programme sauve le fichier sur cassette.

GESTION D'UNE CAVE



```

10 REM GESTION D'UNE CAVE
20 REM -----
40 DIM D$(6,100)
50 L$(1)="CRU":L$(2)="APPELLATION":L$(3)="MILLESIM
E":L$(4)="NB DE BOUTEILLES":L$(5)="MEILLEUR APRES"
:L$(6)="MEILLEUR AVANT"
60 MODE 1:INK 0,0:INK 1,15: BORDER 0
70 INPUT"PREMIERE UTILISATION (0/N)";R$
80 IF R$="0" THEN 200
90 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
100 OPENIN "!CAVE"
110 INPUT#9,N
120 FOR I=1 TO N
130 FOR J=1 TO 6
140 INPUT#9,D$(J,I)
150 NEXT J,I
160 CLOSEIN
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE CRUS:";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INKEY$:IF R$="" THEN 290
300 R=VAL(R$):IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE
400 CLS:PRINT"SAISIE D'UN CRU":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:LINE INPUT
R$:GOTO 200
420 N=N+1
430 FOR I=1 TO 6
440 PRINT L$(I);":":LINE INPUT D$(I,N)
450 IF I=1 AND D$(I,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 6:PRINT I:L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$=INKEY$:IF R$="" THEN 640
650 R=VAL(R$):IF R<1 OR R>6 THEN 200
660 IF R<5 THEN 700
670 INPUT"DATE DEBUT ";DD
680 INPUT"DATE FIN ";DF
690 GOTO 710
700 PRINT L$(R);":":LINE INPUT D$:PRINT
710 PRINT
720 FOR I=1 TO N
730 IF R<5 THEN IF INSTR(D$(R,I),D$)=0 THEN 790 EL
SE 750
740 IF VAL(D$(R,I))<DD OR VAL(D$(R,I))>DF THEN 790
750 F=1:PRINT USING"### ";I;
760 PRINT D$(1,I);" (";D$(2,I);")"
770 PRINT TAB(5);"MILLESIME:";D$(3,I);TAB(20);"NB
DE B.:";D$(4,I)

```

```

780 PRINT TAB(9); "APRES: "; D$(5, I); TAB(23); "AVANT: "
; D$(6, I): PRINT
790 NEXT I: PRINT
800 IF F=0 THEN PRINT "JE N'AI RIEN TROUVE"
810 IF INKEY$="" THEN 810 ELSE 600
880 REM -----
890 REM MODIFICATION
900 CLS: PRINT "MODIFICATION": PRINT
910 INPUT "NUMERO DU CRU"; NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 6
940 PRINT I; L$(I); ": "; D$(I, NU)
950 NEXT I
960 PRINT: INPUT "MODIFICATION DU"; R
970 IF R<1 OR R>6 THEN 200
980 PRINT L$(R); ": "; : LINE INPUT D$(R, NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS: PRINT "SUPPRESSION": PRINT
1110 INPUT "NUMERO DU CRU"; NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 6
1150 PRINT I; L$(I); ": "; D$(I, NU)
1160 NEXT I: PRINT
1170 INPUT "VOUS ETES SUR DE SUPPRIMER CE CRU"; R$
1180 IF R$<>"0" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 6
1220 D$(J, I)=D$(J, I+1)
1230 NEXT J, I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS: F=0
1310 FOR I=1 TO N
1320 F=F+4: PRINT USING "### "; I;
1330 PRINT D$(1, I); " ("; D$(2, I); ")"
1340 PRINT TAB(5); "MILLESIME: "; D$(3, I); TAB(20); "NB
DE B. "; D$(4, I)
1350 PRINT TAB(9); "APRES: "; D$(5, I); TAB(23); "AVANT:
"; D$(6, I): PRINT
1360 IF F>16 THEN PRINT: PRINT "APPUYEZ SUR ENTER PO
UR LA SUITE"; : LINE INPUT R$: CLS: F=0
1370 NEXT I
1380 PRINT: PRINT "APPUYEZ SUR UNE TOUCHE POUR LE ME
NU";
1390 IF INKEY$="" THEN 1390 ELSE 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS: PRINT "APPUYEZ SUR ENTER QUAND LE DATACORD
ER EST PRET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPENOUT "!CAVE"
1530 PRINT#9, N
1540 FOR I=1 TO N
1550 FOR J=1 TO 6
1560 PRINT#9, D$(J, I)
1570 NEXT J, I
1580 CLOSEOUT
1590 CLS: PRINT "A LA PROCHAINE BOUTEILLE !"
1600 END

```

LE PROGRAMME

Lignes 30 à 160 : initialisation

On place dans le tableau L\$() les libellés des six données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'un cru

On saisit par une boucle FOR/NEXT les six données du fichier. Si aucun cru n'a été entré, on revient au menu (ligne 450).

Lignes 600 à 810 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$ pour les quatre premières, dates de début DD et de fin DF pour une période).

Les lignes 720 à 790 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$ ou comprise entre les deux dates DD et DF.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1390 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1360 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
DD	date de début des recherches
DF	date de fin des recherches
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre d'opérations dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$

```
RECHERCHE PAR :
1 CRU
2 APPELLATION
3 MILLESIME
4 NB DE BOUTEILLES
5 MEILLEUR APRES
6 MEILLEUR AVANT

QUEL TYPE DE DONNEE ?

DATE DEBUT ? 80
DATE FIN ? 82

1 DOM PERIGNON (CHAMPAGNE)
  MILLESIME:76   NB DE B.:8
  APRES:82      AVANT:86

2 CH. MARGAUX (BORDEAUX)
  MILLESIME:73   NB DE B.:3
  APRES:81      AVANT:93
```


REPERTOIRE TELEPHONIQUE

Pourquoi ne pas simplifier votre annuaire et vous réserver quelques numéros de téléphone ?


Ce fichier vous permettra de garder les numéros d'une quarantaine de personnes.

Le programme affiche un menu de trois options :

- 1 – **SAISIE** : avant de saisir un nom et un numéro de téléphone, AMSTRAD affiche sur l'écran la totalité du fichier. Vous devez donner en priorité un numéro de libre pour une nouvelle saisie ou le numéro d'enregistrement que vous désirez modifier. Il faut répondre "0" à la question "Numéro:" pour revenir au menu.
- 2 – **RECHERCHE** : dès que vous aurez donné le nom à rechercher, cette option affiche l'intégralité du fichier et s'arrête après un bip sonore sur les noms correspondants. Il n'est pas nécessaire de donner un nom en entier. Ceci permet de retrouver toutes les personnes d'une même famille.
- 3 – **SAUVEGARDE** : sauve le fichier sur cassette.



AGENDA TELEPHONIQUE



```

10 REM REPERTOIRE TELEPHONIQUE
20 REM -----
30 DIM N$(40),T$(40)
40 MODE 1:INK 0,0:INK 1,15:PAPER 0:PEN 1:BORDER 0
50 INPUT"PREMIERE UTILISATION (O/N)";R$
60 IF R$="O" THEN 200
70 CLS:PRINT"METTEZ LE DATACORDER EN LECTURE"
80 OPENIN "!TEL"
90 FOR I=1 TO 40
100 INPUT#9,N$(I),T$(I)
110 NEXT I
120 CLOSEIN
180 REM -----
190 REM MENU
200 CLS
210 PRINT"1 SAISIE PERSONNE"
220 PRINT"2 RECHERCHE"
230 PRINT"3 SAUVEGARDE"
240 INPUT"VOTRE CHOIX";V
250 IF V<1 OR V>3 THEN 200
260 ON V GOTO 300,500,600
280 REM -----
290 REM SAISIE
300 GOSUB 800
310 LOCATE 1,22
320 INPUT"NUMERO";NUS:NU=VAL(NUS)
330 IF NU=0 THEN 200
340 LOCATE 1,22:PRINT STRING$(35,32);CHR$(13);N$(N
U);"   ";T$(NU)
350 INPUT"NOM";N$
360 IF N$<>" " THEN N$(NU)=N$
370 INPUT"TELEPHONE";T$
380 IF T$<>" " THEN T$(NU)=T$
390 GOTO 300
480 REM -----
490 REM RECHERCHE
500 CLS
510 INPUT"NOM A RECHERCHER";NR$
520 GOSUB 800
530 GOTO 200
580 REM -----
590 REM SAUVEGARDE
600 CLS:PRINT"APPUYEZ SUR ENTER QUAND LE DATACORDE
R EST PRET POUR L'ENREGISTREMENT"
610 LINE INPUT R$
620 OPENOUT "!TEL"
630 FOR I=1 TO 40
640 PRINT#9,N$(I);", ";T$(I)
650 NEXT I
660 CLOSEOUT
670 CLS:PRINT"A LA PROCHAINE FOIS !"
680 END
780 REM -----
790 REM AFFICHE LES NOMS
800 CLS:X=1:Y=1
810 FOR I=1 TO 40
820 LOCATE X,Y
830 IF N$(I)="" THEN PRINT I:GOTO 860
840 IF V=1 THEN PRINT I;LEFT$(N$(I),6);" ";LEFT$(T
$(I),10) ELSE PRINT I;N$(I);"   ";T$(I)
850 IF V=2 AND LEFT$(N$(I),LEN(NR$))=NR$ THEN SOUN
D 1,340,2:WHILE INKEY$="":WEND
860 Y=Y+1:IF Y=21 AND V=1 THEN Y=1:X=21
870 IF Y=21 THEN WHILE INKEY$="":WEND:CLS:Y=1
880 NEXT I
890 RETURN

```

LE PROGRAMME

Lignes 30 à 120 : initialisation

Les tableaux N\$() et T\$() correspondent aux noms et numéros de téléphone du fichier. Si vous trouvez que ce n'est pas assez, il est possible de modifier leurs indices maximum (ici 40), par une autre valeur supérieure. Il faut pour cela modifier les chiffres 40 aux lignes 30, 90, 630 et 810.

Si ce n'est pas la première utilisation, on lit sur la cassette le fichier déjà créé antérieurement.

Lignes 200 à 260 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur (variable V).

Lignes 300 à 390 : saisie

On fait d'abord appel au sous-programme en 800 (affiche les noms).

Puis le programme demande le numéro que l'on veut créer ou modifier. Si on répond 0, la ligne 330 renvoie au menu. Dans le cas contraire, les lignes 350 et 370 saisissent le nom et le numéro de téléphone.

Lignes 500 à 530 : recherches

Le programme place dans NR\$ le nom que l'on veut rechercher. Le sous-programme en 800 affiche alors tous les noms correspondants.

Lignes 600 à 680: sauvegarde

Ces lignes sauvent sur cassette le fichier dès qu'une touche a été appuyée.

Lignes 800 à 890 : affichent les noms

Ce sous-programme affiche tous les noms et numéros de téléphone du fichier pour la création (V=1), ou uniquement ceux commençant par la chaîne de caractères NR\$ pour la recherche (V=2).

LISTE DES VARIABLES :

I	indice de boucle FOR/NEXT
NU	numéro d'enregistrement dans le fichier (numérique)
N\$	dernier nom saisi
N\$()	liste des noms
NR\$	nom à rechercher
NU\$	numéro d'enregistrement dans le fichier
R\$	réponse de l'utilisateur
T\$	dernier numéro de téléphone saisi
T\$()	liste des numéros de téléphone
V	choix de l'utilisateur
X et Y	coordonnées sur l'écran du prochain nom à afficher

CONSEILS DE LECTURE

Pour approfondir vos connaissances en BASIC, et mieux connaître le système des CPC 464, 664 et 6128, ainsi que du PCW 8256, P.S.I. vous propose une palette d'ouvrages utiles.

POUR JOUER AVEC OU CONTRE VOTRE AMSTRAD

- **Super jeux Amstrad**, par Jean-François Sehan (Editions du P.S.I.) :
50 programmes BASIC sont ici abondamment commentés et totalement ouverts à toute modification et adaptation. Le niveau de programmation est supérieur à celui du « 102 programmes ».
- **102 programmes pour Amstrad**, par Jacques Deconchat (Editions du P.S.I.) :
Pour débutant en BASIC, voici de nombreux programmes de jeu qui vous permettront, niveau par niveau, de maîtriser le BASIC de votre Amstrad.

POUR MAITRISER LE BASIC AMSTRAD

- **BASIC Amstrad – 1. Méthodes pratiques**, par Jacques Boisgontier et Bruno Césard (Editions du P.S.I.) :
Pour ceux qui connaissent déjà un BASIC, voici un bon ouvrage de perfectionnement à la programmation en BASIC sur Amstrad 464, 664 et 6128.
- **BASIC Amstrad – 2. Programmes et fichiers**, par Jacques Boisgontier (Editions du P.S.I.) :
Pour pratiquer le BASIC Amstrad, cet ouvrage donne de nombreux programmes de gestion, d'éducation et de jeu où le rôle des fichiers est expliqué et largement commenté.
- **BASIC plus 80 routines sur Amstrad**, par Michel Martin (Editions du P.S.I.) :
Pour pousser votre Amstrad au maximum de ses capacités : 80 routines de simulation d'instructions qui n'existent pas en BASIC Amstrad.

POUR MIEUX CONNAITRE LE SYSTEME DES CPC ET DU PCW 8256

- **Clefs pour Amstrad 1 – Système de base**, par Daniel Martin (Editions du P.S.I.) :
Mémento présentant synthétiquement le jeu d'instructions du Z80, les points d'entrée des routines système, les connecteurs et brochages, etc. Le livre de chevet du programmeur sur Amstrad.
- **Clefs pour Amstrad 2 – Système disque**, par Daniel Martin et Philippe Jadoul (Editions du P.S.I.) :
Ce deuxième tome consacré au système disque présente les points d'entrée des routines disque, les blocs de contrôle, la programmation et les brochages des circuits spécialisés... La deuxième partie du livre est aussi destinée aux possesseurs d'Amstrad 8256.
- **CP/M plus sur Amstrad**, par Yvon Dargery (Editions du P.S.I.) :
Toutes les commandes CP/M et CP/M plus pour maîtriser le système des 6128 et 8256 : un ouvrage de référence illustré par de nombreux programmes.
- **Le livre de l'Amstrad, tome 1**, par Daniel Martin et Philippe Jadoul (BCM - diffusé par P.S.I.) :
Ce livre, destiné aux programmeurs des CPC 464 et 664, donne une étude complète de tous les circuits internes, et analyse la structure interne du BASIC. Vous trouverez, en outre, une étude complète des RSX, et des programmes de scrolling, de traçage de rectangles, de coloriage de surface et de manipulation vectorielle.

POUR ETRE INFORME REGULIEREMENT DE L'ACTUALITE DES MICROS AMSTRAD

- **MICROSTRAD**, revue bimestrielle du Groupe Tests.
Pour exploiter au mieux les capacités de votre micro, vous trouverez au sommaire de chaque numéro un rendez-vous avec les rubriques clés :
 - **DECOUVREZ... LA FACE CACHEE DE VOTRE CPC** : astuces, idées, conseils, tout pour comprendre votre micro, son anatomie, son fonctionnement, sa programmation et exploiter ses capacités graphiques et sonores.

-
- DOMPTEZ... VOTRE CPC 464, 664, 6128 ou PCW 6256 : passionnés, petits ou grands, spécialistes ou débutants, une information pratique et la compétence d'experts au service de votre micro.
 - PROGRAMMEZ... VOTRE MICRO AMSTRAD : dans chaque numéro de MICROSTRAD, un cocktail de programmes (dessins, jeux, utilitaires, gestion, etc.) et des trucs de programmation.

Votre avis nous intéresse

- Pour nous permettre de faire de meilleurs livres, adressez-nous vos critiques sur le présent livre.
- Si vous souhaitez des éclaircissements techniques, écrivez-nous, nous adresserons votre demande à l'auteur qui ne manquera pas de vous répondre directement.

- Ce livre vous donne-t-il toute satisfaction?

- Y a-t-il un aspect du problème que vous auriez aimé voir abordé?

Comment avez-vous eu connaissance de ce livre?

- | | |
|---|-------------------------------------|
| <input type="checkbox"/> publicité | <input type="checkbox"/> cadeau |
| <input type="checkbox"/> catalogue | <input type="checkbox"/> librairie |
| <input type="checkbox"/> boutique micro | <input type="checkbox"/> exposition |
| <input type="checkbox"/> autres | |

Avez-vous déjà acquis des livres PSI?

lesquels? _____

qu'en pensez-vous? _____

Nom _____ Prénom _____ Age _____

Adresse _____

Profession _____

Centre d'intérêt _____

CATALOGUE GRATUIT

Vous pouvez obtenir un catalogue complet des ouvrages PSI, sur simple demande, ou en retournant cette page remplie à votre libraire, à votre boutique micro ou aux

Editions du PSI
BP 86
77402 Lagny-sur-Marne Cedex

**Calculer le coût annuel de votre voiture,
apprendre à taper au clavier ou obtenir
des idées de recettes pour votre soirée,
remonter votre arbre généalogique
ou déterminer la période favorable
de votre biorythme, telles sont les rubriques
que vous propose "Amstrad en famille".**

**40 programmes utiles feront de votre
CPC 464 un outil indispensable et
vous aideront à prendre des décisions à bon escient.**

**Tous les programmes contenus dans ce livre
fonctionnent également sur CPC 664 et 6128.**

Ce livre est composé de huit parties :

- **les finances,**
- **la pédagogie,**
- **la cuisine,**
- **les jeux nationaux,**
- **le temps,**
- **votre forme,**
- **le bricolage**
et pour finir
- **quelques utilitaires.**

**Tous les programmes, abondamment commentés,
sont détaillés dans une analyse
ligne par ligne, tandis que leur structure
est représentée systématiquement
par un organigramme.**



**EDITIONS DU P.S.I.
BP 86**

FRANCE - 77402 LAGNY/MARNE CEDEX

ISBN 2.86595.240.1 Imprimé en France

120 FF

MANAGERIAL ECONOMICS

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic

40 programmes utilises on Basic



Document numérisé avec amour par

AMSTRAD

CPC 

MÉMOIRE ÉCRITE



<https://acpc.me/>